Archival-Notebook

This is the accompanying Mathematica notebook for the computations supporting the paper "**COUPLED KPZ EQUATIONS AND THEIR DECOUPLEABILITY**" by Fu, Funaki, Sethuraman, and Venkataramani.

Table of contents:

Sec. 2.2

Sec. 4

Sec. 7.2

Sec. 7.3

Section 2.2: ODE analysis of decoupleability for n = 2.

In[1]:=
```
n = 2;
vars = Table[x[i], {i, 1, n}];
ϵ = Table[0, {i, 1, n}, {j, 1, n}];
ϵ[[1, 2]] = -1; ϵ[[2, 1]] = 1;
```

We being by defining the variables and constructing ϵ permutation matrix/generator for rotations in the plane, as it is used in the Proof of Prop. 2.5

In[5]:=
```
f = a3 x[1]^3 + 3 a2 x[1]^2 × x[2] + 3 a1 x[1] x[2]^2 + a0 x[2]^3
```

Out[5]= $a_3 x[1]^3 + 3 a_2 x[1]^2 x[2] + 3 a_1 x[1] x[2]^2 + a_0 x[2]^3$

In[6]:=
```
Γ = Table[Simplify[D[f / 6, x[i], x[j], x[k]]], {i, 1, n}, {j, 1, n}, {k, 1, n}];
```

In[7]:=
```
dθΓ =
 Table[Sum[Γ[[m, j, k]] × ϵ[[i, m]] + Γ[[i, m, k]] × ϵ[[j, m]] + Γ[[i, j, m]] × ϵ[[k, m]], {m, 1, n}],
  {i, 1, n}, {j, 1, n}, {k, 1, n}]
```

Out[7]= $\{\{\{-3 a_2, -2 a_1 + a_3\}, \{-2 a_1 + a_3, -a_0 + 2 a_2\}\}, \{\{-2 a_1 + a_3, -a_0 + 2 a_2\}, \{-a_0 + 2 a_2, 3 a_1\}\}\}$

The evolution of the tensor under rotations

In[8]:=
```
dΓvec = {dθΓ[[2, 2, 2]], dθΓ[[2, 2, 1]], dθΓ[[2, 1, 1]], dθΓ[[1, 1, 1]]}
```

Out[8]= $\{3 a_1, -a_0 + 2 a_2, -2 a_1 + a_3, -3 a_2\}$

The tensor and its derivative are represented as 4 by 1 column vectors.

In[9]:=
```
ℒ = Grad[dΓvec, {a0, a1, a2, a3}]
```

Out[9]= $\{\{0, 3, 0, 0\}, \{-1, 0, 2, 0\}, \{0, -2, 0, 1\}, \{0, 0, -3, 0\}\}$

Matrix representation of the rotation generator.

In[10]:=
```
MatrixForm[ℒ]
```

Out[10]//MatrixForm=

$$\begin{pmatrix} 0 & 3 & 0 & 0 \\ -1 & 0 & 2 & 0 \\ 0 & -2 & 0 & 1 \\ 0 & 0 & -3 & 0 \end{pmatrix}$$

In[11]:=
```
{vals, vecs} = Eigensystem[ℒ]
```

Out[11]=
```
{{3 ⅈ, -3 ⅈ, ⅈ, -ⅈ}, {{ⅈ, -1, -ⅈ, 1}, {-ⅈ, -1, ⅈ, 1}, {-3 ⅈ, 1, -ⅈ, 3}, {3 ⅈ, 1, ⅈ, 3}}}
```

In[12]:=
```
Λ = DiagonalMatrix[vals]
```

Out[12]=
```
{{3 ⅈ, 0, 0, 0}, {0, -3 ⅈ, 0, 0}, {0, 0, ⅈ, 0}, {0, 0, 0, -ⅈ}}
```

If we treat vecs as a matrix instead of a list of vectors, each eigenvector will be treated as a row. To make them columns, as appropriate for a right eigenvector, we need to take a transpose.

In[13]:=
```
ℒ.Transpose[vecs] - Transpose[vecs].Λ
```

Out[13]=
```
{{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}
```

Transpose[vecs] gives the right eigenvectors of ℒ. To get the Left eigenvectors as rows, we need to invert Transpose[vecs]. Below, we include an additional normalization to clear denominators.

In[14]:=
```
Lvecs = Sqrt[Det[vecs]] Inverse[Transpose[vecs]]
```

Out[14]=
```
{{1, -3 ⅈ, -3, ⅈ}, {-1, -3 ⅈ, 3, ⅈ}, {-1, ⅈ, -1, ⅈ}, {1, ⅈ, 1, ⅈ}}
```

These are the left Eigenvectors of ℒ. Lets Check

In[15]:=
```
Lvecs.ℒ - Λ.Lvecs
```

Out[15]=
```
{{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}
```

In[16]:=
```
MatrixForm[Lvecs]
```

Out[16]//MatrixForm=

$$\begin{pmatrix} 1 & -3 i & -3 & i \\ -1 & -3 i & 3 & i \\ -1 & i & -1 & i \\ 1 & i & 1 & i \end{pmatrix}$$

This is the matrix E in Sec. 2.2 of the paper and the corresponding eigenvalues are λ_1=3i, λ_2=-3i, λ_3=i, λ_4=-i.

For a diagonal tensor (β_1, 0, 0, β_2) we get v_i= β_1+ i β_2 = v_4. This gives Exp[4 i θ] = $v_4(0)$ / $v_1(0)$ = $\frac{(a_0+a_2)+i\,(a_1+a_3)}{(a_0-3\,a_2)+i\,(-3\,a_1+a_3}$. Subject to the necessary condition $(a_0 + a_2)^2 + (a_1 + a_3)^2 = (a_0 - 3\, a_2)^2 + (-3\, a_1 + a_3)^2$, we get 4 solutions for θ, say φ, $\varphi+\pi/2$, $\varphi+\pi$ and $\varphi+3\pi/2$.

These solutions define β_1 and β_2 by β_1+ i β_2 = Exp[-iφ] (($a_0 + a_2$) + $i\,(a_1 + a_3)$). The rotations $\varphi+\pi/2$, $\varphi+\pi$ and $\varphi+3\pi/2$ then correspond, respectively, to the diagonal tensors (β_2, 0, 0, -β_1), (−β_1, 0, 0, -β_2) and

$(-\beta_2, 0, 0, \beta_1)$ respectively.

Section 4

We begin to compute $\sigma_\theta \circ \Gamma$, the action of a rotation on the space of tensors $\mathcal{T}_2$.

In[17]:=
```
Clear[Γ];
n = 2;
vars = Table[x[i], {i, 1, n}];
Γlist = Flatten[Table[Γ[i, j, k], {i, 1, n}, {j, i, n}, {k, j, n}]];
rename = Table[Γlist〚m〛 → Subscript[a, 4 - m], {m, 1, 4}];
f =
 (Sum[Γ @@ Sort[{i, j, k}] × x[i] × x[j] × x[k], {i, 1, n}, {j, 1, n}, {k, 1, n}] /. rename)
```

Out[22]=

$a_3\, x[1]^3 + 3\, a_2\, x[1]^2\, x[2] + 3\, a_1\, x[1]\, x[2]^2 + a_0\, x[2]^3$

The last line is the expression for the cubic polynomial associate to a tensor Γ. Note that the coordinates are x[1] and x[2] where the indices are arguments and not subscripts.

In[23]:=
```
Clear[σ];
σ[θ_] = {{Cos[θ], - Sin[θ]}, {Sin[θ], Cos[θ]}};
MatrixForm[σ[θ]]
```

Out[25]//MatrixForm=

$$\begin{pmatrix} \mathrm{Cos}[\theta] & -\mathrm{Sin}[\theta] \\ \mathrm{Sin}[\theta] & \mathrm{Cos}[\theta] \end{pmatrix}$$

Multiplying by this rotation matrix on the left gives the action of SO(2) on $\mathbb{R}^2$ where the elements are thought of a column vectors. The action corresponds to rotating `counter-clockwise' by an angle θ.

The action on Tensors (or equivalently on polynomials) is given by $\sigma \circ \mathrm{f}(z) = f\left(\sigma^{-1} \cdot z\right)$ for $z \in \mathbb{R}^2$.

In[26]:=
```
Substitution = Table[x[i] → (σ[-θ].{z[1], z[2]})〚i〛, {i, 1, n}]
```

Out[26]=

{x[1] → Cos[θ] z[1] + Sin[θ] z[2], x[2] → -Sin[θ] z[1] + Cos[θ] z[2]}

In[27]:=
```
Transformedf = f //. Substitution
```

Out[27]=

$$\begin{aligned}&a_0\, (-\mathrm{Sin}[\theta]\, z[1] + \mathrm{Cos}[\theta]\, z[2])^3 + \\ &\;3\, a_1\, (-\mathrm{Sin}[\theta]\, z[1] + \mathrm{Cos}[\theta]\, z[2])^2\, (\mathrm{Cos}[\theta]\, z[1] + \mathrm{Sin}[\theta]\, z[2]) + \\ &\;3\, a_2\, (-\mathrm{Sin}[\theta]\, z[1] + \mathrm{Cos}[\theta]\, z[2])\, (\mathrm{Cos}[\theta]\, z[1] + \mathrm{Sin}[\theta]\, z[2])^2 + \\ &\;a_3\, (\mathrm{Cos}[\theta]\, z[1] + \mathrm{Sin}[\theta]\, z[2])^3\end{aligned}$$

This is a cubic polynomial in z[1],z[2]. We can now read off the transformations of the coefficients from $\sigma \circ \mathrm{f}(z) = b_3\, z[1]^3 + 3\, b_2\, z[1]^2\, z[2] + 3\, b_1\, z[1]\, z[2]^2 + b_0\, z[2]^3$. We now account for the factors of 3 in the coefficients b[1] and b[2] and order the coefficients as a column vector from b_0 to $b_{3.}$

In[28]:=
```
newcoeffs = Simplify[DiagonalMatrix[{1, 1/3, 1/3, 1}].
    CoefficientList[Transformedf /. {z[2] → 1}, z[1]]]
```

Out[28]=

$$\Big\{\mathrm{Cos}[\theta]^3\, a_0 + \mathrm{Sin}[\theta]\,\big(3\,\mathrm{Cos}[\theta]^2\, a_1 + \mathrm{Sin}[\theta]\,(3\,\mathrm{Cos}[\theta]\, a_2 + \mathrm{Sin}[\theta]\, a_3)\big),\ \frac{1}{4}\big(-4\,\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta]\, a_0 + (\mathrm{Cos}[\theta] + 3\,\mathrm{Cos}[3\,\theta])\, a_1 + 2\,\mathrm{Sin}[\theta]\,(a_2 + 3\,\mathrm{Cos}[2\,\theta]\, a_2 + \mathrm{Sin}[2\,\theta]\, a_3)\big),\ \mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]^2\, a_0 + \frac{1}{4}\,((\mathrm{Sin}[\theta] - 3\,\mathrm{Sin}[3\,\theta])\, a_1 + 2\,\mathrm{Cos}[\theta]\,((-1 + 3\,\mathrm{Cos}[2\,\theta])\, a_2 + \mathrm{Sin}[2\,\theta]\, a_3)),\ -\mathrm{Sin}[\theta]^3\, a_0 + \mathrm{Cos}[\theta]\,\big(3\,\mathrm{Sin}[\theta]^2\, a_1 + \mathrm{Cos}[\theta]\,(-3\,\mathrm{Sin}[\theta]\, a_2 + \mathrm{Cos}[\theta]\, a_3)\big)\Big\}$$

In[29]:=
```
Lσ = Grad[newcoeffs, Table[a_{i-1}, {i, 1, 4}]]
```

Out[29]=

$$\Big\{\big\{\mathrm{Cos}[\theta]^3,\ 3\,\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta],\ 3\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]^2,\ \mathrm{Sin}[\theta]^3\big\},\ \big\{-\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta],\ \frac{1}{4}\,(\mathrm{Cos}[\theta] + 3\,\mathrm{Cos}[3\,\theta]),\ \frac{1}{2}\,(1 + 3\,\mathrm{Cos}[2\,\theta])\,\mathrm{Sin}[\theta],\ \frac{1}{2}\,\mathrm{Sin}[\theta]\,\mathrm{Sin}[2\,\theta]\big\},\ \big\{\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]^2,\ \frac{1}{4}\,(\mathrm{Sin}[\theta] - 3\,\mathrm{Sin}[3\,\theta]),\ \frac{1}{2}\,\mathrm{Cos}[\theta]\,(-1 + 3\,\mathrm{Cos}[2\,\theta]),\ \frac{1}{2}\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[2\,\theta]\big\},\ \big\{-\mathrm{Sin}[\theta]^3,\ 3\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]^2,\ -3\,\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta],\ \mathrm{Cos}[\theta]^3\big\}\Big\}$$

In[30]:=
```
MatrixForm[Lσ]
```

Out[30]//MatrixForm=

$$\begin{pmatrix} \mathrm{Cos}[\theta]^3 & 3\,\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta] & 3\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]^2 & \mathrm{Sin}[\theta]^3 \\ -\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta] & \frac{1}{4}\,(\mathrm{Cos}[\theta] + 3\,\mathrm{Cos}[3\,\theta]) & \frac{1}{2}\,(1 + 3\,\mathrm{Cos}[2\,\theta])\,\mathrm{Sin}[\theta] & \frac{1}{2}\,\mathrm{Sin}[\theta]\,\mathrm{Sin}[2\,\theta] \\ \mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]^2 & \frac{1}{4}\,(\mathrm{Sin}[\theta] - 3\,\mathrm{Sin}[3\,\theta]) & \frac{1}{2}\,\mathrm{Cos}[\theta]\,(-1 + 3\,\mathrm{Cos}[2\,\theta]) & \frac{1}{2}\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[2\,\theta] \\ -\mathrm{Sin}[\theta]^3 & 3\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]^2 & -3\,\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta] & \mathrm{Cos}[\theta]^3 \end{pmatrix}$$

This is the representation of SO(2) on the space of 2 × 2 × 2 symmetric tensors. This representation is used in Sec. 4 of the paper. We can also determine the generator for this action.

In[31]:=
```
ℒ = D[Lσ, θ] /. {θ → 0}
```

Out[31]=

```
{{0, 3, 0, 0}, {-1, 0, 2, 0}, {0, -2, 0, 1}, {0, 0, -3, 0}}
```

In[32]:=
```
MatrixForm[ℒ]
```

Out[32]//MatrixForm=

$$\begin{pmatrix} 0 & 3 & 0 & 0 \\ -1 & 0 & 2 & 0 \\ 0 & -2 & 0 & 1 \\ 0 & 0 & -3 & 0 \end{pmatrix}$$

This is an alternate derivation to obtain the generator $\mathcal{L}$ in the proof of Prop. 2.5

Hilbert series for the SO(2) invariants.

In[33]:=
```
Simplify[1 / (2 π) Integrate[1 / Simplify[Det[IdentityMatrix[4] - λ Lσ]], {θ, 0, 2 π}]]
```

Out[33]=
$$\begin{cases} \frac{1+\lambda^4}{(-1+\lambda^2)^3 (1+\lambda^2)} & \text{Abs}[\lambda] > 1 \\ \frac{2+\lambda^{2/3} (-1-\lambda^{2/3}+\lambda^{4/3}) (1+\lambda^{2/3}+\lambda^{4/3}+2\lambda^2)}{3 (-1+\lambda^2)^3 (1+\lambda^2)} & \frac{1}{\text{Abs}[\lambda]^{1/3}} < 1 \text{ if } \text{Abs}[\lambda]^{1/3} \neq 1 \\ -\frac{1+\lambda^4}{(-1+\lambda^2)^3 (1+\lambda^2)} & \text{True} \end{cases}$$

We need the result for Abs[λ] < 1, so this is the last line in the piecewise defined integral.

In[34]:=
$$\Phi SO_2 = \text{Simplify}\left[-\frac{1+\lambda^4}{(-1+\lambda^2)^3 (1+\lambda^2)}\right]$$

Out[34]=
$$-\frac{1+\lambda^4}{(-1+\lambda^2)^3 (1+\lambda^2)}$$

Hilbert series for the O(2) invariants.

We now compute the action of O(2) by adding a reflection operator corresponding to x[1]-> x[1], x[2]->-x[2]. In terms of the tensor coefficients, this action is given by the matrix

In[35]:=
```
N = {{0, 0, 0, 1}, {0, 0, 1, 0}, {0, 1, 0, 0}, {1, 0, 0, 0}};
```

In[36]:=
```
Simplify[Det[IdentityMatrix[4] - λ N.Lσ]]
```

Out[36]=
$$(-1+\lambda^2)^2$$

This determinant does not depend explicitly on θ, so it is easy to integrate the reciprocal.

In[37]:=
```
ΦO2 = Simplify[ (1 / Det[IdentityMatrix[4] - λ N.Lσ] + ΦSO2) / 2 ]
```

Out[37]=
$$-\frac{1}{(-1+\lambda^2)^3 (1+\lambda^2)}$$

Computation of the invariants

In[39]:=
```
u = Simplify[ Grad[Laplacian[f, vars], vars] / 6]
```

Out[39]=
$$\{a_1 + a_3, a_0 + a_2\}$$

This is the trace vector. We ‘lift’ this vector to form the cubic polynomial f_1

In[40]:=
```
f1 = 3 u.vars (vars.vars) / (n + 2)
```

Out[40]=
$$\frac{3}{4} ((a_1 + a_3)\, x[1] + (a_0 + a_2)\, x[2]) \left(x[1]^2 + x[2]^2\right)$$

```
In[42]:= ℬ = Table[D[f1 / 6, x[i], x[j], x[k]], {i, 1, n}, {j, 1, n}, {k, 1, n}];

In[43]:= Table[MatrixForm[ℬ⟦i⟧], {i, 1, n}]
```

Out[43]=

$$\left\{\begin{pmatrix} \frac{3}{4}\ (a_1 + a_3) & \frac{1}{4}\ (a_0 + a_2) \\ \frac{1}{4}\ (a_0 + a_2) & \frac{1}{4}\ (a_1 + a_3) \end{pmatrix}, \begin{pmatrix} \frac{1}{4}\ (a_0 + a_2) & \frac{1}{4}\ (a_1 + a_3) \\ \frac{1}{4}\ (a_1 + a_3) & \frac{3}{4}\ (a_0 + a_2) \end{pmatrix}\right\}$$

$\mathcal{B}$ is the tensor corresponding to the trace part. With our normalization, the trace-free part is given by $f_3 = (n+2)\,f\ -\ f_1$

```
In[44]:= f3 = Collect[Expand[(n + 2) (f - f1)], vars]
```

Out[44]=

$$(-3\ a_1 + a_3)\ x[1]^3 + (-3\ a_0 + 9\ a_2)\ x[1]^2\ x[2] + (9\ a_1 - 3\ a_3)\ x[1]\ x[2]^2 + (a_0 - 3\ a_2)\ x[2]^3$$

f_3 corresponds to a trace-free $2 \times 2 \times 2$ symmetric tensor $\mathcal{D}$. To eliminate denominators, we multiply by a factor of (n+2), which equals 4 in the case n=2.

```
In[45]:= 𝒟 = Table[Simplify[D[f3 / 6, x[i], x[j], x[k]]], {i, 1, n}, {j, 1, n}, {k, 1, n}];

In[46]:= Table[MatrixForm[𝒟⟦i⟧], {i, 1, n}]
```

Out[46]=

$$\left\{\begin{pmatrix} -3\ a_1 + a_3 & -a_0 + 3\ a_2 \\ -a_0 + 3\ a_2 & 3\ a_1 - a_3 \end{pmatrix}, \begin{pmatrix} -a_0 + 3\ a_2 & 3\ a_1 - a_3 \\ 3\ a_1 - a_3 & a_0 - 3\ a_2 \end{pmatrix}\right\}$$

```
In[47]:= Dstarsqrd = Simplify[TensorContract[𝒟⊗𝒟, {{1, 4}, {2, 5}}]]
```

Out[47]=

$$\left\{\left\{2\ \left((a_0 - 3\ a_2)^2 + (-3\ a_1 + a_3)^2\right),\ 0\right\},\ \left\{0,\ 2\ \left((a_0 - 3\ a_2)^2 + (-3\ a_1 + a_3)^2\right)\right\}\right\}$$

```
In[51]:= w = Expand[𝒟.u.u]
```

Out[51]=

$$\left\{a_0^2\ a_1 - 3\ a_1^3 + 10\ a_0\ a_1\ a_2 + 9\ a_1\ a_2^2 - 3\ a_0^2\ a_3 - 5\ a_1^2\ a_3 + 2\ a_0\ a_2\ a_3 + 5\ a_2^2\ a_3 - a_1\ a_3^2 + a_3^3,\right.$$
$$\left. a_0^3 + 5\ a_0\ a_1^2 - a_0^2\ a_2 + 9\ a_1^2\ a_2 - 5\ a_0\ a_2^2 - 3\ a_2^3 + 2\ a_0\ a_1\ a_3 + 10\ a_1\ a_2\ a_3 - 3\ a_0\ a_3^2 + a_2\ a_3^2\right\}$$

This is the vector w defined right after (4.4).

```
In[53]:= Simplify[u.w]
```

Out[53]=

$$a_0^4 - 3\ a_1^4 - 3\ a_2^4 - 8\ a_1^3\ a_3 + 24\ a_1\ a_2^2\ a_3 + 6\ a_2^2\ a_3^2 + a_3^4 -$$
$$8\ a_0\ a_2\ \left(-3\ a_1^2 + a_2^2 - 3\ a_1\ a_3\right) + 6\ a_0^2\ \left(a_1^2 - a_2^2 - a_3^2\right) + 6\ a_1^2\ \left(3\ a_2^2 - a_3^2\right)$$

This defines the invariant I_4

```
In[54]:= Simplify[Det[{u, w}]]
```

Out[54]=

$$4\ \left(-3\ a_0^2\ a_1\ a_2 + a_0^3\ a_3 +\right.$$
$$\left. a_2\ \left(3\ a_1^3 + 6\ a_1^2\ a_3 - 2\ a_2^2\ a_3 - 3\ a_1\ \left(a_2^2 - a_3^2\right)\right) + a_0\ \left(2\ a_1^3 - 6\ a_1\ a_2^2 + 3\ a_1^2\ a_3 - a_3\ \left(3\ a_2^2 + a_3^2\right)\right)\right)$$

This defines the invariant m_4. We can now write down the definitions of the SO(2) and the ideal generated by these definitions,

In[55]:=
```
Clear[j, h, l, m]
Invariants =
 {j₂ - u.u, h₂ - Tr[TensorContract[𝒟⊗𝒟, {{1, 4}, {2, 5}}]], l₄ - w.u, m₄ - Det[{u, w}]}
```

Out[56]=

$$\left\{-(a_0+a_2)^2-(a_1+a_3)^2+j_2,\ -(a_0-3a_2)^2-3(-a_0+3a_2)^2-3(3a_1-a_3)^2-(-3a_1+a_3)^2+h_2,\right.$$
$$-\left((a_0+a_2)\left(a_0^3+5a_0a_1^2-a_0^2a_2+9a_1^2a_2-5a_0a_2^2-3a_2^3+2a_0a_1a_3+10a_1a_2a_3-3a_0a_3^2+a_2a_3^2\right)\right)-$$
$$(a_1+a_3)\left(a_0^2a_1-3a_1^3+10a_0a_1a_2+9a_1a_2^2-3a_0^2a_3-5a_1^2a_3+2a_0a_2a_3+5a_2^2a_3-a_1a_3^2+a_3^3\right)+$$
$$l_4,\ -8a_0a_1^3+12a_0^2a_1a_2-12a_1^3a_2+24a_0a_1a_2^2+12a_1a_2^3-4a_0^3a_3-$$
$$\left.12a_0a_1^2a_3-24a_1^2a_2a_3+12a_0a_2^2a_3+8a_2^3a_3-12a_1a_2a_3^2+4a_0a_3^3+m_4\right\}$$

This is the basis for the ideal of polynomials on $\mathbb{R}^8$ (corresponding to the 4 coefficients a_0,a_1,a_2,a_3 and the 4 invariants j_2,h_2,l_4,m_4

We seek potential relations among the invariants by finding a basis for the ideal generated by the definitions of the invariants intersected with the polynomials in j_2,h_2,l_4,m_4 that do not depend on a_0,a_1,a_2,a_3, i.e. we are eliminating the coefficients between the relations defining the invariants.

In[57]:=
```
GroebnerBasis[Invariants, {m₄, l₄, h₂, j₂},
 {a₀, a₁, a₂, a₃}, MonomialOrder → EliminationOrder]
```

Out[57]=

$$\left\{h_2\,j_2^3-4\,l_4^2-4\,m_4^2\right\}$$

We see that there is one identity that allows us to replace $m_4{}^2$ by an expression in the other invariants. There are no further relations, so this implies l_4, h_2 and j_2 are algebraically independent.

Alternate choices for the fundamental invariants are the trace and determinant of Γ^{*^2}, which are themselves O(2) invariants, so they can be expressed in terms of the fundamental invariants j_2, h_2 and l_4

In[58]:=
```
Γ = Table[D[f / 6, x[i], x[j], x[k]], {i, 1, n}, {j, 1, n}, {k, 1, n}];
Q = Simplify[TensorContract[Γ⊗Γ, {{1, 4}, {2, 5}}]];
MatrixForm[Q]
```

Out[59]//MatrixForm=

$$\begin{pmatrix} a_1^2+2a_2^2+a_3^2 & a_0a_1+a_2(2a_1+a_3) \\ a_0a_1+a_2(2a_1+a_3) & a_0^2+2a_1^2+a_2^2 \end{pmatrix}$$

This is the matrix Γ^{*2}.

In[60]:=
```
NewInvariants = {τ - Tr[Q], δ - Det[Q]}
```

Out[60]=

$$\left\{\tau-a_0^2-3a_1^2-3a_2^2-a_3^2,\right.$$
$$\left.\delta-2a_1^4+4a_0a_1^2a_2-2a_0^2a_2^2-a_1^2a_2^2-2a_2^4+2a_0a_1a_2a_3+4a_1a_2^2a_3-a_0^2a_3^2-2a_1^2a_3^2\right\}$$

These are the alternate invariants which are examples of the invariants used in the general framework for fully decoupled tensors in Sec. 6.

In[61]:=
```
GroebnerBasis[Join[NewInvariants, Invariants],
 {τ, δ, l₄, h₂, j₂}, {m₄, a₀, a₁, a₂, a₃}, MonomialOrder → EliminationOrder]
```

Out[61]=

$$\left\{16\,\tau-h_2-12\,j_2,\ -1024\,\delta+h_2^2+8\,h_2\,j_2+80\,j_2^2-128\,l_4\right\}$$

These are the relations right after (4.5) at the end of section 4. $\mathrm{Tr}\left[\Gamma^{*2}\right] = \frac{h_2+12\, j_2}{16}$, $\mathrm{Det}\left[\Gamma^{*2}\right] = \frac{h_2^2+8\, h_2\, j_2+80\, j_2^2-128\, l_4}{1024}$

Section 7.2

Fully decoupleable 3 × 3 × 3 tensors

In[71]:=
```
n = 3; vars = Table[x[i], {i, 1, n}]; f = Sum[βi x[i]^3, {i, 1, n}]
```

Out[71]=

$$\beta_1\, x[1]^3 + \beta_2\, x[2]^3 + \beta_3\, x[3]^3$$

This is the cubic polynomial corresponding to a fully decoupleable tensor.

In[72]:=
```
Γ = Simplify[Table[D[f, x[i], x[j], x[k]], {k, 1, n}, {j, 1, n}, {i, 1, n}] / 6];
u = Simplify[Table[D[Laplacian[f, vars], x[i]], {i, 1, n}] / 6];
f3 = (n + 2) f - 3 (u.vars) (vars.vars)
```

Out[74]=

$$-3\, (\beta_1\, x[1] + \beta_2\, x[2] + \beta_3\, x[3])\, \left(x[1]^2 + x[2]^2 + x[3]^2\right) + 5\, \left(\beta_1\, x[1]^3 + \beta_2\, x[2]^3 + \beta_3\, x[3]^3\right)$$

This is the trace-free part

In[75]:=
```
𝒟 = Simplify[Table[D[f3, x[i], x[j], x[k]], {k, 1, n}, {j, 1, n}, {i, 1, n}] / 6];
v = Simplify[TensorContract[(𝒟⊗𝒟)⊗𝒟, {{1, 4}, {2, 5}, {3, 7}, {6, 8}}]];
w = Simplify[𝒟.u.u];
```

We have now computed the vectors u,v and w and the trace-free tensor $\mathcal{D}$, which are the ingredients needed to compute the Integrity basis given by Olive and Auffray.

In[78]:=
```
Clear[H, J, K, L, M, Q];
Q = Simplify[TensorContract[𝒟⊗𝒟, {{1, 4}, {2, 5}}]];
γuu = w;
Coeffs = Table[βi, {i, 1, n}];
Trivialize = Table[Coeffs[[j]] → 0, {j, 1, Length[Coeffs]}];
IntegrityPolys =
  {H[2] - Simplify[Tr[Q]], H[4] - Simplify[Tr[Q.Q]], J[2] - Simplify[u.u],
   L[4] - Simplify[γuu.u], H[6] - Simplify[v.v], H[10] - Simplify[𝒟.v.v.v] ,
   J[4] - Simplify[u.Q.u], K[4] - Simplify[Tr[Q.(𝒟.u)]],
   J[6] - Simplify[(u.Q).γuu], K[6] - Simplify[v.w], L[6] - Simplify[(u.Q).v],
   M[6] - Simplify[γuu.γuu], H[8] - Simplify[(u.Q).(Q.v)]};
OAInvariants = IntegrityPolys /. Trivialize;
```

The Ideal IntegrityPolys is generated by the polynomials defining the Integrity basis elements in terms of the coefficients of a fully decoupled tensor, with the labels of the Integrity invariants as the slack variables.

In[84]:=
```
IntegrityPolys
```
Out[84]=

$$\{H[2]-10\left(\beta_1^2+\beta_2^2+\beta_3^2\right),\ H[4]-44\beta_1^4-44\beta_2^4-58\beta_2^2\beta_3^2-44\beta_3^4-58\beta_1^2\left(\beta_2^2+\beta_3^2\right),$$
$$J[2]-\beta_1^2-\beta_2^2-\beta_3^2,\ L[4]-2\left(\beta_1^4+\beta_2^4-3\beta_2^2\beta_3^2+\beta_3^4-3\beta_1^2\left(\beta_2^2+\beta_3^2\right)\right),$$
$$H[6]-4\left(\beta_3^2\left(\beta_1^2+\beta_2^2-4\beta_3^2\right)^2+\beta_2^2\left(\beta_1^2-4\beta_2^2+\beta_3^2\right)^2+\beta_1^2\left(-4\beta_1^2+\beta_2^2+\beta_3^2\right)^2\right),$$
$$H[10]-8\left(128\beta_1^{10}+128\beta_2^{10}-60\beta_2^8\beta_3^2-95\beta_2^6\beta_3^4-95\beta_2^4\beta_3^6-\right.$$
$$60\beta_2^2\beta_3^8+128\beta_3^{10}-60\beta_1^8\left(\beta_2^2+\beta_3^2\right)+\beta_1^6\left(-95\beta_2^4+60\beta_2^2\beta_3^2-95\beta_3^4\right)+$$
$$\left.\beta_1^4\left(-95\beta_2^6+90\beta_2^4\beta_3^2+90\beta_2^2\beta_3^4-95\beta_3^6\right)-30\beta_1^2\left(2\beta_2^8-2\beta_2^6\beta_3^2-3\beta_2^4\beta_3^4-2\beta_2^2\beta_3^6+2\beta_3^8\right)\right),$$
$$J[4]-2\left(3\beta_1^4+3\beta_2^4+\beta_2^2\beta_3^2+3\beta_3^4+\beta_1^2\left(\beta_2^2+\beta_3^2\right)\right),\ K[4]-8\beta_1^4-8\beta_2^4+4\beta_2^2\beta_3^2-8\beta_3^4+4\beta_1^2\left(\beta_2^2+\beta_3^2\right),$$
$$J[6]-12\beta_1^6-12\beta_2^6+19\beta_2^4\beta_3^2+19\beta_2^2\beta_3^4-12\beta_3^6+19\beta_1^4\left(\beta_2^2+\beta_3^2\right)+\beta_1^2\left(19\beta_2^4+18\beta_2^2\beta_3^2+19\beta_3^4\right),$$
$$K[6]-2\left(8\beta_1^6+8\beta_2^6-11\beta_2^4\beta_3^2-11\beta_2^2\beta_3^4+8\beta_3^6-11\beta_1^4\left(\beta_2^2+\beta_3^2\right)+\beta_1^2\left(-11\beta_2^4+18\beta_2^2\beta_3^2-11\beta_3^4\right)\right),$$
$$L[6]-6\left(8\beta_1^6+8\beta_2^6-\beta_2^4\beta_3^2-\beta_2^2\beta_3^4+8\beta_3^6-\beta_1^4\left(\beta_2^2+\beta_3^2\right)-\beta_1^2\left(\beta_2^2+\beta_3^2\right)^2\right),$$
$$M[6]-\beta_3^2\left(3\beta_1^2+3\beta_2^2-2\beta_3^2\right)^2-\beta_2^2\left(3\beta_1^2-2\beta_2^2+3\beta_3^2\right)^2-\left(2\beta_1^3-3\beta_1\left(\beta_2^2+\beta_3^2\right)\right)^2,$$
$$H[8]-4\left(72\beta_1^8+18\beta_1^6\left(\beta_2^2+\beta_3^2\right)-11\beta_1^4\left(3\beta_2^4+\beta_2^2\beta_3^2+3\beta_3^4\right)+\right.$$
$$\left.\beta_1^2\left(18\beta_2^6-11\beta_2^4\beta_3^2-11\beta_2^2\beta_3^4+18\beta_3^6\right)+3\left(24\beta_2^8+6\beta_2^6\beta_3^2-11\beta_2^4\beta_3^4+6\beta_2^2\beta_3^6+24\beta_3^8\right)\right)\}$$

Characteristic Polynomial coefficients

In[85]:=
```
Γstarsqrd = Simplify[TensorContract[Γ⊗Γ, {{1, 4}, {2, 5}}]];
```

In[86]:=
```
Clear[q, ξ];
ξ = Rest[Reverse[CoefficientList[Simplify[Det[λ IdentityMatrix[3] + Γstarsqrd]], λ]]]
```
Out[86]=

$$\{\beta_1^2+\beta_2^2+\beta_3^2,\ \beta_1^2\beta_2^2+\beta_1^2\beta_3^2+\beta_2^2\beta_3^2,\ \beta_1^2\beta_2^2\beta_3^2\}$$

As expected, these are the elementary symmetric polynomials of the quantities β_i^2.

In[87]:=
```
DiagInvars = Table[qᵢ - ξ⟦i⟧, {i, 1, 3}]
```
Out[87]=

$$\{q_1-\beta_1^2-\beta_2^2-\beta_3^2,\ q_2-\beta_1^2\beta_2^2-\beta_1^2\beta_3^2-\beta_2^2\beta_3^2,\ q_3-\beta_1^2\beta_2^2\beta_3^2\}$$

This is the basis of invariants for the group $G_R = S_3\times(\mathbb{Z}_2)^3$. Since all the Olive and Auffray invariants, when restricted to fully decoupled tensors, are also G_R invariants, we can express them in terms of the quantities q_i

In[88]:=
```
Table[GroebnerBasis[Join[{IntegrityPolys⟦i⟧}, DiagInvars],
   Join[{OAInvariants⟦i⟧}, {q₁, q₂, q₃}], {β₁, β₂, β₃},
   MonomialOrder → EliminationOrder]⟦1⟧, {i, 1, Length[OAInvariants]}]
```
Out[88]=

$$\{H[2]-10q_1,\ -H[4]+44q_1^2-30q_2,\ J[2]-q_1,$$
$$-L[4]+2q_1^2-10q_2,\ -H[6]+64q_1^3-220q_1q_2+300q_3,$$
$$-H[10]+1024q_1^5-5600q_1^3q_2+5800q_1q_2^2+7600q_1^2q_3-7000q_2q_3,$$
$$-J[4]+6q_1^2-10q_2,\ -K[4]+8q_1^2-20q_2,\ -J[6]+12q_1^3-55q_1q_2+75q_3,$$
$$-K[6]+16q_1^3-70q_1q_2+150q_3,\ -L[6]+48q_1^3-150q_1q_2+150q_3,$$
$$-M[6]+4q_1^3-15q_1q_2+75q_3,\ -H[8]+288q_1^4-1080q_1^2q_2+300q_2^2+1300q_1q_3\}$$

This is the ideal corresponding to the relations in Eq. (7.1).

Section 7.3

Partially decoupleable 3 × 3 × 3 tensors

In[89]:=
```
n = 3;
vars = Table[x[i], {i, 1, n}];
f = 3 α x[1] (x[1]^2 + x[2]^2) +
   γ1 (3 x[2]^2 × x[1] - x[1]^3) + γ2 (3 x[1]^2 × x[2] - x[2]^3) + β3 x[3]^3
```

Out[89]=

$$3\,\alpha\,x[1]\,\left(x[1]^2+x[2]^2\right)+\gamma_1\left(-x[1]^3+3\,x[1]\,x[2]^2\right)+\gamma_2\left(3\,x[1]^2\,x[2]-x[2]^3\right)+\beta_3\,x[3]^3$$

This is the canonical form corresponding to a partially decoupleable tensor.

In[90]:=
```
Coeffs = {α, γ1, γ2, β3};
```

This is the list ot tensor coefficients in the canonical form.

In[91]:=
```
Γ = Simplify[Table[D[f, x[i], x[j], x[k]], {k, 1, n}, {j, 1, n}, {i, 1, n}] / 6]
```

Out[91]=

$$\{\{\{3\,\alpha-\gamma_1,\gamma_2,0\},\{\gamma_2,\alpha+\gamma_1,0\},\{0,0,0\}\},\{\{\gamma_2,\alpha+\gamma_1,0\},\{\alpha+\gamma_1,-\gamma_2,0\},\{0,0,0\}\},\{\{0,0,0\},\{0,0,0\},\{0,0,\beta_3\}\}\}$$

In[92]:=
```
Table[MatrixForm[Γ[[i]]], {i, 1, n}]
```

Out[92]=

$$\left\{\begin{pmatrix}3\,\alpha-\gamma_1 & \gamma_2 & 0\\ \gamma_2 & \alpha+\gamma_1 & 0\\ 0 & 0 & 0\end{pmatrix},\begin{pmatrix}\gamma_2 & \alpha+\gamma_1 & 0\\ \alpha+\gamma_1 & -\gamma_2 & 0\\ 0 & 0 & 0\end{pmatrix},\begin{pmatrix}0 & 0 & 0\\ 0 & 0 & 0\\ 0 & 0 & \beta_3\end{pmatrix}\right\}$$

In[93]:=
```
u = Simplify[Table[D[Laplacian[f, vars], x[i]], {i, 1, n}] / 6];
f3 = (n + 2) f - 3 (u.vars) (vars.vars)
```

Out[94]=

$$-3\,(4\,\alpha\,x[1]+\beta_3\,x[3])\left(x[1]^2+x[2]^2+x[3]^2\right)+5\left(3\,\alpha\,x[1]\left(x[1]^2+x[2]^2\right)+\gamma_1\left(-x[1]^3+3\,x[1]\,x[2]^2\right)+\gamma_2\left(3\,x[1]^2\,x[2]-x[2]^3\right)+\beta_3\,x[3]^3\right)$$

This is the trace-free part

In[95]:=
```
𝒟 = Simplify[Table[D[f3, x[i], x[j], x[k]], {k, 1, n}, {j, 1, n}, {i, 1, n}] / 6];
v = Simplify[TensorContract[(𝒟⊗𝒟)⊗𝒟, {{1, 4}, {2, 5}, {3, 7}, {6, 8}}]];
w = Simplify[𝒟.u.u];
```

The calculations follow the same steps as in the fully decoupled case.

In[98]:=
```
Clear[H, J, K, L, M, Q];
v = Simplify[TensorContract[(𝒟⊗𝒟)⊗𝒟, {{1, 4}, {2, 5}, {3, 7}, {6, 8}}]];
w = Simplify[𝒟.u.u];
Q = Simplify[TensorContract[𝒟⊗𝒟, {{1, 4}, {2, 5}}]];
γuu = w;
Trivialize = Table[Coeffs⟦j⟧ → 0, {j, 1, Length[Coeffs]}];
```

In what follows, we use the decomposition $\mathcal{I} = \mathcal{I}^{+} \cup \mathcal{I}'$, corresponding to the ‘Fundamental’ and ‘Secondary’ invariants. This is not standard terminology!!

In[104]:=
```
FundamentalRelations = {H[2] - Simplify[Tr[Q]],
   H[4] - Simplify[Tr[Q.Q]], J[2] - Simplify[u.u], L[4] - Simplify[γuu.u]};
FundamentalInvariants = FundamentalRelations /. Trivialize;
SecondaryRelations = { H[6] - Simplify[v.v], H[10] - Simplify[𝒟.v.v.v] ,
   J[4] - Simplify[u.Q.u], K[4] - Simplify[Tr[Q.(𝒟.u)]],
   J[6] - Simplify[(u.Q).γuu], K[6] - Simplify[v.w], L[6] - Simplify[(u.Q).v],
   M[6] - Simplify[γuu.γuu], H[8] - Simplify[(u.Q).(Q.v)]};
SecondaryInvariants = SecondaryRelations /. Trivialize;
PartialDecoupleRelations = Join[FundamentalRelations, SecondaryRelations];
```

In[109]:=
```
G = Table[Γ⟦i, j, k⟧, {i, 1, 2}, {j, 1, 2}, {k, 1, 2}]
```

Out[109]=
$\{\{\{3\,\alpha - \gamma_1,\ \gamma_2\},\ \{\gamma_2,\ \alpha + \gamma_1\}\},\ \{\{\gamma_2,\ \alpha + \gamma_1\},\ \{\alpha + \gamma_1,\ -\gamma_2\}\}\}$

This is the 2 × 2 × 2 block of R

In[110]:=
```
GStarSqrd = Expand[TensorContract[G⊗G, {{1, 4}, {2, 5}}]]
```

Out[110]=
$\{\{10\,\alpha^2 - 4\,\alpha\,\gamma_1 + 2\,\gamma_1^2 + 2\,\gamma_2^2,\ 4\,\alpha\,\gamma_2\},\ \{4\,\alpha\,\gamma_2,\ 2\,\alpha^2 + 4\,\alpha\,\gamma_1 + 2\,\gamma_1^2 + 2\,\gamma_2^2\}\}$

In[111]:=
```
MatrixForm[GStarSqrd]
```

Out[111]//MatrixForm=
$$\begin{pmatrix} 10\,\alpha^2 - 4\,\alpha\,\gamma_1 + 2\,\gamma_1^2 + 2\,\gamma_2^2 & 4\,\alpha\,\gamma_2 \\ 4\,\alpha\,\gamma_2 & 2\,\alpha^2 + 4\,\alpha\,\gamma_1 + 2\,\gamma_1^2 + 2\,\gamma_2^2 \end{pmatrix}$$

In[112]:=
```
ured = TensorContract[G, {1, 2}]
```

Out[112]=
$\{4\,\alpha,\ 0\}$

This is the trace of the 2 × 2 × 2 block

In[113]:=

```
Clear[q];
defns = {q₁ - β₃², q₂ - ured.ured, q₄ - Tr[GStarSqrd], q₃ - Det[GStarSqrd]}
```

Out[114]=

$$\{q_1 - \beta_3^2,\ -16\,\alpha^2 + q_2,\ -12\,\alpha^2 + q_4 - 4\,\gamma_1^2 - 4\,\gamma_2^2,\ -20\,\alpha^4 + q_3 - 32\,\alpha^3\,\gamma_1 - 8\,\alpha^2\,\gamma_1^2 - 4\,\gamma_1^4 - 8\,\alpha^2\,\gamma_2^2 - 8\,\gamma_1^2\,\gamma_2^2 - 4\,\gamma_2^4\}$$

These are the expressions of the O(2) x $\mathbb{Z}_2$ invariants (the quantities q_i) in terms of the parameters defining the Canonical form.

In[115]:=

```
EliminateCoeffs = Table[
  GroebnerBasis[Join[{FundamentalRelations[[i]]}, defns], Join[FundamentalInvariants,
    {q₃, q₄, q₂, q₁}], Coeffs, MonomialOrder → EliminationOrder][[1]], {i, 1, 4}]
```

Out[115]=

$$\{H[2] - 10\,q_1 + 15\,q_2 - 25\,q_4,\ -H[4] + 44\,q_1^2 - 42\,q_1\,q_2 + 144\,q_2^2 - 30\,q_3 + 100\,q_1\,q_4 - 420\,q_2\,q_4 + 320\,q_4^2,\ J[2] - q_1 - q_2,\ -2\,L[4] + 4\,q_1^2 - 12\,q_1\,q_2 + 4\,q_2^2 - 20\,q_3 - 5\,q_2\,q_4 + 5\,q_4^2\}$$

This is Eq. (7.8) expressing the elements in $\mathcal{I}^+$ in terms of the O(2) x $\mathbb{Z}_2$ invariants q_i. We can invert these relations and solve for the quantities q_i.

In[116]:=

```
TriangularSystem =
 GroebnerBasis[EliminateCoeffs, {q₄, q₃, q₂, q₁, H[2], H[4], J[2], L[4]}]
```

Out[116]=

$$\{-H[2]^2 + 2\,H[4] + 3\,H[2] \times J[2] - 6\,J[2]^2 - 6\,L[4] + 9\,H[2]\,q_1 - 90\,J[2]\,q_1,\ -J[2] + q_1 + q_2,\ 8\,H[2]^2 - 25\,H[4] - 60\,H[2] \times J[2] - 1500\,J[2]^2 + 1200\,L[4] + 11\,250\,J[2]\,q_1 - 11\,250\,q_1^2 + 11\,250\,q_3,\ -H[2] - 15\,J[2] + 25\,q_1 + 25\,q_4\}$$

In[117]:=

```
Substitutions = Table[Solve[TriangularSystem[[i]] == 0, qᵢ][[1, 1]], {i, 1, 4}]
```

Out[117]=

$$\left\{q_1 \to \frac{H[2]^2 - 2\,H[4] - 3\,H[2] \times J[2] + 6\,J[2]^2 + 6\,L[4]}{9\,(H[2] - 10\,J[2])},\ q_2 \to J[2] - q_1,\ q_3 \to \frac{-8\,H[2]^2 + 25\,H[4] + 60\,H[2] \times J[2] + 1500\,J[2]^2 - 1200\,L[4] - 11\,250\,J[2]\,q_1 + 11\,250\,q_1^2}{11\,250},\ q_4 \to \frac{1}{25}\,(H[2] + 15\,J[2] - 25\,q_1)\right\}$$

These are the substitutions implied by Eq. (7.9).

Theorem 7.2: Expressing the invariants in $\mathcal{I}'$ in terms of $\mathcal{I}^+$

In[118]:=

```
Eliminated = {α, γ₁, γ₂};
```

We are making a choice to include β_3 in defining the necessary/sufficient conditions and only eliminating α, γ_1 and γ_2. The rationale is that among all the relations between the coefficients of the canonical form and the invariants, the only relation which is not a polynomial is the relation for $\beta_3{}^2$ in terms of the OA invariants, so we can get more compact expressions without denominators if we also include it in the set of `basic' quantities for expressing the rest of the invariants.

In[119]:=

```
GroebnerBasis[FundamentalRelations, FundamentalInvariants, Eliminated]
```

Out[119]=

$$\left\{H[2]^2-2\,H[4]-3\,H[2]\times J[2]+6\,J[2]^2+6\,L[4]-9\,H[2]\,\beta_3^2+90\,J[2]\,\beta_3^2\right\}$$

This is the relation for $\beta_3{}^2$ in terms of the OA invariants. Inverting gives a rational function for $\beta_3{}^2$ which is uniquely defined only if H[2] ≠ 10 J[2]. We now calculate the polynomial expressions for the Secondary invariants in terms of the Fundamental invariants *and* $\beta_3{}^2$ with an ordering that promotes low order polynomials in β_3.

In[120]:=

```
NeccSuffRelations =
 Join[GroebnerBasis[FundamentalRelations, FundamentalInvariants, Eliminated],
  Table[GroebnerBasis[Join[{SecondaryRelations〚i〛}, FundamentalRelations], Join[
      {SecondaryInvariants〚i〛}, FundamentalInvariants], Eliminated]〚2〛, {i, 1, 9}]]
```

Out[120]=

$$\begin{aligned}\{&H[2]^2-2\,H[4]-3\,H[2]\times J[2]+6\,J[2]^2+6\,L[4]-9\,H[2]\,\beta_3^2+90\,J[2]\,\beta_3^2,\\
&H[6]-2\,H[4]\times J[2]+8\,H[2]\,J[2]^2-24\,J[2]^3-4\,H[2]\times L[4]+24\,J[2]\times L[4]-\\
&\quad 15\,H[2]\times J[2]\,\beta_3^2+90\,J[2]^2\,\beta_3^2+30\,L[4]\,\beta_3^2+39\,H[2]\,\beta_3^4-90\,J[2]\,\beta_3^4-300\,\beta_3^6,\\
&H[10]-3\,H[2]\times H[4]\,J[2]^2+40\,H[4]\,J[2]^3-63\,H[2]\,J[2]^4+114\,J[2]^5-\\
&\quad 2\,H[2]\times H[4]\times L[4]+6\,H[4]\times J[2]\times L[4]+42\,H[2]\,J[2]^2\,L[4]-246\,J[2]^3\,L[4]-\\
&\quad 4\,H[2]\,L[4]^2+24\,J[2]\,L[4]^2-6\,H[2]\times H[4]\times J[2]\,\beta_3^2-36\,H[4]\,J[2]^2\,\beta_3^2+\\
&\quad 485\,H[2]\,J[2]^3\,\beta_3^2-2442\,J[2]^4\,\beta_3^2+30\,H[4]\times L[4]\,\beta_3^2-72\,H[2]\times J[2]\times L[4]\,\beta_3^2+\\
&\quad 348\,J[2]^2\,L[4]\,\beta_3^2-20\,L[4]^2\,\beta_3^2+17\,H[2]\times H[4]\,\beta_3^4+36\,H[4]\times J[2]\,\beta_3^4-\\
&\quad 363\,H[2]\,J[2]^2\,\beta_3^4+2002\,J[2]^3\,\beta_3^4+70\,H[2]\times L[4]\,\beta_3^4-618\,J[2]\times L[4]\,\beta_3^4-280\,H[4]\,\beta_3^6+\\
&\quad 807\,H[2]\times J[2]\,\beta_3^6-3630\,J[2]^2\,\beta_3^6+140\,L[4]\,\beta_3^6-210\,H[2]\,\beta_3^8+2100\,J[2]\,\beta_3^8,\\
&-H[2]\times J[2]+2\,J[2]^2+2\,J[4]-2\,L[4]+H[2]\,\beta_3^2-10\,J[2]\,\beta_3^2,\\
&-H[2]\times J[2]+6\,J[2]^2+K[4]-2\,L[4]+2\,H[2]\,\beta_3^2-20\,J[2]\,\beta_3^2,\\
&-H[2]\,J[2]^2+6\,J[2]^3+4\,J[6]-2\,H[2]\times L[4]-2\,J[2]\times L[4]-\\
&\quad 4\,H[2]\times J[2]\,\beta_3^2-20\,J[2]^2\,\beta_3^2+30\,L[4]\,\beta_3^2+9\,H[2]\,\beta_3^4+210\,J[2]\,\beta_3^4-300\,\beta_3^6,\\
&-H[2]\,J[2]^2+6\,J[2]^3+2\,K[6]-2\,H[2]\times L[4]+6\,J[2]\times L[4]-10\,H[2]\times J[2]\,\beta_3^2+\\
&\quad 40\,J[2]^2\,\beta_3^2+30\,L[4]\,\beta_3^2+19\,H[2]\,\beta_3^4+110\,J[2]\,\beta_3^4-300\,\beta_3^6,\\
&-H[4]\times J[2]+2\,H[2]\,J[2]^2-2\,H[2]\times L[4]+8\,J[2]\times L[4]+L[6]+H[4]\,\beta_3^2-\\
&\quad 11\,H[2]\times J[2]\,\beta_3^2+42\,J[2]^2\,\beta_3^2+12\,L[4]\,\beta_3^2+19\,H[2]\,\beta_3^4-40\,J[2]\,\beta_3^4-150\,\beta_3^6,\\
&-H[2]\,J[2]^2+6\,J[2]^3-6\,J[2]\times L[4]+4\,M[6]+2\,H[2]\times J[2]\,\beta_3^2-\\
&\quad 80\,J[2]^2\,\beta_3^2+30\,L[4]\,\beta_3^2-H[2]\,\beta_3^4+310\,J[2]\,\beta_3^4-300\,\beta_3^6,\\
&2\,H[8]-H[2]\times H[4]\times J[2]+4\,H[4]\,J[2]^2+9\,H[2]\,J[2]^3-30\,J[2]^4-6\,H[4]\times L[4]+\\
&\quad 4\,H[2]\times J[2]\times L[4]+14\,J[2]^2\,L[4]+12\,L[4]^2+H[2]\times H[4]\,\beta_3^2-32\,H[4]\times J[2]\,\beta_3^2+\\
&\quad 36\,H[2]\,J[2]^2\,\beta_3^2-44\,J[2]^3\,\beta_3^2+10\,H[2]\times L[4]\,\beta_3^2+226\,J[2]\times L[4]\,\beta_3^2+40\,H[4]\,\beta_3^4+\\
&\quad 69\,H[2]\times J[2]\,\beta_3^4+390\,J[2]^2\,\beta_3^4-120\,L[4]\,\beta_3^4-70\,H[2]\,\beta_3^6-1900\,J[2]\,\beta_3^6\}\end{aligned}$$

Since we are adding an extra quantity, we have an extra equation, and we can ‘pretend’ that H[4] is also an secondary invariant.

In[121]:=

```
SecondaryInvariants = Join[{H[4]}, SecondaryInvariants]
```

Out[121]=

```
{H[4], H[6], H[10], J[4], K[4], J[6], K[6], L[6], M[6], H[8]}
```

We identify the denominators so we can make sure to get relations with integer coefficients.

In[122]:=

```
Normalizations = Table[D[NeccSuffRelations〚i〛, SecondaryInvariants〚i〛], {i, 1, 10}]
```

Out[122]=

```
{-2, 1, 1, 2, 1, 4, 2, 1, 4, 2}
```

Replacing $\beta_3{}^2$ by $q_{1.}$

In[123]:=

```
SolveNeccSuff =
 Table[Solve[NeccSuffRelations[[i]] == 0, SecondaryInvariants[[i]]][[1, 1]], {i, 1, 10}] /.
  {β3^k_ → q1^(k/2)}
```

Out[123]=

$$\left\{H[4] \to \frac{1}{2} \left(H[2]^2 - 3 H[2] \times J[2] + 6 J[2]^2 + 6 L[4] - 9 H[2] q_1 + 90 J[2] q_1\right),\right.$$

$$H[6] \to 2 H[4] \times J[2] - 8 H[2] J[2]^2 + 24 J[2]^3 + 4 H[2] \times L[4] - 24 J[2] \times L[4] + 15 H[2] \times J[2] q_1 - 90 J[2]^2 q_1 - 30 L[4] q_1 - 39 H[2] q_1^2 + 90 J[2] q_1^2 + 300 q_1^3,$$

$$H[10] \to 3 H[2] \times H[4] J[2]^2 - 40 H[4] J[2]^3 + 63 H[2] J[2]^4 - 114 J[2]^5 + 2 H[2] \times H[4] \times L[4] - 6 H[4] \times J[2] \times L[4] - 42 H[2] J[2]^2 L[4] + 246 J[2]^3 L[4] + 4 H[2] L[4]^2 - 24 J[2] L[4]^2 + 6 H[2] \times H[4] \times J[2] q_1 + 36 H[4] J[2]^2 q_1 - 485 H[2] J[2]^3 q_1 + 2442 J[2]^4 q_1 - 30 H[4] \times L[4] q_1 + 72 H[2] \times J[2] \times L[4] q_1 - 348 J[2]^2 L[4] q_1 + 20 L[4]^2 q_1 - 17 H[2] \times H[4] q_1^2 - 36 H[4] \times J[2] q_1^2 + 363 H[2] J[2]^2 q_1^2 - 2002 J[2]^3 q_1^2 - 70 H[2] \times L[4] q_1^2 + 618 J[2] \times L[4] q_1^2 + 280 H[4] q_1^3 - 807 H[2] \times J[2] q_1^3 + 3630 J[2]^2 q_1^3 - 140 L[4] q_1^3 + 210 H[2] q_1^4 - 2100 J[2] q_1^4,$$

$$J[4] \to \frac{1}{2} \left(H[2] \times J[2] - 2 J[2]^2 + 2 L[4] - H[2] q_1 + 10 J[2] q_1\right),$$

$$K[4] \to H[2] \times J[2] - 6 J[2]^2 + 2 L[4] - 2 H[2] q_1 + 20 J[2] q_1,$$

$$J[6] \to \frac{1}{4} \left(H[2] J[2]^2 - 6 J[2]^3 + 2 H[2] \times L[4] + 2 J[2] \times L[4] + 4 H[2] \times J[2] q_1 + 20 J[2]^2 q_1 - 30 L[4] q_1 - 9 H[2] q_1^2 - 210 J[2] q_1^2 + 300 q_1^3\right),$$

$$K[6] \to \frac{1}{2} \left(H[2] J[2]^2 - 6 J[2]^3 + 2 H[2] \times L[4] - 6 J[2] \times L[4] + 10 H[2] \times J[2] q_1 - 40 J[2]^2 q_1 - 30 L[4] q_1 - 19 H[2] q_1^2 - 110 J[2] q_1^2 + 300 q_1^3\right),$$

$$L[6] \to H[4] \times J[2] - 2 H[2] J[2]^2 + 2 H[2] \times L[4] - 8 J[2] \times L[4] - H[4] q_1 + 11 H[2] \times J[2] q_1 - 42 J[2]^2 q_1 - 12 L[4] q_1 - 19 H[2] q_1^2 + 40 J[2] q_1^2 + 150 q_1^3,$$

$$M[6] \to \frac{1}{4} \left(H[2] J[2]^2 - 6 J[2]^3 + 6 J[2] \times L[4] - 2 H[2] \times J[2] q_1 + 80 J[2]^2 q_1 - 30 L[4] q_1 + H[2] q_1^2 - 310 J[2] q_1^2 + 300 q_1^3\right),$$

$$\left. H[8] \to \frac{1}{2} \left(H[2] \times H[4] \times J[2] - 4 H[4] J[2]^2 - 9 H[2] J[2]^3 + 30 J[2]^4 + 6 H[4] \times L[4] - 4 H[2] \times J[2] \times L[4] - 14 J[2]^2 L[4] - 12 L[4]^2 - H[2] \times H[4] q_1 + 32 H[4] \times J[2] q_1 - 36 H[2] J[2]^2 q_1 + 44 J[2]^3 q_1 - 10 H[2] \times L[4] q_1 - 226 J[2] \times L[4] q_1 - 40 H[4] q_1^2 - 69 H[2] \times J[2] q_1^2 - 390 J[2]^2 q_1^2 + 120 L[4] q_1^2 + 70 H[2] q_1^3 + 1900 J[2] q_1^3\right)\right\}$$

Multiplying out the denominators to get relations with integer coefficients.

In[124]:=

```
FormatAsEquations = Table[Normalizations⟦i⟧ × SecondaryInvariants⟦i⟧ → Collect[
    (Normalizations⟦i⟧ × SecondaryInvariants⟦i⟧ /. SolveNeccSuff), q1], {i, 1, 10}]
```

Out[124]=

$$\{-2\,H[4] \to -H[2]^2 + 3\,H[2]\times J[2] - 6\,J[2]^2 - 6\,L[4] + (9\,H[2] - 90\,J[2])\,q_1,$$
$$H[6] \to 2\,H[4]\times J[2] - 8\,H[2]\,J[2]^2 + 24\,J[2]^3 + 4\,H[2]\times L[4] - 24\,J[2]\times L[4] + \left(15\,H[2]\times J[2] - 90\,J[2]^2 - 30\,L[4]\right)q_1 + (-39\,H[2] + 90\,J[2])\,q_1^2 + 300\,q_1^3,$$
$$H[10] \to 3\,H[2]\times H[4]\,J[2]^2 - 40\,H[4]\,J[2]^3 + 63\,H[2]\,J[2]^4 - 114\,J[2]^5 + 2\,H[2]\times H[4]\times L[4] - 6\,H[4]\times J[2]\times L[4] - 42\,H[2]\,J[2]^2\,L[4] + 246\,J[2]^3\,L[4] + 4\,H[2]\,L[4]^2 - 24\,J[2]\,L[4]^2 + \left(6\,H[2]\times H[4]\times J[2] + 36\,H[4]\,J[2]^2 - 485\,H[2]\,J[2]^3 + 2442\,J[2]^4 - 30\,H[4]\times L[4] + 72\,H[2]\times J[2]\times L[4] - 348\,J[2]^2\,L[4] + 20\,L[4]^2\right)q_1 + \left(-17\,H[2]\times H[4] - 36\,H[4]\times J[2] + 363\,H[2]\,J[2]^2 - 2002\,J[2]^3 - 70\,H[2]\times L[4] + 618\,J[2]\times L[4]\right)q_1^2 + \left(280\,H[4] - 807\,H[2]\times J[2] + 3630\,J[2]^2 - 140\,L[4]\right)q_1^3 + (210\,H[2] - 2100\,J[2])\,q_1^4,$$
$$2\,J[4] \to H[2]\times J[2] - 2\,J[2]^2 + 2\,L[4] + (-H[2] + 10\,J[2])\,q_1,$$
$$K[4] \to H[2]\times J[2] - 6\,J[2]^2 + 2\,L[4] + (-2\,H[2] + 20\,J[2])\,q_1,$$
$$4\,J[6] \to H[2]\,J[2]^2 - 6\,J[2]^3 + 2\,H[2]\times L[4] + 2\,J[2]\times L[4] + \left(4\,H[2]\times J[2] + 20\,J[2]^2 - 30\,L[4]\right)q_1 + (-9\,H[2] - 210\,J[2])\,q_1^2 + 300\,q_1^3,$$
$$2\,K[6] \to H[2]\,J[2]^2 - 6\,J[2]^3 + 2\,H[2]\times L[4] - 6\,J[2]\times L[4] + \left(10\,H[2]\times J[2] - 40\,J[2]^2 - 30\,L[4]\right)q_1 + (-19\,H[2] - 110\,J[2])\,q_1^2 + 300\,q_1^3,$$
$$L[6] \to H[4]\times J[2] - 2\,H[2]\,J[2]^2 + 2\,H[2]\times L[4] - 8\,J[2]\times L[4] + \left(-H[4] + 11\,H[2]\times J[2] - 42\,J[2]^2 - 12\,L[4]\right)q_1 + (-19\,H[2] + 40\,J[2])\,q_1^2 + 150\,q_1^3,$$
$$4\,M[6] \to H[2]\,J[2]^2 - 6\,J[2]^3 + 6\,J[2]\times L[4] + \left(-2\,H[2]\times J[2] + 80\,J[2]^2 - 30\,L[4]\right)q_1 + (H[2] - 310\,J[2])\,q_1^2 + 300\,q_1^3,$$
$$2\,H[8] \to H[2]\times H[4]\times J[2] - 4\,H[4]\,J[2]^2 - 9\,H[2]\,J[2]^3 + 30\,J[2]^4 + 6\,H[4]\times L[4] - 4\,H[2]\times J[2]\times L[4] - 14\,J[2]^2\,L[4] - 12\,L[4]^2 + \left(-H[2]\times H[4] + 32\,H[4]\times J[2] - 36\,H[2]\,J[2]^2 + 44\,J[2]^3 - 10\,H[2]\times L[4] - 226\,J[2]\times L[4]\right)q_1 + \left(-40\,H[4] - 69\,H[2]\times J[2] - 390\,J[2]^2 + 120\,L[4]\right)q_1^2 + (70\,H[2] + 1900\,J[2])\,q_1^3\}$$

Calculations with respect to Lemma 7.3.

In[125]:=

```
GroebnerBasis[defns, Join[{β3}, Table[qi, {i, 1, 4}]], {α, γ1, γ2}]
```

Out[125]=

$$\{-q_1 + \beta_3^2\}$$

We need $q_1 \geq 0$ to solve for a real β_3.

In[126]:=

```
GroebnerBasis[defns, Join[{γ2}, Table[qi, {i, 1, 4}]], {α, γ1, β3}]
```

Out[126]=

$$\{q_2^4 - 4\,q_2^2\,q_3 + 16\,q_3^2 - 2\,q_2^3\,q_4 + 8\,q_2\,q_3\,q_4 + 2\,q_2^2\,q_4^2 - 8\,q_3\,q_4^2 - 2\,q_2\,q_4^3 + q_4^4 + 4\,q_2^3\,\gamma_2^2\}$$

In[127]:=

```
Collect[%⟦1⟧, γ2]
```

Out[127]=

$$q_2^4 - 4\,q_2^2\,q_3 + 16\,q_3^2 - 2\,q_2^3\,q_4 + 8\,q_2\,q_3\,q_4 + 2\,q_2^2\,q_4^2 - 8\,q_3\,q_4^2 - 2\,q_2\,q_4^3 + q_4^4 + 4\,q_2^3\,\gamma_2^2$$

We get a linear equation for $\gamma_2{}^2$.

In[128]:=

```
Eqnγ2 = (% /. {γ2^2 → γ2sqrd}) == 0
```

Out[128]=

$$4\,\gamma2sqrd\,q_2^3 + q_2^4 - 4\,q_2^2\,q_3 + 16\,q_3^2 - 2\,q_2^3\,q_4 + 8\,q_2\,q_3\,q_4 + 2\,q_2^2\,q_4^2 - 8\,q_3\,q_4^2 - 2\,q_2\,q_4^3 + q_4^4 == 0$$

In[129]:=

```
Solve[Eqnγ2, γ2sqrd][[1]]
```

Out[129]=

$$\left\{\gamma2sqrd \to \frac{-q_2^4 + 4\,q_2^2\,q_3 - 16\,q_3^2 + 2\,q_2^3\,q_4 - 8\,q_2\,q_3\,q_4 - 2\,q_2^2\,q_4^2 + 8\,q_3\,q_4^2 + 2\,q_2\,q_4^3 - q_4^4}{4\,q_2^3}\right\}$$

To get a real solution, we therefore need $q_2 \neq 0$ and the fraction (or equivalently the product of the numerator and the denominator in the above expression) is non-negative. We see below that $q_2 > 0$ is necessary, and along with this condition, we will need that the numerator be greater than or equal to zero.

In[130]:=

```
Eqnsaαγ = GroebnerBasis[defns, Join[{α, γ1}, Table[qi, {i, 1, 4}]], {γ2, β3}]
```

Out[130]=

$$\{-q_2^4 + 16\,q_2^2\,q_3 - 64\,q_3^2 + 4\,q_2^3\,q_4 - 32\,q_2\,q_3\,q_4 - 8\,q_2^2\,q_4^2 + 32\,q_3\,q_4^2 + 8\,q_2\,q_4^3 - 4\,q_4^4 + 16\,q_2^3\,\gamma_1^2,$$
$$\alpha\,q_2^2 - 8\,\alpha\,q_3 - 2\,\alpha\,q_2\,q_4 + 2\,\alpha\,q_4^2 + q_2^2\,\gamma_1,$$
$$q_2^3 - 8\,q_2\,q_3 - 4\,q_2^2\,q_4 + 16\,q_3\,q_4 + 6\,q_2\,q_4^2 - 4\,q_4^3 + 128\,\alpha\,q_3\,\gamma_1 - 32\,\alpha\,q_4^2\,\gamma_1 - 16\,q_2^2\,\gamma_1^2,$$
$$q_2^2 - 8\,q_3 - 2\,q_2\,q_4 + 2\,q_4^2 + 16\,\alpha\,q_2\,\gamma_1,\ 16\,\alpha^2 - q_2\}$$

In[131]:=

```
Eqnsaαγ[[5]] == 0
```

Out[131]=

$$16\,\alpha^2 - q_2 == 0$$

To find a real solution, we need $q_2 \geq 0$. This, along with the earlier requirement $q_2 \neq 0$ implies that $q_2 > 0$.

In[132]:=

```
Eqnsaαγ[[2]] == 0
```

Out[132]=

$$\alpha\,q_2^2 - 8\,\alpha\,q_3 - 2\,\alpha\,q_2\,q_4 + 2\,\alpha\,q_4^2 + q_2^2\,\gamma_1 == 0$$

In[133]:=

```
Simplify[Solve[Eqnsaαγ[[2]] == 0, γ1][[1]]]
```

Out[133]=

$$\left\{\gamma_1 \to -\frac{\alpha\,\left(q_2^2 - 8\,q_3 - 2\,q_2\,q_4 + 2\,q_4^2\right)}{q_2^2}\right\}$$

We get no further conditions from the solvability for γ_1

Example 7.4

In[134]:=

```
n = 3;
vars = Table[x[i], {i, 1, n}];
f = Sum[2 i x[i]^3, {i, 1, n}] + (3 x[1]^2×x[2] - x[2]^3) - 12 x[1]×x[2]×x[3]
```

Out[134]=

$$2\,x[1]^3 + 3\,x[1]^2\,x[2] + 3\,x[2]^3 - 12\,x[1]\times x[2]\times x[3] + 6\,x[3]^3$$

This is an explicit numerical example.

In[135]:=

```
Γ = Simplify[Table[D[f, x[i], x[j], x[k]], {k, 1, n}, {j, 1, n}, {i, 1, n}] / 6];
u = Simplify[Table[D[Laplacian[f, vars], x[i]], {i, 1, n}] / 6];
f₃ = (n + 2) f - 3 (u.vars) (vars.vars);
𝒟 = Simplify[Table[D[f₃, x[i], x[j], x[k]], {k, 1, n}, {j, 1, n}, {i, 1, n}] / 6];
Clear[H, J, K, L, M, Q];
v = Simplify[TensorContract[(𝒟⊗𝒟)⊗𝒟, {{1, 4}, {2, 5}, {3, 7}, {6, 8}}]];
w = Simplify[𝒟.u.u];
Q = Simplify[TensorContract[𝒟⊗𝒟, {{1, 4}, {2, 5}}]];
γuu = w;
FundamentalValues = {H[2] → Simplify[Tr[Q]],
   H[4] → Simplify[Tr[Q.Q]], J[2] → Simplify[u.u], L[4] → Simplify[γuu.u]};
SecondaryValues =
  { H[6] → Simplify[v.v], H[10] - Simplify[𝒟.v.v.v] , J[4] - Simplify[u.Q.u],
   K[4] - Simplify[Tr[Q.(𝒟.u)]], J[6] - Simplify[(u.Q).γuu], K[6] - Simplify[v.w],
   L[6] - Simplify[(u.Q).v], M[6] - Simplify[γuu.γuu], H[8] - Simplify[(u.Q).(Q.v)]};
```

In[146]:=

```
FundamentalValues
```

Out[146]=

$$\{H[2]\to 1060,\ H[4]\to 518\,384,\ J[2]\to 56,\ L[4]\to -4528\}$$

In[147]:=

```
Specialization = FundamentalRelations /. FundamentalValues
```

Out[147]=

$$\Big\{1060 - 10\left(6\,\alpha^2 + \beta_3^2 + 10\,\gamma_1^2 + 10\,\gamma_2^2\right),\ 518\,384 - 32\,\alpha^2\,\beta_3^2 - \left(32\,\alpha^2 + 6\,\beta_3^2\right)^2 - 800\,\alpha^2\,\gamma_2^2 - 4\left(13\,\alpha^2 + \beta_3^2 - 10\,\alpha\,\gamma_1 + 25\,\gamma_1^2 + 25\,\gamma_2^2\right)^2 - 4\left(\beta_3^2 + (\alpha + 5\,\gamma_1)^2 + 25\,\gamma_2^2\right)^2,\ 56 - 16\,\alpha^2 - \beta_3^2,\ -4528 - 2\left(-48\,\alpha^2\,\beta_3^2 + \beta_3^4 + 32\,\alpha^3\,(3\,\alpha - 5\,\gamma_1)\right)\Big\}$$

In[148]:=

```
GroebnerBasis[Specialization, Coeffs]
```

Out[148]=

$$\{332 + 15\,\beta_3^2,\ 3\,173\,103\,609 + 125\,768\,785\,\gamma_2^2,\ -52\,993\,421\,209 + 1\,509\,225\,420\,\gamma_1^2,\ 230\,203\,\alpha - 85\,849\,\gamma_1\}$$