Section 4

We compute the linear representation of SO(2) on $\mathcal{T}_2$

```
In[1]:= Clear[Γ];
n = 2;
vars = Table[x[i], {i, 1, n}];
Γlist = Flatten[Table[Γ[i, j, k], {i, 1, n}, {j, i, n}, {k, j, n}]];
rename = Table[Γlist⟦m⟧ → Subscript[a, 4 - m], {m, 1, 4}];
f =
 (Sum[Γ @@ Sort[{i, j, k}] × x[i] × x[j] × x[k], {i, 1, n}, {j, 1, n}, {k, 1, n}] /. rename)
```

Out[6]= $a_3\, x[1]^3 + 3\, a_2\, x[1]^2\, x[2] + 3\, a_1\, x[1]\, x[2]^2 + a_0\, x[2]^3$

The last line is the expression for the cubic polynomial associate to a tensor Γ. Note that the coordinates are x[1] and x[2] where the indices are arguments and not subscripts.

```
In[7]:= Clear[σ];
σ[θ_] = {{Cos[θ], - Sin[θ]}, {Sin[θ], Cos[θ]}};
MatrixForm[σ[θ]]
```

Out[9]//MatrixForm=

$$\begin{pmatrix} \mathrm{Cos}[\theta] & -\mathrm{Sin}[\theta] \\ \mathrm{Sin}[\theta] & \mathrm{Cos}[\theta] \end{pmatrix}$$

Multiplying by this rotation matrix on the left gives the action of SO(2) on $\mathbb{R}^2$ where the elements are thought of a column vectors. The action corresponds to rotating `counter-clockwise' by an angle θ.

The action on Tensors (or equivalently on polynomials) is given by $\sigma \circ \mathrm{f(z)} = f\left(\sigma^{-1}\cdot \mathrm{z}\right)$ for $\mathrm{z} \in \mathbb{R}^2$.

```
In[10]:= Substitution = Table[x[i] → (σ[-θ].{z[1], z[2]})⟦i⟧, {i, 1, n}]
```

Out[10]=

```
{x[1] → Cos[θ] z[1] + Sin[θ] z[2], x[2] → -Sin[θ] z[1] + Cos[θ] z[2]}
```

```
In[11]:= Transformedf = f //. Substitution
```

Out[11]=

$$a_0\,(-\mathrm{Sin}[\theta]\, z[1] + \mathrm{Cos}[\theta]\, z[2])^3 + 3\, a_1\,(-\mathrm{Sin}[\theta]\, z[1] + \mathrm{Cos}[\theta]\, z[2])^2\,(\mathrm{Cos}[\theta]\, z[1] + \mathrm{Sin}[\theta]\, z[2]) + 3\, a_2\,(-\mathrm{Sin}[\theta]\, z[1] + \mathrm{Cos}[\theta]\, z[2])\,(\mathrm{Cos}[\theta]\, z[1] + \mathrm{Sin}[\theta]\, z[2])^2 + a_3\,(\mathrm{Cos}[\theta]\, z[1] + \mathrm{Sin}[\theta]\, z[2])^3$$

This is a cubic polynomial in z[1],z[2]. We can now read off the transformations of the coefficients from $\sigma\circ \mathrm{f(z)} = b_3\, z[1]^3 + 3\, b_2\, z[1]^2\, z[2] + 3\, b_1\, z[1]\, z[2]^2 + b_0\, z[2]^3$. We account for the factors of 3 in the coefficients b[1] and b[2] and order the coefficients as a column vector from b_0 to b_3.

In[12]:=
```
newcoeffs = Simplify[DiagonalMatrix[{1, 1/3, 1/3, 1}].
   CoefficientList[Transformedf /. {z[2] → 1}, z[1]]]
```

Out[12]=

$$\Big\{\mathrm{Cos}[\theta]^3\, a_0 + \mathrm{Sin}[\theta]\,\left(3\,\mathrm{Cos}[\theta]^2\, a_1 + \mathrm{Sin}[\theta]\,(3\,\mathrm{Cos}[\theta]\, a_2 + \mathrm{Sin}[\theta]\, a_3)\right),$$
$$\frac{1}{4}\left(-4\,\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta]\, a_0 + (\mathrm{Cos}[\theta] + 3\,\mathrm{Cos}[3\,\theta])\, a_1 + 2\,\mathrm{Sin}[\theta]\,(a_2 + 3\,\mathrm{Cos}[2\,\theta]\, a_2 + \mathrm{Sin}[2\,\theta]\, a_3)\right),\ \mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]^2\, a_0 +$$
$$\frac{1}{4}\,((\mathrm{Sin}[\theta] - 3\,\mathrm{Sin}[3\,\theta])\, a_1 + 2\,\mathrm{Cos}[\theta]\,((-1 + 3\,\mathrm{Cos}[2\,\theta])\, a_2 + \mathrm{Sin}[2\,\theta]\, a_3)),$$
$$-\mathrm{Sin}[\theta]^3\, a_0 + \mathrm{Cos}[\theta]\,\left(3\,\mathrm{Sin}[\theta]^2\, a_1 + \mathrm{Cos}[\theta]\,(-3\,\mathrm{Sin}[\theta]\, a_2 + \mathrm{Cos}[\theta]\, a_3)\right)\Big\}$$

In[13]:=
```
Lσ = Grad[newcoeffs, Table[a_{i-1}, {i, 1, 4}]]
```

Out[13]=

$$\Big\{\{\mathrm{Cos}[\theta]^3,\ 3\,\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta],\ 3\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]^2,\ \mathrm{Sin}[\theta]^3\},$$
$$\left\{-\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta],\ \frac{1}{4}\,(\mathrm{Cos}[\theta] + 3\,\mathrm{Cos}[3\,\theta]),\ \frac{1}{2}\,(1 + 3\,\mathrm{Cos}[2\,\theta])\,\mathrm{Sin}[\theta],\ \frac{1}{2}\,\mathrm{Sin}[\theta]\,\mathrm{Sin}[2\,\theta]\right\},$$
$$\left\{\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]^2,\ \frac{1}{4}\,(\mathrm{Sin}[\theta] - 3\,\mathrm{Sin}[3\,\theta]),\ \frac{1}{2}\,\mathrm{Cos}[\theta]\,(-1 + 3\,\mathrm{Cos}[2\,\theta]),\ \frac{1}{2}\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[2\,\theta]\right\},$$
$$\{-\mathrm{Sin}[\theta]^3,\ 3\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]^2,\ -3\,\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta],\ \mathrm{Cos}[\theta]^3\}\Big\}$$

In[14]:=
```
MatrixForm[Lσ]
```

Out[14]//MatrixForm=

$$\begin{pmatrix} \mathrm{Cos}[\theta]^3 & 3\,\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta] & 3\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]^2 & \mathrm{Sin}[\theta]^3 \\ -\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta] & \frac{1}{4}\,(\mathrm{Cos}[\theta] + 3\,\mathrm{Cos}[3\,\theta]) & \frac{1}{2}\,(1 + 3\,\mathrm{Cos}[2\,\theta])\,\mathrm{Sin}[\theta] & \frac{1}{2}\,\mathrm{Sin}[\theta]\,\mathrm{Sin}[2\,\theta] \\ \mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]^2 & \frac{1}{4}\,(\mathrm{Sin}[\theta] - 3\,\mathrm{Sin}[3\,\theta]) & \frac{1}{2}\,\mathrm{Cos}[\theta]\,(-1 + 3\,\mathrm{Cos}[2\,\theta]) & \frac{1}{2}\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[2\,\theta] \\ -\mathrm{Sin}[\theta]^3 & 3\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]^2 & -3\,\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta] & \mathrm{Cos}[\theta]^3 \end{pmatrix}$$

This is the representation of SO(2) on the space of 2 × 2 × 2 symmetric tensors. This representation is used in Sec. 4 (Pg. 15) of the paper. We can also determine the generator for this action.

Section 2.2

In[15]:=
```
ℒ = D[Lσ, θ] /. {θ → 0}
```

Out[15]=

```
{{0, 3, 0, 0}, {-1, 0, 2, 0}, {0, -2, 0, 1}, {0, 0, -3, 0}}
```

In[16]:=
```
MatrixForm[ℒ]
```

Out[16]//MatrixForm=

$$\begin{pmatrix} 0 & 3 & 0 & 0 \\ -1 & 0 & 2 & 0 \\ 0 & -2 & 0 & 1 \\ 0 & 0 & -3 & 0 \end{pmatrix}$$

This is the matrix L in Sec. 2.1 (Pg. 11)

In[17]:=
```
{vals, vecs} = Eigensystem[ℒ]
```
Out[17]=
```
{{3 i, -3 i, i, -i}, {{i, -1, -i, 1}, {-i, -1, i, 1}, {-3 i, 1, -i, 3}, {3 i, 1, i, 3}}}
```

In[18]:=
```
Λ = DiagonalMatrix[vals]
```
Out[18]=
```
{{3 i, 0, 0, 0}, {0, -3 i, 0, 0}, {0, 0, i, 0}, {0, 0, 0, -i}}
```

If we treat vecs as a matrix instead of a list of vectors, each eigenvector will be treated as a row. To make them columns, as appropriate for a right eigenvector, we need to take a transpose.

In[19]:=
```
ℒ.Transpose[vecs] - Transpose[vecs].Λ
```
Out[19]=
```
{{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}
```

Transpose[vecs] gives the right eigenvectors of $\mathcal{L}$. To get the Left eigenvectors as rows, we need to invert Transpose[vecs]. Below, we include an additional normalization to clear denominators.

In[20]:=
```
Lvecs = Sqrt[Det[vecs]] Inverse[Transpose[vecs]]
```
Out[20]=
```
{{1, -3 i, -3, i}, {-1, -3 i, 3, i}, {-1, i, -1, i}, {1, i, 1, i}}
```

These are the left Eigenvectors of $\mathcal{L}$. Lets Check

In[21]:=
```
Lvecs.ℒ - Λ.Lvecs
```
Out[21]=
```
{{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}
```

In[22]:=
```
MatrixForm[Lvecs]
```
Out[22]//MatrixForm=

$$\begin{pmatrix} 1 & -3i & -3 & i \\ -1 & -3i & 3 & i \\ -1 & i & -1 & i \\ 1 & i & 1 & i \end{pmatrix}$$

This is the matrix E in Sec. 2.1 of the paper and the corresponding eigenvalues are λ_1=3i, λ_2=-3i, λ_3=i, λ_4=-i.

For a diagonal tensor (β_1, 0, 0, β_2) we get v_i= β_1+ i β_2 = v_4. This gives Exp[4 i θ] = $v_4(0)$ / $v_1(0)$ = $\frac{(a_0+a_2)+i\,(a_1+a_3)}{(a_0-3\,a_2)+i\,(-3\,a_1+a_3}$. Subject to the necessary condition $(a_0 + a_2)^2 + (a_1 + a_3)^2 = (a_0 - 3\,a_2)^2 + (-3\,a_1 + a_3)^2$, we get 4 solutions for θ, say φ, φ+π/2, φ+ π and φ+3π/2.

These solutions define β_1 and β_2 by β_1+ i β_2 = Exp[-iφ] $((a_0 + a_2) + i\,(a_1 + a_3))$. The rotations φ+π/2, φ+π and φ+3π/2 then correspond, respectively, to the diagonal tensors (β_2, 0, 0, -β_1), (−β_1, 0, 0, -β_2) and (−β_2, 0, 0, β_1) respectively.

Hilbert series for the SO(2) invariants.

In[23]:=
```
Simplify[1 / (2 π) Integrate[1 / Simplify[Det[IdentityMatrix[4] - λ Lσ]], {θ, 0, 2 π}]]
```

Out[23]=

$$\begin{cases} \frac{1+\lambda^4}{(-1+\lambda^2)^3\,(1+\lambda^2)} & \text{Abs}[\lambda] > 1 \\ \frac{2+\lambda^{2/3}\,(-1-\lambda^{2/3}+\lambda^{4/3})\,(1+\lambda^{2/3}+\lambda^{4/3}+2\,\lambda^2)}{3\,(-1+\lambda^2)^3\,(1+\lambda^2)} & \frac{1}{\text{Abs}[\lambda]^{1/3}} < 1 \text{ if } \text{Abs}[\lambda]^{1/3} \neq 1 \\ -\frac{1+\lambda^4}{(-1+\lambda^2)^3\,(1+\lambda^2)} & \text{True} \end{cases}$$

We need the result for Abs[λ] < 1, so this is the last line in the piecewise defined integral.

In[24]:=

$$\Phi SO_2 = \text{Simplify}\left[-\frac{1+\lambda^4}{(-1+\lambda^2)^3\,(1+\lambda^2)}\right]$$

Out[24]=

$$-\frac{1+\lambda^4}{(-1+\lambda^2)^3\,(1+\lambda^2)}$$

Hilbert series for the O(2) invariants.

We now compute the action of O(2) by adding a reflection operator corresponding to x[1]-> x[1], x[2]->-x[2]. In terms of the tensor coefficients, this action is given by the matrix

In[25]:=
```
N = {{0, 0, 0, 1}, {0, 0, 1, 0}, {0, 1, 0, 0}, {1, 0, 0, 0}};
```

In[26]:=
```
Simplify[Det[IdentityMatrix[4] - λ N.Lσ]]
```

Out[26]=

$$(-1+\lambda^2)^2$$

This determinant does not depend explicitly on θ, so it is easy to integrate the reciprocal.

In[27]:=
```
ΦO2 = Simplify[ (1 / Det[IdentityMatrix[4] - λ N.Lσ] + ΦSO2) / 2 ]
```

Out[27]=

$$-\frac{1}{(-1+\lambda^2)^3\,(1+\lambda^2)}$$

For O(2) covariants corresponding to linear forms, the generating function is 1/2 the generating function for SO(2). Does this make sense?

If we take u and w as the fundamental vector covariants, then other covariants are obtained by taking linear combinations of u and w with coefficients given by O(2) invariants. Consequently, the generating function is

In[28]:=
```
Simplify[(λ + λ^3) ΦO2]
```

Out[28]=

$$-\frac{\lambda}{(-1+\lambda^2)^3}$$

Computation of the invariants

In[29]:=
```
u = Simplify[ Grad[Laplacian[f, vars], vars] / 6]
```
Out[29]=

$\{a_1 + a_3, a_0 + a_2\}$

This is the trace vector. We ‘lift’ this vector to form the cubic polynomial f_1

In[30]:=
```
f1 = 3 u.vars (vars.vars) / (n + 2)
```
Out[30]=

$$\frac{3}{4} ((a_1 + a_3) x[1] + (a_0 + a_2) x[2]) \left(x[1]^2 + x[2]^2\right)$$

We can form an additional SO(2) covariant vector by rotating u counterclockwise by $\pi/2$.

In[31]:=
```
uperp = σ[π / 2].u
```
Out[31]=

$\{-a_0 - a_2, a_1 + a_3\}$

f_1 corresponds to a $2 \times 2 \times 2$ symmetric tensor $\mathcal{B}$

In[32]:=
```
ℬ = Table[D[f1 / 6, x[i], x[j], x[k]], {i, 1, n}, {j, 1, n}, {k, 1, n}];
```
In[33]:=
```
Table[MatrixForm[ℬ⟦i⟧], {i, 1, n}]
```
Out[33]=

$$\left\{\begin{pmatrix} \frac{3}{4} (a_1 + a_3) & \frac{1}{4} (a_0 + a_2) \\ \frac{1}{4} (a_0 + a_2) & \frac{1}{4} (a_1 + a_3) \end{pmatrix}, \begin{pmatrix} \frac{1}{4} (a_0 + a_2) & \frac{1}{4} (a_1 + a_3) \\ \frac{1}{4} (a_1 + a_3) & \frac{3}{4} (a_0 + a_2) \end{pmatrix}\right\}$$

With our normalization, the trace-free part is given by $f_3 = (n+2) f - f_1$

In[34]:=
```
f3 = Collect[Expand[(n + 2) (f - f1)], vars]
```
Out[34]=

$$(-3 a_1 + a_3) x[1]^3 + (-3 a_0 + 9 a_2) x[1]^2 x[2] + (9 a_1 - 3 a_3) x[1] x[2]^2 + (a_0 - 3 a_2) x[2]^3$$

f_3 corresponds to a trace-free $2 \times 2 \times 2$ symmetric tensor $\mathcal{D}$. To eliminate denominators, we multiply by a factor of (n+2), which equals 4 in the case n=2.

In[35]:=
```
𝒟 = Table[Simplify[D[f3 / 6, x[i], x[j], x[k]]], {i, 1, n}, {j, 1, n}, {k, 1, n}];
```
In[36]:=
```
Table[MatrixForm[𝒟⟦i⟧], {i, 1, n}]
```
Out[36]=

$$\left\{\begin{pmatrix} -3 a_1 + a_3 & -a_0 + 3 a_2 \\ -a_0 + 3 a_2 & 3 a_1 - a_3 \end{pmatrix}, \begin{pmatrix} -a_0 + 3 a_2 & 3 a_1 - a_3 \\ 3 a_1 - a_3 & a_0 - 3 a_2 \end{pmatrix}\right\}$$

In[37]:=
```
Dstarsqrd = Simplify[TensorContract[𝒟⊗𝒟, {{1, 4}, {2, 5}}]]
```
Out[37]=

$$\left\{\left\{2 \left((a_0 - 3 a_2)^2 + (-3 a_1 + a_3)^2\right), 0\right\}, \left\{0, 2 \left((a_0 - 3 a_2)^2 + (-3 a_1 + a_3)^2\right)\right\}\right\}$$

In[38]:=
```
v = Simplify[TensorContract[(𝒟⊗𝒟)⊗𝒟, {{1, 4}, {2, 5}, {3, 7}, {6, 8}}]]
```
Out[38]=

$\{0, 0\}$

In[39]:=
```
Simplify[Dstarsqrd.𝒟]
```

Out[39]=

$$\{\{\{2\ (-3\ a_1+a_3)\ \left((a_0-3\ a_2)^2+(-3\ a_1+a_3)^2\right),\ 2\ (-a_0+3\ a_2)\ \left((a_0-3\ a_2)^2+(-3\ a_1+a_3)^2\right)\},\ \{2\ (-a_0+3\ a_2)\ \left((a_0-3\ a_2)^2+(-3\ a_1+a_3)^2\right),\ 2\ (3\ a_1-a_3)\ \left((a_0-3\ a_2)^2+(-3\ a_1+a_3)^2\right)\}\},\ \{\{2\ (-a_0+3\ a_2)\ \left((a_0-3\ a_2)^2+(-3\ a_1+a_3)^2\right),\ 2\ (3\ a_1-a_3)\ \left((a_0-3\ a_2)^2+(-3\ a_1+a_3)^2\right)\},\ \{2\ (3\ a_1-a_3)\ \left((a_0-3\ a_2)^2+(-3\ a_1+a_3)^2\right),\ 2\ (a_0-3\ a_2)\ \left((a_0-3\ a_2)^2+(-3\ a_1+a_3)^2\right)\}\}\}$$

In[40]:=
```
Simplify[Table[Tr[Simplify[Dstarsqrd.𝒟][[i]]], {i, 1, n}]]
```

Out[40]=

$$\{0,\ 0\}$$

In[41]:=
```
w = Expand[𝒟.u.u]
```

Out[41]=

$$\{a_0^2\ a_1-3\ a_1^3+10\ a_0\ a_1\ a_2+9\ a_1\ a_2^2-3\ a_0^2\ a_3-5\ a_1^2\ a_3+2\ a_0\ a_2\ a_3+5\ a_2^2\ a_3-a_1\ a_3^2+a_3^3,\ a_0^3+5\ a_0\ a_1^2-a_0^2\ a_2+9\ a_1^2\ a_2-5\ a_0\ a_2^2-3\ a_2^3+2\ a_0\ a_1\ a_3+10\ a_1\ a_2\ a_3-3\ a_0\ a_3^2+a_2\ a_3^2\}$$

In[42]:=
```
TeXForm[w]
```

Out[42]//TeXForm=

```
\left\{-3 a_1^3-5 a_3 a_1^2+a_0^2 a_1+9 a_2^2 a_1-a_3^2 a_1+10 a_0 a_2 a_1+a_3^3-3
   a_0^2 a_3+5 a_2^2 a_3+2 a_0 a_2 a_3,a_0^3-a_2 a_0^2+5 a_1^2 a_0-5 a_2^2 a_0-3 a_
   a_0+2 a_1 a_3 a_0-3 a_2^3+a_2 a_3^2+9 a_1^2 a_2+10 a_1 a_2 a_3\right\}
```

In[43]:=
```
Simplify[u.w]
```

Out[43]=

$$a_0^4-3\ a_1^4-3\ a_2^4-8\ a_1^3\ a_3+24\ a_1\ a_2^2\ a_3+6\ a_2^2\ a_3^2+a_3^4-8\ a_0\ a_2\ \left(-3\ a_1^2+a_2^2-3\ a_1\ a_3\right)+6\ a_0^2\ \left(a_1^2-a_2^2-a_3^2\right)+6\ a_1^2\ \left(3\ a_2^2-a_3^2\right)$$

In[44]:=
```
Simplify[Det[{u, w}]]
```

Out[44]=

$$4\ \left(-3\ a_0^2\ a_1\ a_2+a_0^3\ a_3+a_2\ \left(3\ a_1^3+6\ a_1^2\ a_3-2\ a_2^2\ a_3-3\ a_1\ \left(a_2^2-a_3^2\right)\right)+a_0\ \left(2\ a_1^3-6\ a_1\ a_2^2+3\ a_1^2\ a_3-a_3\ \left(3\ a_2^2+a_3^2\right)\right)\right)$$

We can now compute SO(2) invariants from u, w and $\mathcal{D}$.

In[45]:=
```
Clear[j, h, l, m]
Invariants =
 {j₂ - u.u, h₂ - Tr[TensorContract[𝒟⊗𝒟, {{1, 4}, {2, 5}}]], l₄ - w.u, m₄ - Det[{u, w}]}
```

Out[46]=

$$\{-(a_0+a_2)^2-(a_1+a_3)^2+j_2,\ -(a_0-3\ a_2)^2-3\ (-a_0+3\ a_2)^2-3\ (3\ a_1-a_3)^2-(-3\ a_1+a_3)^2+h_2,\ -\left((a_0+a_2)\ \left(a_0^3+5\ a_0\ a_1^2-a_0^2\ a_2+9\ a_1^2\ a_2-5\ a_0\ a_2^2-3\ a_2^3+2\ a_0\ a_1\ a_3+10\ a_1\ a_2\ a_3-3\ a_0\ a_3^2+a_2\ a_3^2\right)\right)-(a_1+a_3)\ \left(a_0^2\ a_1-3\ a_1^3+10\ a_0\ a_1\ a_2+9\ a_1\ a_2^2-3\ a_0^2\ a_3-5\ a_1^2\ a_3+2\ a_0\ a_2\ a_3+5\ a_2^2\ a_3-a_1\ a_3^2+a_3^3\right)+l_4,\ -8\ a_0\ a_1^3+12\ a_0^2\ a_1\ a_2-12\ a_1^3\ a_2+24\ a_0\ a_1\ a_2^2+12\ a_1\ a_2^3-4\ a_0^3\ a_3-12\ a_0\ a_1^2\ a_3-24\ a_1^2\ a_2\ a_3+12\ a_0\ a_2^2\ a_3+8\ a_2^3\ a_3-12\ a_1\ a_2\ a_3^2+4\ a_0\ a_3^3+m_4\}$$

This is the basis for the ideal of polynomials on $\mathbb{R}^8$ (corresponding to the 4 coefficients a_0,a_1,a_2,a_3 and the 4 invariants j_2,h_2,l_4,m_4

We seek potential relations among the invariants by finding a basis for the ideal generated by the definitions of the invariants intersected with the polynomials in j_2,h_2,l_4,m_4 that do not depend on a_0,a_1,a_2,a_3, i.e. we are eliminating the coefficients between the relations defining the invariants.

In[47]:=
```
GroebnerBasis[Invariants, {m₄, l₄, h₂, j₂},
  {a₀, a₁, a₂, a₃}, MonomialOrder → EliminationOrder]
```

Out[47]=

$$\{h_2 j_2^3 - 4 l_4^2 - 4 m_4^2\}$$

We see that there is one identity that allows us to replace $m_4{}^2$ by an expression in the other invariants. There are no further relations, so this implies l_4, h_2 and j_2 are algebraically independent.

Alternate choices for the fundamental invariants are the trace and determinant of Γ^{*^2}, which are themselves O(2) invariants, so they can be expressed in terms of the fundamental invariants j_2, h_2 and l_4

In[48]:=
```
Γ = Table[D[f / 6, x[i], x[j], x[k]], {i, 1, n}, {j, 1, n}, {k, 1, n}];
Q = Simplify[TensorContract[Γ⊗Γ, {{1, 4}, {2, 5}}]];
MatrixForm[Q]
```

Out[49]//MatrixForm=

$$\begin{pmatrix} a_1^2 + 2 a_2^2 + a_3^2 & a_0 a_1 + a_2 (2 a_1 + a_3) \\ a_0 a_1 + a_2 (2 a_1 + a_3) & a_0^2 + 2 a_1^2 + a_2^2 \end{pmatrix}$$

This is the matrix Γ^{*2}.

In[50]:=
```
NewInvariants = {τ - Tr[Q], δ - Det[Q]}
```

Out[50]=

$$\{\tau - a_0^2 - 3 a_1^2 - 3 a_2^2 - a_3^2,$$
$$\delta - 2 a_1^4 + 4 a_0 a_1^2 a_2 - 2 a_0^2 a_2^2 - a_1^2 a_2^2 - 2 a_2^4 + 2 a_0 a_1 a_2 a_3 + 4 a_1 a_2^2 a_3 - a_0^2 a_3^2 - 2 a_1^2 a_3^2\}$$

In[51]:=
```
GroebnerBasis[Join[NewInvariants, Invariants],
  {τ, δ, l₄, h₂, j₂}, {m₄, a₀, a₁, a₂, a₃}, MonomialOrder → EliminationOrder]
```

Out[51]=

$$\{16 \tau - h_2 - 12 j_2, -1024 \delta + h_2^2 + 8 h_2 j_2 + 80 j_2^2 - 128 l_4\}$$

This shows that $\mathrm{Tr}[\Gamma^{*2}] = \frac{h_2+12 j_2}{16}$, $\mathrm{Det}[\Gamma^{*2}] = \frac{h_2^2+8 h_2 j_2+80 j_2^2-128 l_4}{1024}$

In[52]:=
```
DiagSetting = {a₃ → α, a₂ → 0, a₁ → 0, a₀ → β}
```

Out[52]=

$$\{a_3 \to \alpha, a_2 \to 0, a_1 \to 0, a_0 \to \beta\}$$

In[53]:=
```
Simplify[Invariants /. DiagSetting]
```

Out[53]=

$$\{-\alpha^2 - \beta^2 + j_2, -4 (\alpha^2 + \beta^2) + h_2, -\alpha^4 + 6 \alpha^2 \beta^2 - \beta^4 + l_4, 4 \alpha^3 \beta - 4 \alpha \beta^3 + m_4\}$$

In[54]:=
```
GroebnerBasis[{-α² - β² + j₂, -4 (α² + β²) + h₂, -α⁴ + 6 α² β² - β⁴ + l₄, 4 α³ β - 4 α β³ + m₄},
  {α, h₂, j₂, l₄, m₄}, {β}]
```

Out[54]=

$$\{j_2^4 - l_4^2 - m_4^2, h_2 - 4 j_2, 8 \alpha^4 - 8 \alpha^2 j_2 + j_2^2 - l_4\}$$

In[55]:=
```
u /. DiagSetting
```
Out[55]=
$$\{\alpha, \beta\}$$

In[56]:=
```
w /. DiagSetting
```
Out[56]=
$$\{\alpha^3 - 3\,\alpha\,\beta^2, -3\,\alpha^2\,\beta + \beta^3\}$$

In[57]:=
```
Simplify[(Invariants /. DiagSetting) /. {β → α}]
```
Out[57]=
$$\{-2\,\alpha^2 + j_2, -8\,\alpha^2 + h_2, 4\,\alpha^4 + l_4, m_4\}$$

In[58]:=
```
GroebnerBasis[{-2 α^2 + j2, -8 α^2 + h2, 4 α^4 + l4, m4}, {α, h2, j2, l4, m4}]
```
Out[58]=
$$\{m_4, j_2^2 + l_4, h_2 - 4\,j_2, 2\,\alpha^2 - j_2\}$$

In[59]:=
```
Simplify[(Invariants /. DiagSetting) /. {β → 0}]
```
Out[59]=
$$\{-\alpha^2 + j_2, -4\,\alpha^2 + h_2, -\alpha^4 + l_4, m_4\}$$

In[60]:=
```
GroebnerBasis[{-α^2 + j2, -4 α^2 + h2, -α^4 + l4, m4}, {α, h2, j2, l4, m4}]
```
Out[60]=
$$\{m_4, j_2^2 - l_4, h_2 - 4\,j_2, \alpha^2 - j_2\}$$

Section 7.2

Fully decoupleable 3 × 3 × 3 tensors

In[61]:=
```
n = 3; vars = Table[x[i], {i, 1, n}]; f = Sum[β_i x[i]^3, {i, 1, n}]
```
Out[61]=
$$\beta_1\,x[1]^3 + \beta_2\,x[2]^3 + \beta_3\,x[3]^3$$

This is the cubic polynomial corresponding to a fully decoupleable tensor.

In[62]:=
```
Γ = Simplify[Table[D[f, x[i], x[j], x[k]], {k, 1, n}, {j, 1, n}, {i, 1, n}] / 6];
u = Simplify[Table[D[Laplacian[f, vars], x[i]], {i, 1, n}] / 6];
f3 = (n + 2) f - 3 (u.vars) (vars.vars)
```
Out[64]=
$$-3\,(\beta_1\,x[1] + \beta_2\,x[2] + \beta_3\,x[3])\,\left(x[1]^2 + x[2]^2 + x[3]^2\right) + 5\,\left(\beta_1\,x[1]^3 + \beta_2\,x[2]^3 + \beta_3\,x[3]^3\right)$$

This is the trace-free part

In[65]:=
```
𝒟 = Simplify[Table[D[f3, x[i], x[j], x[k]], {k, 1, n}, {j, 1, n}, {i, 1, n}] / 6];
v = Simplify[TensorContract[(𝒟⊗𝒟)⊗𝒟, {{1, 4}, {2, 5}, {3, 7}, {6, 8}}]];
w = Simplify[𝒟.u.u];
```

We have now computed the vectors u,v and w and the trace-free tensor $\mathcal{D}$, which are the ingredients needed to compute the Integrity basis given by Olive and Auffray.

In[68]:=
```
Clear[H, J, K, L, M, Q];
Q = Simplify[TensorContract[𝒟⊗𝒟, {{1, 4}, {2, 5}}]];
γuu = w;
Coeffs = Table[βi, {i, 1, n}];
Trivialize = Table[Coeffs[[j]] → 0, {j, 1, Length[Coeffs]}];
IntegrityPolys =
  {H[2] - Simplify[Tr[Q]], H[4] - Simplify[Tr[Q.Q]], J[2] - Simplify[u.u],
   L[4] - Simplify[γuu.u], H[6] - Simplify[v.v], H[10] - Simplify[𝒟.v.v.v] ,
   J[4] - Simplify[u.Q.u], K[4] - Simplify[Tr[Q.(𝒟.u)]],
   J[6] - Simplify[(u.Q).γuu], K[6] - Simplify[v.w], L[6] - Simplify[(u.Q).v],
   M[6] - Simplify[γuu.γuu], H[8] - Simplify[(u.Q).(Q.v)]};
OAInvariants = IntegrityPolys /. Trivialize;
```

The Ideal IntegrityPolys is generated by the polynomials defining the Integrity basis elements in terms of the coefficients of a fully decoupled tensor, with the labels of the Integrity invariants as the slack variables.

In[74]:=
```
IntegrityPolys
```

Out[74]=

$$\begin{aligned}&\{H[2]-10\left(\beta_1^2+\beta_2^2+\beta_3^2\right),\ H[4]-44\beta_1^4-44\beta_2^4-58\beta_2^2\beta_3^2-44\beta_3^4-58\beta_1^2\left(\beta_2^2+\beta_3^2\right),\\
&J[2]-\beta_1^2-\beta_2^2-\beta_3^2,\ L[4]-2\left(\beta_1^4+\beta_2^4-3\beta_2^2\beta_3^2+\beta_3^4-3\beta_1^2\left(\beta_2^2+\beta_3^2\right)\right),\\
&H[6]-4\left(\beta_3^2\left(\beta_1^2+\beta_2^2-4\beta_3^2\right)^2+\beta_2^2\left(\beta_1^2-4\beta_2^2+\beta_3^2\right)^2+\beta_1^2\left(-4\beta_1^2+\beta_2^2+\beta_3^2\right)^2\right),\\
&H[10]-8\left(128\beta_1^{10}+128\beta_2^{10}-60\beta_2^8\beta_3^2-95\beta_2^6\beta_3^4-95\beta_2^4\beta_3^6-\right.\\
&\quad 60\beta_2^2\beta_3^8+128\beta_3^{10}-60\beta_1^8\left(\beta_2^2+\beta_3^2\right)+\beta_1^6\left(-95\beta_2^4+60\beta_2^2\beta_3^2-95\beta_3^4\right)+\\
&\quad\left.\beta_1^4\left(-95\beta_2^6+90\beta_2^4\beta_3^2+90\beta_2^2\beta_3^4-95\beta_3^6\right)-30\beta_1^2\left(2\beta_2^8-2\beta_2^6\beta_3^2-3\beta_2^4\beta_3^4-2\beta_2^2\beta_3^6+2\beta_3^8\right)\right),\\
&J[4]-2\left(3\beta_1^4+3\beta_2^4+\beta_2^2\beta_3^2+3\beta_3^4+\beta_1^2\left(\beta_2^2+\beta_3^2\right)\right),\ K[4]-8\beta_1^4-8\beta_2^4+4\beta_2^2\beta_3^2-8\beta_3^4+4\beta_1^2\left(\beta_2^2+\beta_3^2\right),\\
&J[6]-12\beta_1^6-12\beta_2^6+19\beta_2^4\beta_3^2+19\beta_2^2\beta_3^4-12\beta_3^6+19\beta_1^4\left(\beta_2^2+\beta_3^2\right)+\beta_1^2\left(19\beta_2^4+18\beta_2^2\beta_3^2+19\beta_3^4\right),\\
&K[6]-2\left(8\beta_1^6+8\beta_2^6-11\beta_2^4\beta_3^2-11\beta_2^2\beta_3^4+8\beta_3^6-11\beta_1^4\left(\beta_2^2+\beta_3^2\right)+\beta_1^2\left(-11\beta_2^4+18\beta_2^2\beta_3^2-11\beta_3^4\right)\right),\\
&L[6]-6\left(8\beta_1^6+8\beta_2^6-\beta_2^4\beta_3^2-\beta_2^2\beta_3^4+8\beta_3^6-\beta_1^4\left(\beta_2^2+\beta_3^2\right)-\beta_1^2\left(\beta_2^2+\beta_3^2\right)^2\right),\\
&M[6]-\beta_3^2\left(3\beta_1^2+3\beta_2^2-2\beta_3^2\right)^2-\beta_2^2\left(3\beta_1^2-2\beta_2^2+3\beta_3^2\right)^2-\left(2\beta_1^3-3\beta_1\left(\beta_2^2+\beta_3^2\right)\right)^2,\\
&H[8]-4\left(72\beta_1^8+18\beta_1^6\left(\beta_2^2+\beta_3^2\right)-11\beta_1^4\left(3\beta_2^4+\beta_2^2\beta_3^2+3\beta_3^4\right)+\right.\\
&\quad\left.\beta_1^2\left(18\beta_2^6-11\beta_2^4\beta_3^2-11\beta_2^2\beta_3^4+18\beta_3^6\right)+3\left(24\beta_2^8+6\beta_2^6\beta_3^2-11\beta_2^4\beta_3^4+6\beta_2^2\beta_3^6+24\beta_3^8\right)\right)\}\end{aligned}$$

Characteristic Polynomial coefficients

In[75]:=
```
Γstarsqrd = Simplify[TensorContract[Γ⊗Γ, {{1, 4}, {2, 5}}]];
```

In[76]:=
```
Clear[q, ξ];
ξ = Rest[Reverse[CoefficientList[Simplify[Det[λ IdentityMatrix[3] + Γstarsqrd]], λ]]]
```

Out[76]=

$$\left\{\beta_1^2+\beta_2^2+\beta_3^2,\ \beta_1^2\beta_2^2+\beta_1^2\beta_3^2+\beta_2^2\beta_3^2,\ \beta_1^2\beta_2^2\beta_3^2\right\}$$

As expected, these are the elementary symmetric polynomials of the quantities $\beta_i{}^2$.

In[77]:=
```
DiagInvars = Table[qᵢ - ξ⟦i⟧, {i, 1, 3}]
```
Out[77]=

$$\{q_1 - \beta_1^2 - \beta_2^2 - \beta_3^2,\ q_2 - \beta_1^2\beta_2^2 - \beta_1^2\beta_3^2 - \beta_2^2\beta_3^2,\ q_3 - \beta_1^2\beta_2^2\beta_3^2\}$$

This is the basis of invariants for the group $G_R = S_3 \times (\mathbb{Z}_2)^3$. Since all the Olive and Auffray invariants, when restricted to fully decoupled tensors, are also G_R invariants, we can express them in terms of the quantities q_i

In[78]:=
```
Table[GroebnerBasis[Join[{IntegrityPolys⟦i⟧}, DiagInvars],
   Join[{OAInvariants⟦i⟧}, {q₁, q₂, q₃}], {β₁, β₂, β₃},
   MonomialOrder → EliminationOrder]⟦1⟧, {i, 1, Length[OAInvariants]}]
```
Out[78]=

$$\{H[2] - 10 q_1,\ -H[4] + 44 q_1^2 - 30 q_2,\ J[2] - q_1,$$
$$-L[4] + 2 q_1^2 - 10 q_2,\ -H[6] + 64 q_1^3 - 220 q_1 q_2 + 300 q_3,$$
$$-H[10] + 1024 q_1^5 - 5600 q_1^3 q_2 + 5800 q_1 q_2^2 + 7600 q_1^2 q_3 - 7000 q_2 q_3,$$
$$-J[4] + 6 q_1^2 - 10 q_2,\ -K[4] + 8 q_1^2 - 20 q_2,\ -J[6] + 12 q_1^3 - 55 q_1 q_2 + 75 q_3,$$
$$-K[6] + 16 q_1^3 - 70 q_1 q_2 + 150 q_3,\ -L[6] + 48 q_1^3 - 150 q_1 q_2 + 150 q_3,$$
$$-M[6] + 4 q_1^3 - 15 q_1 q_2 + 75 q_3,\ -H[8] + 288 q_1^4 - 1080 q_1^2 q_2 + 300 q_2^2 + 1300 q_1 q_3\}$$

This is the ideal corresponding to the relations in Eq. (7.1).

Section 7.3

Partially decoupleable 3 × 3 × 3 tensors

In[79]:=
```
n = 3;
vars = Table[x[i], {i, 1, n}];
f = 3 α x[1] (x[1]^2 + x[2]^2) +
   γ₁ (3 x[2]^2 × x[1] - x[1]^3) + γ₂ (3 x[1]^2 × x[2] - x[2]^3) + β₃ x[3]^3
```
Out[79]=

$$3\alpha\, x[1]\,(x[1]^2 + x[2]^2) + \gamma_1\,(-x[1]^3 + 3\, x[1]\, x[2]^2) + \gamma_2\,(3\, x[1]^2\, x[2] - x[2]^3) + \beta_3\, x[3]^3$$

This is the canonical form corresponding to a partially decoupleable tensor.

In[80]:=
```
Coeffs = {α, γ₁, γ₂, β₃};
```

This is the list ot tensor coefficients in the canonical form.

In[81]:=
```
Γ = Simplify[Table[D[f, x[i], x[j], x[k]], {k, 1, n}, {j, 1, n}, {i, 1, n}] / 6]
```
Out[81]=

$$\{\{\{3\alpha - \gamma_1, \gamma_2, 0\}, \{\gamma_2, \alpha + \gamma_1, 0\}, \{0, 0, 0\}\},$$
$$\{\{\gamma_2, \alpha + \gamma_1, 0\}, \{\alpha + \gamma_1, -\gamma_2, 0\}, \{0, 0, 0\}\}, \{\{0, 0, 0\}, \{0, 0, 0\}, \{0, 0, \beta_3\}\}\}$$

In[82]:=
```
Table[MatrixForm[Γ⟦i⟧], {i, 1, n}]
```
Out[82]=

$$\left\{\begin{pmatrix} 3\alpha - \gamma_1 & \gamma_2 & 0 \\ \gamma_2 & \alpha + \gamma_1 & 0 \\ 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} \gamma_2 & \alpha + \gamma_1 & 0 \\ \alpha + \gamma_1 & -\gamma_2 & 0 \\ 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & \beta_3 \end{pmatrix}\right\}$$

In[83]:=
```
u =  Simplify[Table[D[Laplacian[f, vars], x[i]], {i, 1, n}] / 6];
f3 = (n + 2) f - 3 (u.vars) (vars.vars)
```

Out[84]=

$$-3\ (4\,\alpha\,x[1] + \beta_3\,x[3])\ \left(x[1]^2 + x[2]^2 + x[3]^2\right) + 5\ \left(3\,\alpha\,x[1]\ \left(x[1]^2 + x[2]^2\right) + \gamma_1\ \left(-x[1]^3 + 3\,x[1]\,x[2]^2\right) + \gamma_2\ \left(3\,x[1]^2\,x[2] - x[2]^3\right) + \beta_3\,x[3]^3\right)$$

This is the trace-free part

In[85]:=
```
𝒟 = Simplify[Table[D[f3, x[i], x[j], x[k]], {k, 1, n}, {j, 1, n}, {i, 1, n}] / 6];
v = Simplify[TensorContract[(𝒟⊗𝒟)⊗𝒟, {{1, 4}, {2, 5}, {3, 7}, {6, 8}}]];
w = Simplify[𝒟.u.u];
```

The calculations follow the same steps as in the fully decoupled case.

In[88]:=
```
Clear[H, J, K, L, M, Q];
v = Simplify[TensorContract[(𝒟⊗𝒟)⊗𝒟, {{1, 4}, {2, 5}, {3, 7}, {6, 8}}]];
w = Simplify[𝒟.u.u];
Q = Simplify[TensorContract[𝒟⊗𝒟, {{1, 4}, {2, 5}}]];
γuu = w;
Trivialize = Table[Coeffs[[j]] → 0, {j, 1, Length[Coeffs]}];
```

In what follows, we use the decomposition $\mathcal{I} = \mathcal{I}^{+} \cup \mathcal{I}'$, corresponding to the ‘Fundamental’ and ‘Secondary’ invariants. This is not standard terminology!!

In[94]:=
```
FundamentalRelations = {H[2] - Simplify[Tr[Q]],
    H[4] - Simplify[Tr[Q.Q]], J[2] - Simplify[u.u], L[4] - Simplify[γuu.u]};
FundamentalInvariants = FundamentalRelations /. Trivialize;
SecondaryRelations = { H[6] - Simplify[v.v], H[10] - Simplify[𝒟.v.v.v] ,
    J[4] - Simplify[u.Q.u], K[4] - Simplify[Tr[Q.(𝒟.u)]],
    J[6] - Simplify[(u.Q).γuu], K[6] - Simplify[v.w], L[6] - Simplify[(u.Q).v],
    M[6] - Simplify[γuu.γuu], H[8] - Simplify[(u.Q).(Q.v)]};
SecondaryInvariants = SecondaryRelations /. Trivialize;
PartialDecoupleRelations = Join[FundamentalRelations, SecondaryRelations];
```

In[99]:=
```
G = Table[Γ[[i, j, k]], {i, 1, 2}, {j, 1, 2}, {k, 1, 2}]
```

Out[99]=

$$\{\{\{3\,\alpha - \gamma_1,\ \gamma_2\},\ \{\gamma_2,\ \alpha + \gamma_1\}\},\ \{\{\gamma_2,\ \alpha + \gamma_1\},\ \{\alpha + \gamma_1,\ -\gamma_2\}\}\}$$

This is the 2 × 2 × 2 block of Γ

In[100]:=
```
GStarSqrd = Expand[TensorContract[G⊗G, {{1, 4}, {2, 5}}]]
```

Out[100]=

$$\left\{\left\{10\,\alpha^2 - 4\,\alpha\,\gamma_1 + 2\,\gamma_1^2 + 2\,\gamma_2^2,\ 4\,\alpha\,\gamma_2\right\},\ \left\{4\,\alpha\,\gamma_2,\ 2\,\alpha^2 + 4\,\alpha\,\gamma_1 + 2\,\gamma_1^2 + 2\,\gamma_2^2\right\}\right\}$$

In[101]:=
```
MatrixForm[GStarSqrd]
```

Out[101]//MatrixForm=

$$\begin{pmatrix} 10\,\alpha^2 - 4\,\alpha\,\gamma_1 + 2\,\gamma_1^2 + 2\,\gamma_2^2 & 4\,\alpha\,\gamma_2 \\ 4\,\alpha\,\gamma_2 & 2\,\alpha^2 + 4\,\alpha\,\gamma_1 + 2\,\gamma_1^2 + 2\,\gamma_2^2 \end{pmatrix}$$

In[102]:=

```
ured = TensorContract[G, {1, 2}]
```

Out[102]=

$\{4\,\alpha, 0\}$

This is the trace of the 2 × 2 × 2 block

In[103]:=

```
Clear[q];
defns = {q₁ - β₃², q₂ - ured.ured, q₄ - Tr[GStarSqrd], q₃ - Det[GStarSqrd]}
```

Out[104]=

$$\{q_1 - \beta_3^2,\ -16\,\alpha^2 + q_2,\ -12\,\alpha^2 + q_4 - 4\,\gamma_1^2 - 4\,\gamma_2^2,\ -20\,\alpha^4 + q_3 - 32\,\alpha^3\,\gamma_1 - 8\,\alpha^2\,\gamma_1^2 - 4\,\gamma_1^4 - 8\,\alpha^2\,\gamma_2^2 - 8\,\gamma_1^2\,\gamma_2^2 - 4\,\gamma_2^4\}$$

These are the expressions of the O(2) x $\mathbb{Z}_2$ invariants (the quantities q_i) in terms of the parameters defining the Canonical form.

In[105]:=

```
EliminateCoeffs = Table[
  GroebnerBasis[Join[{FundamentalRelations[[i]]}, defns], Join[FundamentalInvariants,
     {q₃, q₄, q₂, q₁}], Coeffs, MonomialOrder → EliminationOrder][[1]], {i, 1, 4}]
```

Out[105]=

$$\{H[2] - 10\,q_1 + 15\,q_2 - 25\,q_4,\ -H[4] + 44\,q_1^2 - 42\,q_1\,q_2 + 144\,q_2^2 - 30\,q_3 + 100\,q_1\,q_4 - 420\,q_2\,q_4 + 320\,q_4^2,\ J[2] - q_1 - q_2,\ -2\,L[4] + 4\,q_1^2 - 12\,q_1\,q_2 + 4\,q_2^2 - 20\,q_3 - 5\,q_2\,q_4 + 5\,q_4^2\}$$

These are the relations right after (7.2) expressing the elements in $\mathcal{I}^+$ in terms of the O(2) x $\mathbb{Z}_2$ invariants q_i. We can invert these relations and solve for the quantities q_i.

In[106]:=

```
TriangularSystem =
 GroebnerBasis[EliminateCoeffs, {q₄, q₃, q₂, q₁, H[2], H[4], J[2], L[4]}]
```

Out[106]=

$$\{-H[2]^2 + 2\,H[4] + 3\,H[2] \times J[2] - 6\,J[2]^2 - 6\,L[4] + 9\,H[2]\,q_1 - 90\,J[2]\,q_1,\ -J[2] + q_1 + q_2,\ 8\,H[2]^2 - 25\,H[4] - 60\,H[2] \times J[2] - 1500\,J[2]^2 + 1200\,L[4] + 11\,250\,J[2]\,q_1 - 11\,250\,q_1^2 + 11\,250\,q_3,\ -H[2] - 15\,J[2] + 25\,q_1 + 25\,q_4\}$$

In[107]:=

```
Substitutions = Table[Solve[TriangularSystem[[i]] == 0, qᵢ][[1, 1]], {i, 1, 4}]
```

Out[107]=

$$\left\{q_1 \to \frac{H[2]^2 - 2\,H[4] - 3\,H[2] \times J[2] + 6\,J[2]^2 + 6\,L[4]}{9\,(H[2] - 10\,J[2])},\ q_2 \to J[2] - q_1,\ q_3 \to \frac{-8\,H[2]^2 + 25\,H[4] + 60\,H[2] \times J[2] + 1500\,J[2]^2 - 1200\,L[4] - 11\,250\,J[2]\,q_1 + 11\,250\,q_1^2}{11\,250},\ q_4 \to \frac{1}{25}\,(H[2] + 15\,J[2] - 25\,q_1)\right\}$$

These are the substitutions implied by Eq. (7.3).

Lemma 7.3. Finding real solutions for the parameters in terms of the qtildes and

the associated inequalities that are needed for solvability

In[108]:=

```
GroebnerBasis[defns, Join[{β3}, Table[qi, {i, 1, 4}]], {α, γ1, γ2}]
```

Out[108]=

$$\{-q_1 + \beta_3^2\}$$

We need $q_1 \geq 0$ to solve for a real β_3.

In[109]:=

```
GroebnerBasis[defns, Join[{γ2}, Table[qi, {i, 1, 4}]], {α, γ1, β3}]
```

Out[109]=

$$\{q_2^4 - 4 q_2^2 q_3 + 16 q_3^2 - 2 q_2^3 q_4 + 8 q_2 q_3 q_4 + 2 q_2^2 q_4^2 - 8 q_3 q_4^2 - 2 q_2 q_4^3 + q_4^4 + 4 q_2^3 \gamma_2^2\}$$

In[110]:=

```
Collect[%[[1]], γ2]
```

Out[110]=

$$q_2^4 - 4 q_2^2 q_3 + 16 q_3^2 - 2 q_2^3 q_4 + 8 q_2 q_3 q_4 + 2 q_2^2 q_4^2 - 8 q_3 q_4^2 - 2 q_2 q_4^3 + q_4^4 + 4 q_2^3 \gamma_2^2$$

We get a linear equation for $\gamma_2{}^2$.

In[111]:=

```
Eqnγ2 = (% /. {γ2^2 → γ2sqrd}) == 0
```

Out[111]=

$$4\, \gamma 2\text{sqrd}\, q_2^3 + q_2^4 - 4 q_2^2 q_3 + 16 q_3^2 - 2 q_2^3 q_4 + 8 q_2 q_3 q_4 + 2 q_2^2 q_4^2 - 8 q_3 q_4^2 - 2 q_2 q_4^3 + q_4^4 == 0$$

In[112]:=

```
Solve[Eqnγ2, γ2sqrd][[1]]
```

Out[112]=

$$\left\{\gamma 2\text{sqrd} \to \frac{-q_2^4 + 4 q_2^2 q_3 - 16 q_3^2 + 2 q_2^3 q_4 - 8 q_2 q_3 q_4 - 2 q_2^2 q_4^2 + 8 q_3 q_4^2 + 2 q_2 q_4^3 - q_4^4}{4 q_2^3}\right\}$$

To get a real solution, we therefore need $q_2 \neq 0$ and the fraction (or equivalently the product of the numerator and the denominator in the above expression) is non-negative. We see below that $q_2 > 0$ is necessary, and along with this condition, we will need that the numerator be greater than or equal to zero.

In[113]:=

```
Eqnsaαγ = GroebnerBasis[defns, Join[{α, γ1}, Table[qi, {i, 1, 4}]], {γ2, β3}]
```

Out[113]=

$$\{-q_2^4 + 16 q_2^2 q_3 - 64 q_3^2 + 4 q_2^3 q_4 - 32 q_2 q_3 q_4 - 8 q_2^2 q_4^2 + 32 q_3 q_4^2 + 8 q_2 q_4^3 - 4 q_4^4 + 16 q_2^3 \gamma_1^2,$$
$$\alpha q_2^2 - 8 \alpha q_3 - 2 \alpha q_2 q_4 + 2 \alpha q_4^2 + q_2^2 \gamma_1,$$
$$q_2^3 - 8 q_2 q_3 - 4 q_2^2 q_4 + 16 q_3 q_4 + 6 q_2 q_4^2 - 4 q_4^3 + 128 \alpha q_3 \gamma_1 - 32 \alpha q_4^2 \gamma_1 - 16 q_2^2 \gamma_1^2,$$
$$q_2^2 - 8 q_3 - 2 q_2 q_4 + 2 q_4^2 + 16 \alpha q_2 \gamma_1,\ 16 \alpha^2 - q_2\}$$

In[114]:=

```
Eqnsaαγ[[5]] == 0
```

Out[114]=

$$16 \alpha^2 - q_2 == 0$$

To find a real solution, we need $q_2 \geq 0$. This, along with the earlier requirement $q_2 \neq 0$ implies that $q_2 > 0$.

In[115]:=

```
Eqnsaαγ[[2]] == 0
```

Out[115]=

$$\alpha q_2^2 - 8 \alpha q_3 - 2 \alpha q_2 q_4 + 2 \alpha q_4^2 + q_2^2 \gamma_1 == 0$$

In[116]:=

```
Simplify[Solve[Eqnsaαγ[[2]] == 0, γ1][[1]]]
```

Out[116]=

$$\left\{\gamma_1 \to -\frac{\alpha \left(q_2^2 - 8 q_3 - 2 q_2 q_4 + 2 q_4^2\right)}{q_2^2}\right\}$$

We get no further conditions from the solvability for γ_1

Example 7.4

In[117]:=

```
n = 3;
vars = Table[x[i], {i, 1, n}];
f = Sum[2 i x[i] ^3, {i, 1, n}] + (3 x[1] ^2 × x[2] - x[2] ^3) - 12 x[1] × x[2] × x[3]
```

Out[117]=

$$2 x[1]^3 + 3 x[1]^2 x[2] + 3 x[2]^3 - 12 x[1] \times x[2] \times x[3] + 6 x[3]^3$$

This is an explicit numerical example.

In[118]:=

```
Γ = Simplify[Table[D[f, x[i], x[j], x[k]], {k, 1, n}, {j, 1, n}, {i, 1, n}] / 6];
u = Simplify[Table[D[Laplacian[f, vars], x[i]], {i, 1, n}] / 6];
f3 = (n + 2) f - 3 (u.vars) (vars.vars);
𝒟 = Simplify[Table[D[f3, x[i], x[j], x[k]], {k, 1, n}, {j, 1, n}, {i, 1, n}] / 6];
Clear[H, J, K, L, M, Q];
v = Simplify[TensorContract[(𝒟⊗𝒟)⊗𝒟, {{1, 4}, {2, 5}, {3, 7}, {6, 8}}]];
w = Simplify[𝒟.u.u];
Q = Simplify[TensorContract[𝒟⊗𝒟, {{1, 4}, {2, 5}}]];
γuu = w;
FundamentalValues = {H[2] → Simplify[Tr[Q]],
   H[4] → Simplify[Tr[Q.Q]], J[2] → Simplify[u.u], L[4] → Simplify[γuu.u]};
SecondaryValues =
  { H[6] → Simplify[v.v], H[10] - Simplify[𝒟.v.v.v] , J[4] - Simplify[u.Q.u],
   K[4] - Simplify[Tr[Q.(𝒟.u)]], J[6] - Simplify[(u.Q).γuu], K[6] - Simplify[v.w],
   L[6] - Simplify[(u.Q).v], M[6] - Simplify[γuu.γuu], H[8] - Simplify[(u.Q).(Q.v)]};
```

In[129]:=

```
FundamentalValues
```

Out[129]=

```
{H[2] → 1060, H[4] → 518 384, J[2] → 56, L[4] → -4528}
```

In[130]:=

```
Specialization = FundamentalRelations /. FundamentalValues
```

Out[130]=

$$\{1060 - 10\,(6\,\alpha^2 + \beta_3^2 + 10\,\gamma_1^2 + 10\,\gamma_2^2),\ 518\,384 - 32\,\alpha^2\,\beta_3^2 - (32\,\alpha^2 + 6\,\beta_3^2)^2 - 800\,\alpha^2\,\gamma_2^2 - 4\,(13\,\alpha^2 + \beta_3^2 - 10\,\alpha\,\gamma_1 + 25\,\gamma_1^2 + 25\,\gamma_2^2)^2 - 4\,(\beta_3^2 + (\alpha + 5\,\gamma_1)^2 + 25\,\gamma_2^2)^2,\ 56 - 16\,\alpha^2 - \beta_3^2,\ -4528 - 2\,(-48\,\alpha^2\,\beta_3^2 + \beta_3^4 + 32\,\alpha^3\,(3\,\alpha - 5\,\gamma_1))\}$$

In[131]:=

```
GroebnerBasis[Specialization, Coeffs]
```

Out[131]=

$$\{332 + 15\,\beta_3^2,\ 3\,173\,103\,609 + 125\,768\,785\,\gamma_2^2,\ -52\,993\,421\,209 + 1\,509\,225\,420\,\gamma_1^2,\ 230\,203\,\alpha - 85\,849\,\gamma_1\}$$

Section 7.5

Expressing the invariants in $\mathcal{I}'$ in terms of $\mathcal{I}^+$

In[132]:=

```
n = 3; Clear[x];
vars = Table[x[i], {i, 1, n}];
f = 3 α x[1] (x[1]^2 + x[2]^2) +
  γ1 (3 x[2]^2 × x[1] - x[1]^3) + γ2 (3 x[1]^2 × x[2] - x[2]^3) + β3 x[3]^3
```

Out[132]=

$$3\,\alpha\,x[1]\,(x[1]^2 + x[2]^2) + \gamma_1\,(-x[1]^3 + 3\,x[1]\,x[2]^2) + \gamma_2\,(3\,x[1]^2\,x[2] - x[2]^3) + \beta_3\,x[3]^3$$

In[133]:=

```
Γ = Simplify[Table[D[f, x[i], x[j], x[k]], {k, 1, n}, {j, 1, n}, {i, 1, n}] / 6]
```

Out[133]=

$$\{\{\{3\,\alpha - \gamma_1, \gamma_2, 0\}, \{\gamma_2, \alpha + \gamma_1, 0\}, \{0, 0, 0\}\}, \{\{\gamma_2, \alpha + \gamma_1, 0\}, \{\alpha + \gamma_1, -\gamma_2, 0\}, \{0, 0, 0\}\}, \{\{0, 0, 0\}, \{0, 0, 0\}, \{0, 0, \beta_3\}\}\}$$

In[134]:=

```
u =  Simplify[Table[D[Laplacian[f, vars], x[i]], {i, 1, n}] / 6];
f₃ = f - 3 (u.vars) (vars.vars) / 5;
𝒟 = Simplify[5 Table[D[f₃, x[i], x[j], x[k]], {k, 1, n}, {j, 1, n}, {i, 1, n}] / 6];
Clear[H, J, K, L, M, Q];
v = Simplify[TensorContract[(𝒟⊗𝒟)⊗𝒟, {{1, 4}, {2, 5}, {3, 7}, {6, 8}}]];
w = Simplify[𝒟.u.u];
Q = Simplify[TensorContract[𝒟⊗𝒟, {{1, 4}, {2, 5}}]];
γuu = w;
Coeffs = {α, γ₁, γ₂, β₃};
Eliminated = {α, γ₁, γ₂};
Trivialize = Table[Coeffs⟦j⟧ → 0, {j, 1, Length[Coeffs]}];
FundamentalRelations =
  {H[2] - Simplify[Tr[Q]], J[2] - Simplify[u.u], L[4] - Simplify[γuu.u]};

FundamentalRelations = {H[2] - Simplify[Tr[Q]],
   J[2] - Simplify[u.u], H[4] - Simplify[Tr[Q.Q]], L[4] - Simplify[γuu.u]};
FundamentalInvariants = Join[{β₃}, (FundamentalRelations /. Trivialize)];
SecondaryRelations = { J[4] - Simplify[u.Q.u],
   K[4] - Simplify[Tr[Q.(𝒟.u)]], H[6] - Simplify[v.v], J[6] - Simplify[(u.Q).γuu],
   K[6] - Simplify[v.w], L[6] - Simplify[(u.Q).v], M[6] - Simplify[γuu.γuu],
   H[8] - Simplify[(u.Q).(Q.v)], H[10] - Simplify[𝒟.v.v.v] };
SecondaryInvariants = SecondaryRelations /. Trivialize;
PartialDecoupleRelations = Join[FundamentalRelations, SecondaryRelations];
OAInvariantsList = Join[FundamentalInvariants, SecondaryInvariants];
```

In[152]:=

```
GroebnerBasis[FundamentalRelations, FundamentalInvariants, Eliminated]
```

Out[152]=

$$\left\{-H[2]^2 + 2\,H[4] + 3\,H[2] \times J[2] - 6\,J[2]^2 - 6\,L[4] + 9\,H[2]\,\beta_3^2 - 90\,J[2]\,\beta_3^2\right\}$$

In[153]:=

```
NeccSuffRelations =
 Join[GroebnerBasis[FundamentalRelations, FundamentalInvariants, Eliminated],
  Table[GroebnerBasis[Join[{SecondaryRelations⟦i⟧}, FundamentalRelations], Join[
     {SecondaryInvariants⟦i⟧}, FundamentalInvariants], Eliminated]⟦2⟧, {i, 1, 9}]]
```

Out[153]=

```
{-H[2]^2 + 2 H[4] + 3 H[2] × J[2] - 6 J[2]^2 - 6 L[4] + 9 H[2] β3^2 - 90 J[2] β3^2,
 H[2]^2 - 2 H[4] - 12 H[2] × J[2] + 24 J[2]^2 + 18 J[4] - 12 L[4],
 2 H[2]^2 - 4 H[4] - 15 H[2] × J[2] + 66 J[2]^2 + 9 K[4] - 6 L[4], 13 H[2]^3 - 26 H[2] × H[4] +
  27 H[6] + 7 H[2]^2 J[2] - 146 H[4] × J[2] + 156 H[2] J[2]^2 - 372 J[2]^3 - 30 H[2] × L[4] +
  924 J[2] × L[4] - 234 H[4] β3^2 + 7272 J[2]^2 β3^2 + 1512 L[4] β3^2 + 8100 J[2] β3^4 - 8100 β3^6,
 H[2]^3 - 2 H[2] × H[4] - 6 H[4] × J[2] - 12 H[2] J[2]^2 + 72 J[2]^3 + 36 J[6] -
  12 H[2] × L[4] - 18 H[4] β3^2 + 144 J[2]^2 β3^2 + 324 L[4] β3^2 + 2700 J[2] β3^4 - 2700 β3^6,
 19 H[2]^3 - 38 H[2] × H[4] - 14 H[2]^2 J[2] - 86 H[4] × J[2] - 96 H[2] J[2]^2 +
  744 J[2]^3 + 162 K[6] - 48 H[2] × L[4] + 744 J[2] × L[4] - 342 H[4] β3^2 +
  8136 J[2]^2 β3^2 + 3456 L[4] β3^2 + 24 300 J[2] β3^4 - 24 300 β3^6,
 19 H[2]^3 - 38 H[2] × H[4] - 23 H[2]^2 J[2] - 149 H[4] × J[2] + 174 H[2] J[2]^2 +
  204 J[2]^3 - 48 H[2] × L[4] + 852 J[2] × L[4] + 81 L[6] - 261 H[4] β3^2 + 7488 J[2]^2 β3^2 +
  1998 L[4] β3^2 + 12 150 J[2] β3^4 - 12 150 β3^6, -H[2]^3 + 2 H[2] × H[4] + 14 H[2]^2 J[2] -
  22 H[4] × J[2] - 120 H[2] J[2]^2 + 552 J[2]^3 - 6 H[2] × L[4] - 420 J[2] × L[4] +
  324 M[6] + 18 H[4] β3^2 - 5544 J[2]^2 β3^2 + 2376 L[4] β3^2 + 24 300 J[2] β3^4 - 24 300 β3^6,
 -70 H[2]^4 + 361 H[2]^2 H[4] - 442 H[4]^2 + 1458 H[8] - 149 H[2]^3 J[2] -
  674 H[2] × H[4] × J[2] - 830 H[2]^2 J[2]^2 + 7216 H[4] J[2]^2 + 8868 H[2] J[2]^3 -
  30 792 J[2]^4 - 30 H[2]^2 L[4] - 3828 H[4] × L[4] - 408 H[2] × J[2] × L[4] + 3624 J[2]^2 L[4] +
  11 088 L[4]^2 - 5796 H[4] × J[2] β3^2 - 158 832 J[2]^3 β3^2 + 206 928 J[2] × L[4] β3^2 +
  40 500 H[4] β3^4 + 356 400 J[2]^2 β3^4 - 121 500 L[4] β3^4 - 1 895 400 J[2] β3^6,
 -70 H[2]^5 + 879 H[2]^3 H[4] - 1478 H[2] H[4]^2 + 2187 H[10] + 1161 H[2]^4 J[2] -
  2866 H[2]^2 H[4] × J[2] - 2506 H[4]^2 J[2] - 110 H[2]^3 J[2]^2 - 12 212 H[2] × H[4] J[2]^2 +
  159 160 H[2]^2 J[2]^3 - 230 888 H[4] J[2]^3 - 607 632 H[2] J[2]^4 + 1 226 976 J[2]^5 +
  630 H[2]^3 L[4] - 2040 H[2] × H[4] × L[4] - 204 H[2]^2 J[2] × L[4] +
  26 160 H[4] × J[2] × L[4] + 125 412 H[2] J[2]^2 L[4] + 423 096 J[2]^3 L[4] - 2448 H[2] L[4]^2 +
  35 928 J[2] L[4]^2 - 10 782 H[4]^2 β3^2 + 95 256 H[4] J[2]^2 β3^2 + 9 237 600 J[2]^4 β3^2 +
  71 496 H[4] × L[4] β3^2 + 505 440 J[2]^2 L[4] β3^2 + 35 640 L[4]^2 β3^2 + 251 100 H[4] × J[2] β3^4 +
  4 017 600 J[2]^3 β3^4 + 777 600 J[2] × L[4] β3^4 - 510 300 H[4] β3^6 + 5 832 000 J[2]^2 β3^6}
```

In[154]:=

```
Map[Length, NeccSuffRelations, {1}]
```

Out[154]=

```
{7, 6, 6, 14, 12, 14, 14, 14, 22, 32}
```

In[155]:=

```
SecondaryInvariants = Join[{H[4]}, SecondaryInvariants]
```

Out[155]=

```
{H[4], J[4], K[4], H[6], J[6], K[6], L[6], M[6], H[8], H[10]}
```

Redefine Secondary invariants to include H_4

In[156]:=

```
Normalizations = Table[D[NeccSuffRelations⟦i⟧, SecondaryInvariants⟦i⟧], {i, 1, 10}]
```

Out[156]=

```
{2, 18, 9, 27, 36, 162, 81, 324, 1458, 2187}
```

In[157]:=

```
SolveNeccSuff =
 Table[Solve[NeccSuffRelations⟦i⟧ == 0, SecondaryInvariants⟦i⟧]⟦1, 1⟧, {i, 1, 10}] /.
  {β3^k_ → q1^(k/2)}
```

$\{\beta_3{}^{k\_} \to q_1{}^{k/2}\}$

Out[157]=

$$\Big\{H[4] \to \frac{1}{2}\left(H[2]^2 - 3\,H[2]\times J[2] + 6\,J[2]^2 + 6\,L[4] - 9\,H[2]\,q_1 + 90\,J[2]\,q_1\right),$$

$$J[4] \to \frac{1}{18}\left(-H[2]^2 + 2\,H[4] + 12\,H[2]\times J[2] - 24\,J[2]^2 + 12\,L[4]\right),$$

$$K[4] \to \frac{1}{9}\left(-2\,H[2]^2 + 4\,H[4] + 15\,H[2]\times J[2] - 66\,J[2]^2 + 6\,L[4]\right),$$

$$H[6] \to \frac{1}{27}\Big(-13\,H[2]^3 + 26\,H[2]\times H[4] - 7\,H[2]^2\,J[2] + 146\,H[4]\times J[2] - 156\,H[2]\,J[2]^2 + 372\,J[2]^3 + 30\,H[2]\times L[4] - 924\,J[2]\times L[4] + 234\,H[4]\,q_1 - 7272\,J[2]^2\,q_1 - 1512\,L[4]\,q_1 - 8100\,J[2]\,q_1^2 + 8100\,q_1^3\Big),$$

$$J[6] \to \frac{1}{36}\Big(-H[2]^3 + 2\,H[2]\times H[4] + 6\,H[4]\times J[2] + 12\,H[2]\,J[2]^2 - 72\,J[2]^3 + 12\,H[2]\times L[4] + 18\,H[4]\,q_1 - 144\,J[2]^2\,q_1 - 324\,L[4]\,q_1 - 2700\,J[2]\,q_1^2 + 2700\,q_1^3\Big),$$

$$K[6] \to \frac{1}{162}\Big(-19\,H[2]^3 + 38\,H[2]\times H[4] + 14\,H[2]^2\,J[2] + 86\,H[4]\times J[2] + 96\,H[2]\,J[2]^2 - 744\,J[2]^3 + 48\,H[2]\times L[4] - 744\,J[2]\times L[4] + 342\,H[4]\,q_1 - 8136\,J[2]^2\,q_1 - 3456\,L[4]\,q_1 - 24\,300\,J[2]\,q_1^2 + 24\,300\,q_1^3\Big),$$

$$L[6] \to \frac{1}{81}\Big(-19\,H[2]^3 + 38\,H[2]\times H[4] + 23\,H[2]^2\,J[2] + 149\,H[4]\times J[2] - 174\,H[2]\,J[2]^2 - 204\,J[2]^3 + 48\,H[2]\times L[4] - 852\,J[2]\times L[4] + 261\,H[4]\,q_1 - 7488\,J[2]^2\,q_1 - 1998\,L[4]\,q_1 - 12\,150\,J[2]\,q_1^2 + 12\,150\,q_1^3\Big),$$

$$M[6] \to \frac{1}{324}\Big(H[2]^3 - 2\,H[2]\times H[4] - 14\,H[2]^2\,J[2] + 22\,H[4]\times J[2] + 120\,H[2]\,J[2]^2 - 552\,J[2]^3 + 6\,H[2]\times L[4] + 420\,J[2]\times L[4] - 18\,H[4]\,q_1 + 5544\,J[2]^2\,q_1 - 2376\,L[4]\,q_1 - 24\,300\,J[2]\,q_1^2 + 24\,300\,q_1^3\Big),$$

$$H[8] \to \frac{1}{1458}\Big(70\,H[2]^4 - 361\,H[2]^2\,H[4] + 442\,H[4]^2 + 149\,H[2]^3\,J[2] + 674\,H[2]\times H[4]\times J[2] + 830\,H[2]^2\,J[2]^2 - 7216\,H[4]\,J[2]^2 - 8868\,H[2]\,J[2]^3 + 30\,792\,J[2]^4 + 30\,H[2]^2\,L[4] + 3828\,H[4]\times L[4] + 408\,H[2]\times J[2]\times L[4] - 3624\,J[2]^2\,L[4] - 11\,088\,L[4]^2 + 5796\,H[4]\times J[2]\,q_1 + 158\,832\,J[2]^3\,q_1 - 206\,928\,J[2]\times L[4]\,q_1 - 40\,500\,H[4]\,q_1^2 - 356\,400\,J[2]^2\,q_1^2 + 121\,500\,L[4]\,q_1^2 + 1\,895\,400\,J[2]\,q_1^3\Big),$$

$$H[10] \to \frac{1}{2187}\Big(70\,H[2]^5 - 879\,H[2]^3\,H[4] + 1478\,H[2]\,H[4]^2 - 1161\,H[2]^4\,J[2] + 2866\,H[2]^2\,H[4]\times J[2] + 2506\,H[4]^2\,J[2] + 110\,H[2]^3\,J[2]^2 + 12\,212\,H[2]\times H[4]\,J[2]^2 - 159\,160\,H[2]^2\,J[2]^3 + 230\,888\,H[4]\,J[2]^3 + 607\,632\,H[2]\,J[2]^4 - 1\,226\,976\,J[2]^5 - 630\,H[2]^3\,L[4] + 2040\,H[2]\times H[4]\times L[4] + 204\,H[2]^2\,J[2]\times L[4] - 26\,160\,H[4]\times J[2]\times L[4] - 125\,412\,H[2]\,J[2]^2\,L[4] - 423\,096\,J[2]^3\,L[4] + 2448\,H[2]\,L[4]^2 - 35\,928\,J[2]\,L[4]^2 + 10\,782\,H[4]^2\,q_1 - 95\,256\,H[4]\,J[2]^2\,q_1 - 9\,237\,600\,J[2]^4\,q_1 - 71\,496\,H[4]\times L[4]\,q_1 - 505\,440\,J[2]^2\,L[4]\,q_1 - 35\,640\,L[4]^2\,q_1 - 251\,100\,H[4]\times J[2]\,q_1^2 - 4\,017\,600\,J[2]^3\,q_1^2 - 777\,600\,J[2]\times L[4]\,q_1^2 + 510\,300\,H[4]\,q_1^3 - 5\,832\,000\,J[2]^2\,q_1^3\Big)\Big\}$$

In[159]:=

```
FormatAsEquations = Table[Normalizations[[i]] × SecondaryInvariants[[i]] → Collect[
    (Normalizations[[i]] × SecondaryInvariants[[i]] /. SolveNeccSuff), β3], {i, 1, 10}]
```

Out[159]=

$$\{2H[4] \to H[2]^2 - 3H[2] \times J[2] + 6J[2]^2 + 6L[4] - 9H[2]\,q_1 + 90J[2]\,q_1,$$
$$18J[4] \to -H[2]^2 + 2H[4] + 12H[2] \times J[2] - 24J[2]^2 + 12L[4],$$
$$9K[4] \to -2H[2]^2 + 4H[4] + 15H[2] \times J[2] - 66J[2]^2 + 6L[4],$$
$$27H[6] \to -13H[2]^3 + 26H[2] \times H[4] - 7H[2]^2J[2] + 146H[4] \times J[2] - 156H[2]J[2]^2 + 372J[2]^3 + 30H[2] \times L[4] - 924J[2] \times L[4] + 234H[4]\,q_1 - 7272J[2]^2q_1 - 1512L[4]\,q_1 - 8100J[2]\,q_1^2 + 8100q_1^3,$$
$$36J[6] \to -H[2]^3 + 2H[2] \times H[4] + 6H[4] \times J[2] + 12H[2]J[2]^2 - 72J[2]^3 + 12H[2] \times L[4] + 18H[4]\,q_1 - 144J[2]^2q_1 - 324L[4]\,q_1 - 2700J[2]\,q_1^2 + 2700q_1^3,$$
$$162K[6] \to -19H[2]^3 + 38H[2] \times H[4] + 14H[2]^2J[2] + 86H[4] \times J[2] + 96H[2]J[2]^2 - 744J[2]^3 + 48H[2] \times L[4] - 744J[2] \times L[4] + 342H[4]\,q_1 - 8136J[2]^2q_1 - 3456L[4]\,q_1 - 24\,300J[2]\,q_1^2 + 24\,300q_1^3,$$
$$81L[6] \to -19H[2]^3 + 38H[2] \times H[4] + 23H[2]^2J[2] + 149H[4] \times J[2] - 174H[2]J[2]^2 - 204J[2]^3 + 48H[2] \times L[4] - 852J[2] \times L[4] + 261H[4]\,q_1 - 7488J[2]^2q_1 - 1998L[4]\,q_1 - 12\,150J[2]\,q_1^2 + 12\,150q_1^3,$$
$$324M[6] \to H[2]^3 - 2H[2] \times H[4] - 14H[2]^2J[2] + 22H[4] \times J[2] + 120H[2]J[2]^2 - 552J[2]^3 + 6H[2] \times L[4] + 420J[2] \times L[4] - 18H[4]\,q_1 + 5544J[2]^2q_1 - 2376L[4]\,q_1 - 24\,300J[2]\,q_1^2 + 24\,300q_1^3,$$
$$1458H[8] \to 70H[2]^4 - 361H[2]^2H[4] + 442H[4]^2 + 149H[2]^3J[2] + 674H[2] \times H[4] \times J[2] + 830H[2]^2J[2]^2 - 7216H[4]J[2]^2 - 8868H[2]J[2]^3 + 30\,792J[2]^4 + 30H[2]^2L[4] + 3828H[4] \times L[4] + 408H[2] \times J[2] \times L[4] - 3624J[2]^2L[4] - 11\,088L[4]^2 + 5796H[4] \times J[2]\,q_1 + 158\,832J[2]^3q_1 - 206\,928J[2] \times L[4]\,q_1 - 40\,500H[4]\,q_1^2 - 356\,400J[2]^2q_1^2 + 121\,500L[4]\,q_1^2 + 1\,895\,400J[2]\,q_1^3,$$
$$2187H[10] \to 70H[2]^5 - 879H[2]^3H[4] + 1478H[2]H[4]^2 - 1161H[2]^4J[2] + 2866H[2]^2H[4] \times J[2] + 2506H[4]^2J[2] + 110H[2]^3J[2]^2 + 12\,212H[2] \times H[4]J[2]^2 - 159\,160H[2]^2J[2]^3 + 230\,888H[4]J[2]^3 + 607\,632H[2]J[2]^4 - 1\,226\,976J[2]^5 - 630H[2]^3L[4] + 2040H[2] \times H[4] \times L[4] + 204H[2]^2J[2] \times L[4] - 26\,160H[4] \times J[2] \times L[4] - 125\,412H[2]J[2]^2L[4] - 423\,096J[2]^3L[4] + 2448H[2]L[4]^2 - 35\,928J[2]L[4]^2 + 10\,782H[4]^2q_1 - 95\,256H[4]J[2]^2q_1 - 9\,237\,600J[2]^4q_1 - 71\,496H[4] \times L[4]\,q_1 - 505\,440J[2]^2L[4]\,q_1 - 35\,640L[4]^2q_1 - 251\,100H[4] \times J[2]\,q_1^2 - 4\,017\,600J[2]^3q_1^2 - 777\,600J[2] \times L[4]\,q_1^2 + 510\,300H[4]\,q_1^3 - 5\,832\,000J[2]^2q_1^3\}$$