Mathematica Computations

Sunder Sethuraman and Shankar Venkataramani

This is the accompanying Mathematica notebook for the computations supporting the paper “**COUPLED KPZ EQUATIONS AND THEIR DECOUPLEABILITY**” by Fu, Funaki, Sethuraman, and Venkataramani.

Table of contents:

Section 2.2: ODE analysis of decoupleability for n = 2.

```
In[1]:= n = 2;
        vars = Table[x[i], {i, 1, n}];
        ϵ = Table[0, {i, 1, n}, {j, 1, n}];
        ϵ〚1, 2〛 = -1; ϵ〚2, 1〛 = 1;
```

We begin by defining the variables and constructing ϵ, the permutation matrix/generator for rotations in the plane, as it is used in the Proof of Prop. 2.5.

```
In[5]:= f = a3 x[1]^3 + 3 a2 x[1]^2 × x[2] + 3 a1 x[1] x[2]^2 + a0 x[2]^3
```

$$a_3\, x[1]^3 + 3\, a_2\, x[1]^2\, x[2] + 3\, a_1\, x[1]\, x[2]^2 + a_0\, x[2]^3$$

```
In[6]:= Γ = Table[Simplify[D[f / 6, x[i], x[j], x[k]]], {i, 1, n}, {j, 1, n}, {k, 1, n}];

In[7]:= dθΓ = Table[Sum[Γ〚m, j, k〛 ϵ〚i, m〛 + Γ〚i, m, k〛 ϵ〚j, m〛 + Γ〚i, j, m〛 ϵ〚k, m〛, {m, 1, n}],
          {i, 1, n}, {j, 1, n}, {k, 1, n}]
```

$$\{\{\{-3\, a_2, -2\, a_1 + a_3\}, \{-2\, a_1 + a_3, -a_0 + 2\, a_2\}\}, \{\{-2\, a_1 + a_3, -a_0 + 2\, a_2\}, \{-a_0 + 2\, a_2, 3\, a_1\}\}\}$$

We compute the derivative of the evolution of the tensor under rotations.

```
In[8]:= dΓvec = {dθΓ〚2, 2, 2〛, dθΓ〚2, 2, 1〛, dθΓ〚2, 1, 1〛, dθΓ〚1, 1, 1〛}
```

$$\{3\, a_1, -a_0 + 2\, a_2, -2\, a_1 + a_3, -3\, a_2\}$$

The tensor and its derivative are represented as 4 by 1 column vectors.

```
In[9]:= ℒ = Grad[dΓvec, {a0, a1, a2, a3}]
```

$$\{\{0, 3, 0, 0\}, \{-1, 0, 2, 0\}, \{0, -2, 0, 1\}, \{0, 0, -3, 0\}\}$$

Matrix representation of the rotation generator.

In[10]:=
```
MatrixForm[ℒ]
```

Out[•]//MatrixForm=

$$\begin{pmatrix} 0 & 3 & 0 & 0 \\ -1 & 0 & 2 & 0 \\ 0 & -2 & 0 & 1 \\ 0 & 0 & -3 & 0 \end{pmatrix}$$

In[11]:=
```
{vals, vecs} = Eigensystem[ℒ]
```

Out[•]= {{3 i, -3 i, i, -i}, {{i, -1, -i, 1}, {-i, -1, i, 1}, {-3 i, 1, -i, 3}, {3 i, 1, i, 3}}}

In[12]:=
```
Λ = DiagonalMatrix[vals]
```

Out[•]= {{3 i, 0, 0, 0}, {0, -3 i, 0, 0}, {0, 0, i, 0}, {0, 0, 0, -i}}

If we treat vecs as a matrix instead of a list of vectors, each eigenvector will be treated as a row. To make them columns, as appropriate for a right eigenvector, we need to take a transpose.

In[13]:=
```
ℒ.Transpose[vecs] - Transpose[vecs].Λ
```

Out[•]= {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}

Transpose[vecs] gives the right eigenvectors of $\mathcal{L}$. To get the Left eigenvectors as rows, we need to invert Transpose[vecs]. Below, we include an additional normalization to clear denominators.

In[14]:=
```
Lvecs = Sqrt[Det[vecs]] × Inverse[Transpose[vecs]]
```

Out[•]= {{1, -3 i, -3, i}, {-1, -3 i, 3, i}, {-1, i, -1, i}, {1, i, 1, i}}

These are the left Eigenvectors of $\mathcal{L}$ as can be checked:

In[15]:=
```
Lvecs.ℒ - Λ.Lvecs
```

Out[•]= {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}

In[16]:=
```
MatrixForm[Lvecs]
```

Out[•]//MatrixForm=

$$\begin{pmatrix} 1 & -3\,i & -3 & i \\ -1 & -3\,i & 3 & i \\ -1 & i & -1 & i \\ 1 & i & 1 & i \end{pmatrix}$$

This is the matrix E in Sec. 2.2 and the corresponding eigenvalues are λ_1=3i, λ_2=-3i, λ_3=i, λ_4=-i.

For a decoupled tensor (β_1, 0, 0, β_2) we get v_i= β_1+ i β_2 = v_4. This gives Exp[4 i θ] = $v_4(0)$ / $v_1(0)$ = $\frac{(a_0+a_2)+i\,(a_1+a_3)}{(a_0-3\,a_2)+i\,(-3\,a_1+a_3}$. Subject to the necessary condition $(a_0 + a_2)^2 + (a_1 + a_3)^2 = (a_0 - 3\,a_2)^2 + (-3\,a_1 + a_3)^2$, we get 4 solutions for θ, say φ, $\varphi+\pi/2$, $\varphi+\pi$ and $\varphi+3\pi/2$.

These solutions define β_1 and β_2 by β_1+ i β_2 = Exp[-iφ] $((a_0 + a_2) + i\,(a_1 + a_3))$. The rotations $\varphi+\pi/2$, $\varphi+\pi$ and $\varphi+3\pi/2$ then correspond to the fully decoupled tensors (β_2, 0, 0, -β_1), (−β_1, 0, 0, -β_2) and (−β_2, 0, 0, β_1) respectively.

Section 4

We begin to compute $\sigma_\theta \circ \Gamma$, the action of a rotation on the space of tensors $\mathcal{T}_2$.

In[17]:=
```
Clear[Γ];
n = 2;
vars = Table[x[i], {i, 1, n}];
Γlist = Flatten[Table[Γ[i, j, k], {i, 1, n}, {j, i, n}, {k, j, n}]];
rename = Table[Γlist⟦m⟧ → Subscript[a, 4 - m], {m, 1, 4}];
f = (Sum[Γ @@ Sort[{i, j, k}] × x[i] × x[j] × x[k], {i, 1, n}, {j, 1, n}, {k, 1, n}] /. rename)
```

Out[∘]= $a_3\, x[1]^3 + 3\, a_2\, x[1]^2\, x[2] + 3\, a_1\, x[1]\, x[2]^2 + a_0\, x[2]^3$

The last line is the expression for the cubic polynomial associate to a tensor Γ. Note that the coordinates are x[1] and x[2] where the indices are arguments and not subscripts. We begin by defining a two dimensional rotation matrix $\sigma[\theta]$.

In[23]:=
```
Clear[σ];
σ[θ_] = {{Cos[θ], - Sin[θ]}, {Sin[θ], Cos[θ]}};
MatrixForm[σ[θ]]
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} \mathrm{Cos}[\theta] & -\mathrm{Sin}[\theta] \\ \mathrm{Sin}[\theta] & \mathrm{Cos}[\theta] \end{pmatrix}$$

Multiplying by this rotation matrix on the left gives the action of SO(2) on $\mathbb{R}^2$ where the elements are thought of a column vectors. The action corresponds to rotating `counter-clockwise' by an angle θ.

The action on tensors, or equivalently on polynomials, is given by $\sigma \circ f(z) = f\left(\sigma^{-1} \cdot z\right)$ for $z \in \mathbb{R}^2$.

In[26]:=
```
Substitution = Table[x[i] → (σ[-θ].{z[1], z[2]})⟦i⟧, {i, 1, n}]
```

Out[∘]= $\{x[1] \to \mathrm{Cos}[\theta] \times z[1] + \mathrm{Sin}[\theta] \times z[2],\ x[2] \to -\mathrm{Sin}[\theta] \times z[1] + \mathrm{Cos}[\theta] \times z[2]\}$

In[27]:=
```
Transformedf = f //. Substitution
```

Out[∘]=

$$\begin{aligned}
&a_0\, (-\mathrm{Sin}[\theta] \times z[1] + \mathrm{Cos}[\theta] \times z[2])^3 + \\
&\ 3\, a_1\, (-\mathrm{Sin}[\theta] \times z[1] + \mathrm{Cos}[\theta] \times z[2])^2\, (\mathrm{Cos}[\theta] \times z[1] + \mathrm{Sin}[\theta] \times z[2]) + \\
&\ 3\, a_2\, (-\mathrm{Sin}[\theta] \times z[1] + \mathrm{Cos}[\theta] \times z[2])\, (\mathrm{Cos}[\theta] \times z[1] + \mathrm{Sin}[\theta] \times z[2])^2 + \\
&\ a_3\, (\mathrm{Cos}[\theta] \times z[1] + \mathrm{Sin}[\theta] \times z[2])^3
\end{aligned}$$

This is a cubic polynomial in z[1],z[2]. We can now read off the transformations of the coefficients from $\sigma \circ f(z) = b_3\, z[1]^3 + 3\, b_2\, z[1]^2\, z[2] + 3\, b_1\, z[1]\, z[2]^2 + b_0\, z[2]^3$. We now account for the factors of 3 in the coefficients b[1] and b[2] and order the coefficients as a column vector from b_0 to b_3.

In[28]:=
```
newcoeffs = Simplify[
  DiagonalMatrix[{1, 1 / 3, 1 / 3, 1}].CoefficientList[Transformedf /. {z[2] → 1}, z[1]]]
```

Out[•]=
$$\Big\{\text{Cos}[\theta]^3\, a_0 + \text{Sin}[\theta]\,\left(3\,\text{Cos}[\theta]^2\, a_1 + \text{Sin}[\theta]\,(3\,\text{Cos}[\theta]\, a_2 + \text{Sin}[\theta]\, a_3)\right),$$
$$\frac{1}{4}\left(-4\,\text{Cos}[\theta]^2\,\text{Sin}[\theta]\, a_0 + (\text{Cos}[\theta] + 3\,\text{Cos}[3\,\theta])\, a_1 + 2\,\text{Sin}[\theta]\,(a_2 + 3\,\text{Cos}[2\,\theta]\, a_2 + \text{Sin}[2\,\theta]\, a_3)\right),$$
$$\text{Cos}[\theta]\,\text{Sin}[\theta]^2\, a_0 + \frac{1}{4}\left((\text{Sin}[\theta] - 3\,\text{Sin}[3\,\theta])\, a_1 + 2\,\text{Cos}[\theta]\,((-1 + 3\,\text{Cos}[2\,\theta])\, a_2 + \text{Sin}[2\,\theta]\, a_3)\right),$$
$$-\text{Sin}[\theta]^3\, a_0 + \text{Cos}[\theta]\,\left(3\,\text{Sin}[\theta]^2\, a_1 + \text{Cos}[\theta]\,(-3\,\text{Sin}[\theta]\, a_2 + \text{Cos}[\theta]\, a_3)\right)\Big\}$$

In[29]:=
```
Lσ = Grad[newcoeffs, Table[a_{i-1}, {i, 1, 4}]]
```

Out[•]=
$$\Big\{\left\{\text{Cos}[\theta]^3,\ 3\,\text{Cos}[\theta]^2\,\text{Sin}[\theta],\ 3\,\text{Cos}[\theta]\,\text{Sin}[\theta]^2,\ \text{Sin}[\theta]^3\right\},$$
$$\left\{-\text{Cos}[\theta]^2\,\text{Sin}[\theta],\ \frac{1}{4}(\text{Cos}[\theta] + 3\,\text{Cos}[3\,\theta]),\ \frac{1}{2}(1 + 3\,\text{Cos}[2\,\theta])\,\text{Sin}[\theta],\ \frac{1}{2}\text{Sin}[\theta]\times\text{Sin}[2\,\theta]\right\},$$
$$\left\{\text{Cos}[\theta]\,\text{Sin}[\theta]^2,\ \frac{1}{4}(\text{Sin}[\theta] - 3\,\text{Sin}[3\,\theta]),\ \frac{1}{2}\text{Cos}[\theta]\,(-1 + 3\,\text{Cos}[2\,\theta]),\ \frac{1}{2}\text{Cos}[\theta]\times\text{Sin}[2\,\theta]\right\},$$
$$\left\{-\text{Sin}[\theta]^3,\ 3\,\text{Cos}[\theta]\,\text{Sin}[\theta]^2,\ -3\,\text{Cos}[\theta]^2\,\text{Sin}[\theta],\ \text{Cos}[\theta]^3\right\}\Big\}$$

In[30]:=
```
MatrixForm[Lσ]
```

Out[•]//MatrixForm=
$$\begin{pmatrix} \text{Cos}[\theta]^3 & 3\,\text{Cos}[\theta]^2\,\text{Sin}[\theta] & 3\,\text{Cos}[\theta]\,\text{Sin}[\theta]^2 & \text{Sin}[\theta]^3 \\ -\text{Cos}[\theta]^2\,\text{Sin}[\theta] & \frac{1}{4}(\text{Cos}[\theta] + 3\,\text{Cos}[3\,\theta]) & \frac{1}{2}(1 + 3\,\text{Cos}[2\,\theta])\,\text{Sin}[\theta] & \frac{1}{2}\text{Sin}[\theta]\times\text{Sin}[2\,\theta] \\ \text{Cos}[\theta]\,\text{Sin}[\theta]^2 & \frac{1}{4}(\text{Sin}[\theta] - 3\,\text{Sin}[3\,\theta]) & \frac{1}{2}\text{Cos}[\theta]\,(-1 + 3\,\text{Cos}[2\,\theta]) & \frac{1}{2}\text{Cos}[\theta]\times\text{Sin}[2\,\theta] \\ -\text{Sin}[\theta]^3 & 3\,\text{Cos}[\theta]\,\text{Sin}[\theta]^2 & -3\,\text{Cos}[\theta]^2\,\text{Sin}[\theta] & \text{Cos}[\theta]^3 \end{pmatrix}$$

This is the representation of SO(2) on the space of 2 × 2 × 2 symmetric tensors. We can also compute the generator for this action.

In[31]:=
```
ℒ = D[Lσ, θ] /. {θ → 0}
```

Out[•]= {{0, 3, 0, 0}, {-1, 0, 2, 0}, {0, -2, 0, 1}, {0, 0, -3, 0}}

In[32]:=
```
MatrixForm[ℒ]
```

Out[•]//MatrixForm=
$$\begin{pmatrix} 0 & 3 & 0 & 0 \\ -1 & 0 & 2 & 0 \\ 0 & -2 & 0 & 1 \\ 0 & 0 & -3 & 0 \end{pmatrix}$$

This is an alternate derivation to obtain the generator $\mathcal{L}$ in the proof of Prop. 2.5.

Molien’s formula and Hilbert series for the SO(2) invariants.

In[33]:=
```
Simplify[1 / (2 π) Integrate[1 / Simplify[Det[IdentityMatrix[4] - λ Lσ]], {θ, 0, 2 π}]]
```

Out[•]=
$$\begin{cases} \frac{1+\lambda^4}{(-1+\lambda^2)^3 (1+\lambda^2)} & \mathrm{Abs}[\lambda] > 1 \\ \frac{2+\lambda^{2/3} (-1-\lambda^{2/3}+\lambda^{4/3}) (1+\lambda^{2/3}+\lambda^{4/3}+2\lambda^2)}{3 (-1+\lambda^2)^3 (1+\lambda^2)} & \frac{1}{\mathrm{Abs}[\lambda]^{1/3}} < 1 \text{ if } \mathrm{Abs}[\lambda]^{1/3} \neq 1 \\ -\frac{1+\lambda^4}{(-1+\lambda^2)^3 (1+\lambda^2)} & \mathrm{True} \end{cases}$$

The result for Abs[λ] < 1 corresponds to Molien’s formula.

In[34]:=
$$\Phi SO_2 = \mathrm{Simplify}\left[-\frac{1+\lambda^4}{(-1+\lambda^2)^3 (1+\lambda^2)}\right]$$

Out[•]=
$$-\frac{1+\lambda^4}{(-1+\lambda^2)^3 (1+\lambda^2)}$$

Molien’s formula and Hilbert series for the O(2) invariants.

We now compute the action of O(2) by incorporating a reflection operator corresponding to x[1]-> x[1], x[2]->-x[2]. In terms of the tensor coefficients, this action is given by the matrix:

In[35]:=
```
N = {{0, 0, 0, 1}, {0, 0, 1, 0}, {0, 1, 0, 0}, {1, 0, 0, 0}};
```

In[36]:=
```
Simplify[Det[IdentityMatrix[4] - λ N.Lσ]]
```

Out[•]=
$$(-1+\lambda^2)^2$$

This determinant does not depend explicitly on θ. Therefore the integral of the reciprocal is as follows.

In[37]:=
```
ΦO2 = Simplify[ (1 / Det[IdentityMatrix[4] - λ N.Lσ] + ΦSO2) / 2 ]
```

Out[•]=
$$-\frac{1}{(-1+\lambda^2)^3 (1+\lambda^2)}$$

Computation of the invariants when n=2

In[38]:=
```
u = Simplify[ Grad[Laplacian[f, vars], vars] / 6]
```

Out[•]=
$$\{a_1 + a_3, a_0 + a_2\}$$

This is the trace vector.

In[39]:=
```
u.u
```

Out[•]=
$$(a_0 + a_2)^2 + (a_1 + a_3)^2$$

We will denote this O(2) invariant by j_2. We can ‘lift’ this vector u to form the cubic polynomial f_1.

In[40]:=
```
f1 = 3 u.vars (vars.vars) / (n + 2)
```

Out[•]= $\frac{3}{4}\left((a_1+a_3)\,x[1]+(a_0+a_2)\,x[2]\right)\left(x[1]^2+x[2]^2\right)$

In[41]:=
```
ℬ = Table[D[f1 / 6, x[i], x[j], x[k]], {i, 1, n}, {j, 1, n}, {k, 1, n}];
```

In[42]:=
```
Table[MatrixForm[ℬ〚i〛], {i, 1, n}]
```

Out[•]= $\left\{\begin{pmatrix}\frac{3}{4}(a_1+a_3) & \frac{1}{4}(a_0+a_2)\\ \frac{1}{4}(a_0+a_2) & \frac{1}{4}(a_1+a_3)\end{pmatrix},\ \begin{pmatrix}\frac{1}{4}(a_0+a_2) & \frac{1}{4}(a_1+a_3)\\ \frac{1}{4}(a_1+a_3) & \frac{3}{4}(a_0+a_2)\end{pmatrix}\right\}$

$\mathcal{B}$ is the tensor corresponding to the trace part. With our normalization, the trace-free part is given by $f_3 = (n+2)\,f - f_1$.

In[43]:=
```
f3 = Collect[Expand[(n + 2) (f - f1)], vars]
```

Out[•]= $(-3a_1+a_3)\,x[1]^3+(-3a_0+9a_2)\,x[1]^2\,x[2]+(9a_1-3a_3)\,x[1]\,x[2]^2+(a_0-3a_2)\,x[2]^3$

f_3 corresponds to a trace-free $2\times2\times2$ symmetric tensor $\mathcal{D}$. To eliminate denominators, we multiply by a factor of (n+2), which equals 4 in the case n=2.

In[44]:=
```
𝒟 = Table[Simplify[D[f3 / 6, x[i], x[j], x[k]]], {i, 1, n}, {j, 1, n}, {k, 1, n}];
```

In[45]:=
```
Table[MatrixForm[𝒟〚i〛], {i, 1, n}]
```

Out[•]= $\left\{\begin{pmatrix}-3a_1+a_3 & -a_0+3a_2\\ -a_0+3a_2 & 3a_1-a_3\end{pmatrix},\ \begin{pmatrix}-a_0+3a_2 & 3a_1-a_3\\ 3a_1-a_3 & a_0-3a_2\end{pmatrix}\right\}$

In[46]:=
```
Dstarsqrd = Simplify[TensorContract[𝒟⊗𝒟, {{1, 4}, {2, 5}}]]
```

Out[•]= $\left\{\left\{2\left((a_0-3a_2)^2+(-3a_1+a_3)^2\right),\ 0\right\},\ \left\{0,\ 2\left((a_0-3a_2)^2+(-3a_1+a_3)^2\right)\right\}\right\}$

This is $\mathcal{D}^{*2}$.

In[47]:=
```
w = Expand[𝒟.u.u]
```

Out[•]= $\{a_0^2a_1-3a_1^3+10a_0a_1a_2+9a_1a_2^2-3a_0^2a_3-5a_1^2a_3+2a_0a_2a_3+5a_2^2a_3-a_1a_3^2+a_3^3,$
$a_0^3+5a_0a_1^2-a_0^2a_2+9a_1^2a_2-5a_0a_2^2-3a_2^3+2a_0a_1a_3+10a_1a_2a_3-3a_0a_3^2+a_2a_3^2\}$

This is the vector w defined right after (4.4).

In[48]:=
```
Tr[Dstarsqrd]
```

Out[•]= $4\left((a_0-3a_2)^2+(-3a_1+a_3)^2\right)$

This is another O(2) invariant which we denote by h_2.

In[49]:=
```
Simplify[u.w]
```

Out[•]= $a_0^4-3a_1^4-3a_2^4-8a_1^3a_3+24a_1a_2^2a_3+6a_2^2a_3^2+a_3^4-$
$8a_0a_2\left(-3a_1^2+a_2^2-3a_1a_3\right)+6a_0^2\left(a_1^2-a_2^2-a_3^2\right)+6a_1^2\left(3a_2^2-a_3^2\right)$

This defines the O(2) invariant l_4.

In[50]:=
```
Simplify[Det[{u, w}]]
```

Out[]=
$$4\left(-3 a_0^2 a_1 a_2 + a_0^3 a_3 + a_2\left(3 a_1^3 + 6 a_1^2 a_3 - 2 a_2^2 a_3 - 3 a_1\left(a_2^2 - a_3^2\right)\right) + a_0\left(2 a_1^3 - 6 a_1 a_2^2 + 3 a_1^2 a_3 - a_3\left(3 a_2^2 + a_3^2\right)\right)\right)$$

This defines the invariant m_4.

We can now write down the definitions of the SO(2) invariants and the ideal generated by these definitions, using the labels j_2, h_2, l_4 and m_4 as slack variables.

In[51]:=
```
Clear[j, h, l, m]
Invariants = {j2 - u.u, h2 - Tr[Dstarsqrd], l4 - w.u, m4 - Det[{u, w}]}
```

Out[]=
$$\left\{-(a_0 + a_2)^2 - (a_1 + a_3)^2 + j_2,\ -4\left((a_0 - 3 a_2)^2 + (-3 a_1 + a_3)^2\right) + h_2,\right.$$
$$-\left((a_0 + a_2)\left(a_0^3 + 5 a_0 a_1^2 - a_0^2 a_2 + 9 a_1^2 a_2 - 5 a_0 a_2^2 - 3 a_2^3 + 2 a_0 a_1 a_3 + 10 a_1 a_2 a_3 - 3 a_0 a_3^2 + a_2 a_3^2\right)\right) -$$
$$(a_1 + a_3)\left(a_0^2 a_1 - 3 a_1^3 + 10 a_0 a_1 a_2 + 9 a_1 a_2^2 - 3 a_0^2 a_3 - 5 a_1^2 a_3 + 2 a_0 a_2 a_3 + 5 a_2^2 a_3 - a_1 a_3^2 + a_3^3\right) + l_4,$$
$$-8 a_0 a_1^3 + 12 a_0^2 a_1 a_2 - 12 a_1^3 a_2 + 24 a_0 a_1 a_2^2 + 12 a_1 a_2^3 - 4 a_0^3 a_3 - 12 a_0 a_1^2 a_3 -$$
$$\left.24 a_1^2 a_2 a_3 + 12 a_0 a_2^2 a_3 + 8 a_2^3 a_3 - 12 a_1 a_2 a_3^2 + 4 a_0 a_3^3 + m_4\right\}$$

This is the basis for the ideal of polynomials on $\mathbb{R}^8$, corresponding to the 4 coefficients a_0,a_1,a_2,a_3 and the 4 invariants j_2,h_2,l_4,m_4

We seek potential relations among the invariants by finding a basis for the ideal generated by the definitions of the invariants intersected with the polynomials in j_2,h_2,l_4,m_4 that do not depend on a_0,a_1,a_2,a_3, that is, we are eliminating the coefficients between the relations defining the invariants.

In[53]:=
```
GroebnerBasis[Invariants, {m4, l4, h2, j2}, {a0, a1, a2, a3}, MonomialOrder → EliminationOrder]
```

Out[]=
$$\left\{h_2 j_2^3 - 4 l_4^2 - 4 m_4^2\right\}$$

We see that there is one identity that allows us to replace $m_4{}^2$ by an expression in the other invariants. There are no further relations, so this implies l_4, h_2 and j_2 are algebraically independent.

Alternate choices for the fundamental invariants are the trace and determinant of Γ^{*^2}, which are themselves O(2) invariants.

In[54]:=
```
Γ = Table[D[f / 6, x[i], x[j], x[k]], {i, 1, n}, {j, 1, n}, {k, 1, n}];
Q = Simplify[TensorContract[Γ⊗Γ, {{1, 4}, {2, 5}}]];
MatrixForm[Q]
```

Out[]//MatrixForm=
$$\begin{pmatrix} a_1^2 + 2 a_2^2 + a_3^2 & a_0 a_1 + a_2 (2 a_1 + a_3) \\ a_0 a_1 + a_2 (2 a_1 + a_3) & a_0^2 + 2 a_1^2 + a_2^2 \end{pmatrix}$$

This is the matrix Γ^{*2}.

In[56]:=
```
NewInvariants = {τ - Tr[Q], δ - Det[Q]}
```

Out[]=
$$\left\{\tau - a_0^2 - 3 a_1^2 - 3 a_2^2 - a_3^2,\right.$$
$$\left.\delta - 2 a_1^4 + 4 a_0 a_1^2 a_2 - 2 a_0^2 a_2^2 - a_1^2 a_2^2 - 2 a_2^4 + 2 a_0 a_1 a_2 a_3 + 4 a_1 a_2^2 a_3 - a_0^2 a_3^2 - 2 a_1^2 a_3^2\right\}$$

These are the trace τ and the determinant δ of Γ^{*2}, used in the general framework for fully decoupled tensors in Sec. 6 when n=2. We now express τ and δ in terms of the fundamental invariants j_2, h_2 and l_4, the O(2) invariants found earlier.

In[57]:=
```
GroebnerBasis[Join[NewInvariants, Invariants],
 {τ, δ, l4, h2, j2}, {m4, a0, a1, a2, a3}, MonomialOrder → EliminationOrder]
```

Out[•]= $\{16\,\tau - h_2 - 12\,j_2,\ -1024\,\delta + h_2^2 + 8\,h_2\,j_2 + 80\,j_2^2 - 128\,l_4\}$

These are the relations at the end of section 4 : $\mathrm{Tr}[\Gamma^{*2}] = \frac{h_2+12 j_2}{16}$, $\mathrm{Det}[\Gamma^{*2}] = \frac{h_2^2+8 h_2 j_2+80 j_2^2-128 l_4}{1024}$.

Section 7.2

Fully decoupleable 3 × 3 × 3 tensors

In[58]:=
```
n = 3; vars = Table[x[i], {i, 1, n}]; f = Sum[β_i x[i]^3, {i, 1, n}]
```

Out[•]= $\beta_1\,x[1]^3 + \beta_2\,x[2]^3 + \beta_3\,x[3]^3$

This is the cubic polynomial corresponding to a fully decoupleable tensor when n=3. We now compute the tracial and trace-free parts of Γ.

In[59]:=
```
Γ = Simplify[Table[D[f, x[i], x[j], x[k]], {k, 1, n}, {j, 1, n}, {i, 1, n}] / 6];
u = Simplify[Table[D[Laplacian[f, vars], x[i]], {i, 1, n}] / 6];
f3 = (n + 2) f - 3 (u.vars) (vars.vars)
```

Out[•]= $-3\,(\beta_1\,x[1] + \beta_2\,x[2] + \beta_3\,x[3])\,\left(x[1]^2 + x[2]^2 + x[3]^2\right) + 5\,\left(\beta_1\,x[1]^3 + \beta_2\,x[2]^3 + \beta_3\,x[3]^3\right)$

This is the trace-free part.

In[62]:=
```
𝒟 = Simplify[Table[D[f3, x[i], x[j], x[k]], {k, 1, n}, {j, 1, n}, {i, 1, n}] / 6];
v = Simplify[TensorContract[(𝒟⊗𝒟)⊗𝒟, {{1, 4}, {2, 5}, {3, 7}, {6, 8}}]];
w = Simplify[𝒟.u.u];
```

We have now computed the vectors u,v and w and the trace-free tensor $\mathcal{D}$, which are the ingredients needed to compute the Integrity basis given by Olive and Auffray as follows.

In[65]:=
```
Clear[H, J, K, L, M, Q];
Q = Simplify[TensorContract[𝒟⊗𝒟, {{1, 4}, {2, 5}}]];
γuu = w;
Coeffs = Table[β_i, {i, 1, n}];
Trivialize = Table[Coeffs[[j]] → 0, {j, 1, Length[Coeffs]}];
IntegrityPolys = {H[2] - Simplify[Tr[Q]], H[4] - Simplify[Tr[Q.Q]],
   J[2] - Simplify[u.u], L[4] - Simplify[γuu.u], H[6] - Simplify[v.v],
   H[10] - Simplify[𝒟.v.v.v] , J[4] - Simplify[u.Q.u], K[4] - Simplify[Tr[Q.(𝒟.u)]],
   J[6] - Simplify[(u.Q).γuu], K[6] - Simplify[v.w], L[6] - Simplify[(u.Q).v],
   M[6] - Simplify[γuu.γuu], H[8] - Simplify[(u.Q).(Q.v)]};
OAInvariants = IntegrityPolys /. Trivialize;
```

The Ideal IntegrityPolys is generated by the polynomials defining the Integrity basis elements in terms of the coefficients of a fully decoupled tensor, with diagonal entries β_1, β_2 and β_3, and with the labels of the Integrity invariants as the slack variables.

In[71]:=
```
IntegrityPolys
```

Out[•]=
$$\{H[2]-10\left(\beta_1^2+\beta_2^2+\beta_3^2\right),\ H[4]-44\beta_1^4-44\beta_2^4-58\beta_2^2\beta_3^2-44\beta_3^4-58\beta_1^2\left(\beta_2^2+\beta_3^2\right),$$
$$J[2]-\beta_1^2-\beta_2^2-\beta_3^2,\ L[4]-2\left(\beta_1^4+\beta_2^4-3\beta_2^2\beta_3^2+\beta_3^4-3\beta_1^2\left(\beta_2^2+\beta_3^2\right)\right),$$
$$H[6]-4\left(\beta_3^2\left(\beta_1^2+\beta_2^2-4\beta_3^2\right)^2+\beta_2^2\left(\beta_1^2-4\beta_2^2+\beta_3^2\right)^2+\beta_1^2\left(-4\beta_1^2+\beta_2^2+\beta_3^2\right)^2\right),$$
$$H[10]-8\left(128\beta_1^{10}+128\beta_2^{10}-60\beta_2^8\beta_3^2-95\beta_2^6\beta_3^4-95\beta_2^4\beta_3^6-\right.$$
$$60\beta_2^2\beta_3^8+128\beta_3^{10}-60\beta_1^8\left(\beta_2^2+\beta_3^2\right)+\beta_1^6\left(-95\beta_2^4+60\beta_2^2\beta_3^2-95\beta_3^4\right)+$$
$$\left.\beta_1^4\left(-95\beta_2^6+90\beta_2^4\beta_3^2+90\beta_2^2\beta_3^4-95\beta_3^6\right)-30\beta_1^2\left(2\beta_2^8-2\beta_2^6\beta_3^2-3\beta_2^4\beta_3^4-2\beta_2^2\beta_3^6+2\beta_3^8\right)\right),$$
$$J[4]-2\left(3\beta_1^4+3\beta_2^4+\beta_2^2\beta_3^2+3\beta_3^4+\beta_1^2\left(\beta_2^2+\beta_3^2\right)\right),\ K[4]-8\beta_1^4-8\beta_2^4+4\beta_2^2\beta_3^2-8\beta_3^4+4\beta_1^2\left(\beta_2^2+\beta_3^2\right),$$
$$J[6]-12\beta_1^6-12\beta_2^6+19\beta_2^4\beta_3^2+19\beta_2^2\beta_3^4-12\beta_3^6+19\beta_1^4\left(\beta_2^2+\beta_3^2\right)+\beta_1^2\left(19\beta_2^4+18\beta_2^2\beta_3^2+19\beta_3^4\right),$$
$$K[6]-2\left(8\beta_1^6+8\beta_2^6-11\beta_2^4\beta_3^2-11\beta_2^2\beta_3^4+8\beta_3^6-11\beta_1^4\left(\beta_2^2+\beta_3^2\right)+\beta_1^2\left(-11\beta_2^4+18\beta_2^2\beta_3^2-11\beta_3^4\right)\right),$$
$$L[6]-6\left(8\beta_1^6+8\beta_2^6-\beta_2^4\beta_3^2-\beta_2^2\beta_3^4+8\beta_3^6-\beta_1^4\left(\beta_2^2+\beta_3^2\right)-\beta_1^2\left(\beta_2^2+\beta_3^2\right)^2\right),$$
$$M[6]-\beta_3^2\left(3\beta_1^2+3\beta_2^2-2\beta_3^2\right)^2-\beta_2^2\left(3\beta_1^2-2\beta_2^2+3\beta_3^2\right)^2-\left(2\beta_1^3-3\beta_1\left(\beta_2^2+\beta_3^2\right)\right)^2,$$
$$H[8]-4\left(72\beta_1^8+18\beta_1^6\left(\beta_2^2+\beta_3^2\right)-11\beta_1^4\left(3\beta_2^4+\beta_2^2\beta_3^2+3\beta_3^4\right)+\right.$$
$$\left.\beta_1^2\left(18\beta_2^6-11\beta_2^4\beta_3^2-11\beta_2^2\beta_3^4+18\beta_3^6\right)+3\left(24\beta_2^8+6\beta_2^6\beta_3^2-11\beta_2^4\beta_3^4+6\beta_2^2\beta_3^6+24\beta_3^8\right)\right)\}$$

We will express these later in terms of the q_i's.

Characteristic Polynomial coefficients of Γ^{*2}

In[72]:=
```
Γstarsqrd = Simplify[TensorContract[Γ⊗Γ, {{1, 4}, {2, 5}}]];
```

In[73]:=
```
Clear[q, ξ];
ξ = Rest[Reverse[CoefficientList[Simplify[Det[λ IdentityMatrix[3] + Γstarsqrd]], λ]]]
```

Out[•]=
$$\{\beta_1^2+\beta_2^2+\beta_3^2,\ \beta_1^2\beta_2^2+\beta_1^2\beta_3^2+\beta_2^2\beta_3^2,\ \beta_1^2\beta_2^2\beta_3^2\}$$

As expected, these coefficients are the elementary symmetric polynomials of the quantities β_i^2.

In[74]:=
```
DiagInvars = Table[qi - ξ[[i]], {i, 1, 3}]
```

Out[•]=
$$\{q_1-\beta_1^2-\beta_2^2-\beta_3^2,\ q_2-\beta_1^2\beta_2^2-\beta_1^2\beta_3^2-\beta_2^2\beta_3^2,\ q_3-\beta_1^2\beta_2^2\beta_3^2\}$$

This is the basis of invariants for the stabilizer group $G_R = S_3 \times (\mathbb{Z}_2)^3$ with respect to fully decoupled reduced form tensors. Since all the Olive and Auffray invariants, when restricted to fully decoupled tensors, are also G_R invariants, we can express them in terms of the quantities q_i.

In[75]:=
```
Table[GroebnerBasis[Join[{IntegrityPolys[[i]]}, DiagInvars],
    Join[{OAInvariants[[i]]}, {q1, q2, q3}], {β1, β2, β3},
    MonomialOrder → EliminationOrder][[1]], {i, 1, Length[OAInvariants]}]
```

Out[•]=
$$\{H[2]-10q_1,\ -H[4]+44q_1^2-30q_2,\ J[2]-q_1,\ -L[4]+2q_1^2-10q_2,$$
$$-H[6]+64q_1^3-220q_1q_2+300q_3,\ -H[10]+1024q_1^5-5600q_1^3q_2+5800q_1q_2^2+7600q_1^2q_3-7000q_2q_3,$$
$$-J[4]+6q_1^2-10q_2,\ -K[4]+8q_1^2-20q_2,\ -J[6]+12q_1^3-55q_1q_2+75q_3,$$
$$-K[6]+16q_1^3-70q_1q_2+150q_3,\ -L[6]+48q_1^3-150q_1q_2+150q_3,$$
$$-M[6]+4q_1^3-15q_1q_2+75q_3,\ -H[8]+288q_1^4-1080q_1^2q_2+300q_2^2+1300q_1q_3\}$$

These are the relations in Eq. (7.4).

Section 7.3

Partially decoupleable 3 × 3 × 3 tensors

In[76]:=
```
n = 3;
vars = Table[x[i], {i, 1, n}];
f = 3 α x[1] (x[1]^2 + x[2]^2) +
   γ1 (3 x[2]^2 × x[1] - x[1]^3) + γ2 (3 x[1]^2 × x[2] - x[2]^3) + β3 x[3]^3
```

Out[•]= $3\,\alpha\,x[1]\,\left(x[1]^2+x[2]^2\right)+\gamma_1\,\left(-x[1]^3+3\,x[1]\,x[2]^2\right)+\gamma_2\,\left(3\,x[1]^2\,x[2]-x[2]^3\right)+\beta_3\,x[3]^3$

This is the canonical form R corresponding to a partially decoupleable tensor when n=3.

In[77]:=
```
Coeffs = {α, γ1, γ2, β3};
```

This is the list of the tensor parameters in the canonical form.

In[78]:=
```
Γ = Simplify[Table[D[f, x[i], x[j], x[k]], {k, 1, n}, {j, 1, n}, {i, 1, n}] / 6]
```

Out[•]= $\{\{\{3\alpha-\gamma_1,\gamma_2,0\},\{\gamma_2,\alpha+\gamma_1,0\},\{0,0,0\}\},$
$\{\{\gamma_2,\alpha+\gamma_1,0\},\{\alpha+\gamma_1,-\gamma_2,0\},\{0,0,0\}\},\{\{0,0,0\},\{0,0,0\},\{0,0,\beta_3\}\}\}$

In[79]:=
```
Table[MatrixForm[Γ[[i]]], {i, 1, n}]
```

Out[•]= $\left\{\begin{pmatrix}3\alpha-\gamma_1 & \gamma_2 & 0\\ \gamma_2 & \alpha+\gamma_1 & 0\\ 0 & 0 & 0\end{pmatrix},\begin{pmatrix}\gamma_2 & \alpha+\gamma_1 & 0\\ \alpha+\gamma_1 & -\gamma_2 & 0\\ 0 & 0 & 0\end{pmatrix},\begin{pmatrix}0 & 0 & 0\\ 0 & 0 & 0\\ 0 & 0 & \beta_3\end{pmatrix}\right\}$

In[80]:=
```
u =  Simplify[Table[D[Laplacian[f, vars], x[i]], {i, 1, n}] / 6];
f3 = (n + 2) f - 3 (u.vars) (vars.vars)
```

Out[•]= $-3\,\left(4\,\alpha\,x[1]+\beta_3\,x[3]\right)\,\left(x[1]^2+x[2]^2+x[3]^2\right)+$
$5\,\left(3\,\alpha\,x[1]\,\left(x[1]^2+x[2]^2\right)+\gamma_1\,\left(-x[1]^3+3\,x[1]\,x[2]^2\right)+\gamma_2\,\left(3\,x[1]^2\,x[2]-x[2]^3\right)+\beta_3\,x[3]^3\right)$

This is the trace-free part.

In[82]:=
```
𝒟 = Simplify[Table[D[f3, x[i], x[j], x[k]], {k, 1, n}, {j, 1, n}, {i, 1, n}] / 6];
v = Simplify[TensorContract[(𝒟⊗𝒟)⊗𝒟, {{1, 4}, {2, 5}, {3, 7}, {6, 8}}]];
w = Simplify[𝒟.u.u];
```

The calculations follow the same steps as in the fully decoupled case.

In[85]:=
```
Clear[H, J, K, L, M, Q];
v = Simplify[TensorContract[(𝒟⊗𝒟)⊗𝒟, {{1, 4}, {2, 5}, {3, 7}, {6, 8}}]];
w = Simplify[𝒟.u.u];
Q = Simplify[TensorContract[𝒟⊗𝒟, {{1, 4}, {2, 5}}]];
γuu = w;
Trivialize = Table[Coeffs[[j]] → 0, {j, 1, Length[Coeffs]}];
```

In what follows, we use the decomposition $\mathcal{I}=\mathcal{I}^{+}\cup\mathcal{I}'$. We will call $\mathcal{I}^{+}$ as ‘Fundamental’ and $\mathcal{I}'$ as ‘Secondary’ in the following.

In[91]:=
```
FundamentalRelations = {H[2] - Simplify[Tr[Q]],
    H[4] - Simplify[Tr[Q.Q]], J[2] - Simplify[u.u], L[4] - Simplify[γuu.u]};
FundamentalInvariants = FundamentalRelations /. Trivialize;
SecondaryRelations =
   {H[6] - Simplify[v.v], H[10] - Simplify[𝒟.v.v.v], J[4] - Simplify[u.Q.u],
    K[4] - Simplify[Tr[Q.(𝒟.u)]], J[6] - Simplify[(u.Q).γuu], K[6] - Simplify[v.w],
    L[6] - Simplify[(u.Q).v], M[6] - Simplify[γuu.γuu], H[8] - Simplify[(u.Q).(Q.v)]};
SecondaryInvariants = SecondaryRelations /. Trivialize;
PartialDecoupleRelations = Join[FundamentalRelations, SecondaryRelations];
```

In[96]:=
```
G = Table[Γ⟦i, j, k⟧, {i, 1, 2}, {j, 1, 2}, {k, 1, 2}]
```

Out[ ]= $\{\{\{3\alpha - \gamma_1, \gamma_2\}, \{\gamma_2, \alpha + \gamma_1\}\}, \{\{\gamma_2, \alpha + \gamma_1\}, \{\alpha + \gamma_1, -\gamma_2\}\}\}$

This is the 2 × 2 × 2 block of R.

In[97]:=
```
GStarSqrd = Expand[TensorContract[G⊗G, {{1, 4}, {2, 5}}]]
```

Out[ ]= $\{\{10\alpha^2 - 4\alpha\gamma_1 + 2\gamma_1^2 + 2\gamma_2^2, 4\alpha\gamma_2\}, \{4\alpha\gamma_2, 2\alpha^2 + 4\alpha\gamma_1 + 2\gamma_1^2 + 2\gamma_2^2\}\}$

In[98]:=
```
MatrixForm[GStarSqrd]
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} 10\alpha^2 - 4\alpha\gamma_1 + 2\gamma_1^2 + 2\gamma_2^2 & 4\alpha\gamma_2 \\ 4\alpha\gamma_2 & 2\alpha^2 + 4\alpha\gamma_1 + 2\gamma_1^2 + 2\gamma_2^2 \end{pmatrix}$$

In[99]:=
```
ured = TensorContract[G, {1, 2}]
```

Out[ ]= $\{4\alpha, 0\}$

This is the trace of the 2 × 2 × 2 block of R.

In[100]:=
```
Clear[q];
defns = {q₁ - β₃², q₂ - ured.ured, q₄ - Tr[GStarSqrd], q₃ - Det[GStarSqrd]}
```

Out[ ]= $\{q_1 - \beta_3^2, -16\alpha^2 + q_2, -12\alpha^2 + q_4 - 4\gamma_1^2 - 4\gamma_2^2,$
$-20\alpha^4 + q_3 - 32\alpha^3\gamma_1 - 8\alpha^2\gamma_1^2 - 4\gamma_1^4 - 8\alpha^2\gamma_2^2 - 8\gamma_1^2\gamma_2^2 - 4\gamma_2^4\}$

The quantities q_1, q_2, q_3, q_4 are O(2) x $\mathbb{Z}_2$ invariants in terms of the parameters defining the Canonical form.

In[102]:=
```
EliminateCoeffs =
  Table[GroebnerBasis[Join[{FundamentalRelations⟦i⟧}, defns], Join[FundamentalInvariants,
      {q₃, q₄, q₂, q₁}], Coeffs, MonomialOrder → EliminationOrder]⟦1⟧, {i, 1, 4}]
```

Out[ ]= $\{H[2] - 10 q_1 + 15 q_2 - 25 q_4, -H[4] + 44 q_1^2 - 42 q_1 q_2 + 144 q_2^2 - 30 q_3 + 100 q_1 q_4 - 420 q_2 q_4 + 320 q_4^2,$
$J[2] - q_1 - q_2, -2 L[4] + 4 q_1^2 - 12 q_1 q_2 + 4 q_2^2 - 20 q_3 - 5 q_2 q_4 + 5 q_4^2\}$

Since there are no relations involving only the q_i's, they are independent. Moreover, this is Eq. (7.8) expressing the elements in $\mathcal{I}^+$ in terms of the O(2) x $\mathbb{Z}_2$ invariants q_i. We can invert these relations and solve for the quantities q_i.

In[103]:=
```
TriangularSystem = GroebnerBasis[EliminateCoeffs, {q4, q3, q2, q1, H[2], H[4], J[2], L[4]}]
```

Out[∘]=
$$\{-H[2]^2 + 2\,H[4] + 3\,H[2] \times J[2] - 6\,J[2]^2 - 6\,L[4] + 9\,H[2]\,q_1 - 90\,J[2]\,q_1,\ -J[2] + q_1 + q_2,$$
$$8\,H[2]^2 - 25\,H[4] - 60\,H[2] \times J[2] - 1500\,J[2]^2 + 1200\,L[4] + 11\,250\,J[2]\,q_1 - 11\,250\,q_1^2 + 11\,250\,q_3,$$
$$-H[2] - 15\,J[2] + 25\,q_1 + 25\,q_4\}$$

In[104]:=
```
Substitutions = Table[Solve[TriangularSystem[[i]] == 0, qi][[1, 1]], {i, 1, 4}]
```

Out[∘]=
$$\left\{q_1 \to \frac{H[2]^2 - 2\,H[4] - 3\,H[2] \times J[2] + 6\,J[2]^2 + 6\,L[4]}{9\,(H[2] - 10\,J[2])},\ q_2 \to J[2] - q_1,\right.$$
$$q_3 \to \frac{-8\,H[2]^2 + 25\,H[4] + 60\,H[2] \times J[2] + 1500\,J[2]^2 - 1200\,L[4] - 11\,250\,J[2]\,q_1 + 11\,250\,q_1^2}{11\,250},$$
$$\left.q_4 \to \frac{1}{25}\,(H[2] + 15\,J[2] - 25\,q_1)\right\}$$

These are the substitutions implied by Eq. (7.9).

Theorem 7.2: Expressing the invariants in $\mathcal{I}$’ in terms of $\mathcal{I}^+$

In[105]:=
```
Eliminated = {α, γ1, γ2};
```

We make a choice to include β_3 in defining the necessary/sufficient conditions and only eliminate α, γ_1 and γ_2. The rationale is that among all the relations between the parameters of the canonical form and the invariants, the only relation which is not a polynomial is the relation for ${\beta_3}^2$ in terms of the OA invariants. So, we can get more compact expressions without denominators if we also include it in the set of `basic’ quantities for expressing the rest of the invariants.

In[106]:=
```
GroebnerBasis[FundamentalRelations, FundamentalInvariants, Eliminated]
```

Out[∘]=
$$\{H[2]^2 - 2\,H[4] - 3\,H[2] \times J[2] + 6\,J[2]^2 + 6\,L[4] - 9\,H[2]\,\beta_3^2 + 90\,J[2]\,\beta_3^2\}$$

This is the relation for ${\beta_3}^2$ in terms of the OA invariants. Inverting gives a rational function for ${\beta_3}^2$ which is uniquely defined only if H[2] ≠ 10 J[2]. This will give ${\beta_3}^2 = \frac{H[2]^2-2\,H[4]-3\,H[2]\times J[2]+6\,J[2]^2+6\,L[4]}{9\,(H[2]-10\,J[2])}$. We now calculate the polynomial expressions for the Secondary invariants in terms of the Fundamental invariants *and* ${\beta_3}^2$ with an ordering that promotes low order polynomials in β_3.

In[107]:=
```
NeccSuffRelations =
  Join[GroebnerBasis[FundamentalRelations, FundamentalInvariants, Eliminated],
   Table[GroebnerBasis[Join[{SecondaryRelations[[i]]}, FundamentalRelations],
      Join[{SecondaryInvariants[[i]]}, FundamentalInvariants], Eliminated][[2]], {i, 1, 9}]]
```

Out[•]=
$$\{H[2]^2 - 2H[4] - 3H[2]\times J[2] + 6J[2]^2 + 6L[4] - 9H[2]\,\beta_3^2 + 90J[2]\,\beta_3^2,$$
$$H[6] - 2H[4]\times J[2] + 8H[2]\,J[2]^2 - 24J[2]^3 - 4H[2]\times L[4] + 24J[2]\times L[4] - 15H[2]\times J[2]\,\beta_3^2 + 90J[2]^2\beta_3^2 + 30L[4]\,\beta_3^2 + 39H[2]\,\beta_3^4 - 90J[2]\,\beta_3^4 - 300\beta_3^6,$$
$$H[10] - 3H[2]\times H[4]\,J[2]^2 + 40H[4]\,J[2]^3 - 63H[2]\,J[2]^4 + 114J[2]^5 - 2H[2]\times H[4]\times L[4] + 6H[4]\times J[2]\times L[4] + 42H[2]\,J[2]^2L[4] - 246J[2]^3L[4] - 4H[2]\,L[4]^2 + 24J[2]\,L[4]^2 - 6H[2]\times H[4]\times J[2]\,\beta_3^2 - 36H[4]\,J[2]^2\beta_3^2 + 485H[2]\,J[2]^3\beta_3^2 - 2442J[2]^4\beta_3^2 + 30H[4]\times L[4]\,\beta_3^2 - 72H[2]\times J[2]\times L[4]\,\beta_3^2 + 348J[2]^2L[4]\,\beta_3^2 - 20L[4]^2\beta_3^2 + 17H[2]\times H[4]\,\beta_3^4 + 36H[4]\times J[2]\,\beta_3^4 - 363H[2]\,J[2]^2\beta_3^4 + 2002J[2]^3\beta_3^4 + 70H[2]\times L[4]\,\beta_3^4 - 618J[2]\times L[4]\,\beta_3^4 - 280H[4]\,\beta_3^6 + 807H[2]\times J[2]\,\beta_3^6 - 3630J[2]^2\beta_3^6 + 140L[4]\,\beta_3^6 - 210H[2]\,\beta_3^8 + 2100J[2]\,\beta_3^8,$$
$$-H[2]\times J[2] + 2J[2]^2 + 2J[4] - 2L[4] + H[2]\,\beta_3^2 - 10J[2]\,\beta_3^2,$$
$$-H[2]\times J[2] + 6J[2]^2 + K[4] - 2L[4] + 2H[2]\,\beta_3^2 - 20J[2]\,\beta_3^2,$$
$$-H[2]\,J[2]^2 + 6J[2]^3 + 4J[6] - 2H[2]\times L[4] - 2J[2]\times L[4] - 4H[2]\times J[2]\,\beta_3^2 - 20J[2]^2\beta_3^2 + 30L[4]\,\beta_3^2 + 9H[2]\,\beta_3^4 + 210J[2]\,\beta_3^4 - 300\beta_3^6,$$
$$-H[2]\,J[2]^2 + 6J[2]^3 + 2K[6] - 2H[2]\times L[4] + 6J[2]\times L[4] - 10H[2]\times J[2]\,\beta_3^2 + 40J[2]^2\beta_3^2 + 30L[4]\,\beta_3^2 + 19H[2]\,\beta_3^4 + 110J[2]\,\beta_3^4 - 300\beta_3^6,$$
$$-H[4]\times J[2] + 2H[2]\,J[2]^2 - 2H[2]\times L[4] + 8J[2]\times L[4] + L[6] + H[4]\,\beta_3^2 - 11H[2]\times J[2]\,\beta_3^2 + 42J[2]^2\beta_3^2 + 12L[4]\,\beta_3^2 + 19H[2]\,\beta_3^4 - 40J[2]\,\beta_3^4 - 150\beta_3^6,$$
$$-H[2]\,J[2]^2 + 6J[2]^3 - 6J[2]\times L[4] + 4M[6] + 2H[2]\times J[2]\,\beta_3^2 - 80J[2]^2\beta_3^2 + 30L[4]\,\beta_3^2 - H[2]\,\beta_3^4 + 310J[2]\,\beta_3^4 - 300\beta_3^6,\ 2H[8] - H[2]\times H[4]\times J[2] + 4H[4]\,J[2]^2 + 9H[2]\,J[2]^3 - 30J[2]^4 - 6H[4]\times L[4] + 4H[2]\times J[2]\times L[4] + 14J[2]^2L[4] + 12L[4]^2 + H[2]\times H[4]\,\beta_3^2 - 32H[4]\times J[2]\,\beta_3^2 + 36H[2]\,J[2]^2\beta_3^2 - 44J[2]^3\beta_3^2 + 10H[2]\times L[4]\,\beta_3^2 + 226J[2]\times L[4]\,\beta_3^2 + 40H[4]\,\beta_3^4 + 69H[2]\times J[2]\,\beta_3^4 + 390J[2]^2\beta_3^4 - 120L[4]\,\beta_3^4 - 70H[2]\,\beta_3^6 - 1900J[2]\,\beta_3^6\}$$

The first relation defines β_3. We only need consider the rest of the relations since we allow ${\beta_3}^2$ as a ‘variable’.

In[108]:=
```
NeccSuffRelations = Rest[NeccSuffRelations]
```

Out[∘]=

$$
\begin{array}{l}
\{H[6] - 2\,H[4] \times J[2] + 8\,H[2]\,J[2]^2 - 24\,J[2]^3 - 4\,H[2] \times L[4] + 24\,J[2] \times L[4] - \\
\quad 15\,H[2] \times J[2]\,\beta_3^2 + 90\,J[2]^2\,\beta_3^2 + 30\,L[4]\,\beta_3^2 + 39\,H[2]\,\beta_3^4 - 90\,J[2]\,\beta_3^4 - 300\,\beta_3^6, \\
H[10] - 3\,H[2] \times H[4]\,J[2]^2 + 40\,H[4]\,J[2]^3 - 63\,H[2]\,J[2]^4 + 114\,J[2]^5 - 2\,H[2] \times H[4] \times L[4] + \\
\quad 6\,H[4] \times J[2] \times L[4] + 42\,H[2]\,J[2]^2\,L[4] - 246\,J[2]^3\,L[4] - 4\,H[2]\,L[4]^2 + 24\,J[2]\,L[4]^2 - \\
\quad 6\,H[2] \times H[4] \times J[2]\,\beta_3^2 - 36\,H[4]\,J[2]^2\,\beta_3^2 + 485\,H[2]\,J[2]^3\,\beta_3^2 - 2442\,J[2]^4\,\beta_3^2 + \\
\quad 30\,H[4] \times L[4]\,\beta_3^2 - 72\,H[2] \times J[2] \times L[4]\,\beta_3^2 + 348\,J[2]^2\,L[4]\,\beta_3^2 - 20\,L[4]^2\,\beta_3^2 + 17\,H[2] \times H[4]\,\beta_3^4 + \\
\quad 36\,H[4] \times J[2]\,\beta_3^4 - 363\,H[2]\,J[2]^2\,\beta_3^4 + 2002\,J[2]^3\,\beta_3^4 + 70\,H[2] \times L[4]\,\beta_3^4 - 618\,J[2] \times L[4]\,\beta_3^4 - \\
\quad 280\,H[4]\,\beta_3^6 + 807\,H[2] \times J[2]\,\beta_3^6 - 3630\,J[2]^2\,\beta_3^6 + 140\,L[4]\,\beta_3^6 - 210\,H[2]\,\beta_3^8 + 2100\,J[2]\,\beta_3^8, \\
-H[2] \times J[2] + 2\,J[2]^2 + 2\,J[4] - 2\,L[4] + H[2]\,\beta_3^2 - 10\,J[2]\,\beta_3^2, \\
-H[2] \times J[2] + 6\,J[2]^2 + K[4] - 2\,L[4] + 2\,H[2]\,\beta_3^2 - 20\,J[2]\,\beta_3^2, \\
-H[2]\,J[2]^2 + 6\,J[2]^3 + 4\,J[6] - 2\,H[2] \times L[4] - 2\,J[2] \times L[4] - \\
\quad 4\,H[2] \times J[2]\,\beta_3^2 - 20\,J[2]^2\,\beta_3^2 + 30\,L[4]\,\beta_3^2 + 9\,H[2]\,\beta_3^4 + 210\,J[2]\,\beta_3^4 - 300\,\beta_3^6, \\
-H[2]\,J[2]^2 + 6\,J[2]^3 + 2\,K[6] - 2\,H[2] \times L[4] + 6\,J[2] \times L[4] - 10\,H[2] \times J[2]\,\beta_3^2 + \\
\quad 40\,J[2]^2\,\beta_3^2 + 30\,L[4]\,\beta_3^2 + 19\,H[2]\,\beta_3^4 + 110\,J[2]\,\beta_3^4 - 300\,\beta_3^6, \\
-H[4] \times J[2] + 2\,H[2]\,J[2]^2 - 2\,H[2] \times L[4] + 8\,J[2] \times L[4] + L[6] + H[4]\,\beta_3^2 - \\
\quad 11\,H[2] \times J[2]\,\beta_3^2 + 42\,J[2]^2\,\beta_3^2 + 12\,L[4]\,\beta_3^2 + 19\,H[2]\,\beta_3^4 - 40\,J[2]\,\beta_3^4 - 150\,\beta_3^6, \\
-H[2]\,J[2]^2 + 6\,J[2]^3 - 6\,J[2] \times L[4] + 4\,M[6] + 2\,H[2] \times J[2]\,\beta_3^2 - 80\,J[2]^2\,\beta_3^2 + 30\,L[4]\,\beta_3^2 - \\
\quad H[2]\,\beta_3^4 + 310\,J[2]\,\beta_3^4 - 300\,\beta_3^6,\ 2\,H[8] - H[2] \times H[4] \times J[2] + 4\,H[4]\,J[2]^2 + 9\,H[2]\,J[2]^3 - \\
\quad 30\,J[2]^4 - 6\,H[4] \times L[4] + 4\,H[2] \times J[2] \times L[4] + 14\,J[2]^2\,L[4] + 12\,L[4]^2 + H[2] \times H[4]\,\beta_3^2 - \\
\quad 32\,H[4] \times J[2]\,\beta_3^2 + 36\,H[2]\,J[2]^2\,\beta_3^2 - 44\,J[2]^3\,\beta_3^2 + 10\,H[2] \times L[4]\,\beta_3^2 + 226\,J[2] \times L[4]\,\beta_3^2 + \\
\quad 40\,H[4]\,\beta_3^4 + 69\,H[2] \times J[2]\,\beta_3^4 + 390\,J[2]^2\,\beta_3^4 - 120\,L[4]\,\beta_3^4 - 70\,H[2]\,\beta_3^6 - 1900\,J[2]\,\beta_3^6\}
\end{array}
$$

In[109]:=
```
Normalizations = Table[D[NeccSuffRelations[[i]], SecondaryInvariants[[i]]], {i, 1, 9}]
```

Out[∘]=
```
{1, 1, 2, 1, 4, 2, 1, 4, 2}
```

These relations are linear in the Secondary invariants. We compute the coefficients to ensure they are nonzero constants. This also identifies the denominators that we will get in solving for the Secondary invariants.

The next step solves for the Secondary invariants and replaces ${\beta_3}^2$ by q_1.

In[110]:=
```
SolveNeccSuff =
  Table[Solve[NeccSuffRelations[[i]] == 0, SecondaryInvariants[[i]]][[1, 1]], {i, 1, 9}] /.
   {β3^k_ → q1^(k/2)}
```

Out[•]=

$$\{H[6] \to 2H[4]\times J[2] - 8H[2]\,J[2]^2 + 24J[2]^3 + 4H[2]\times L[4] - 24J[2]\times L[4] + 15H[2]\times J[2]\,q_1 - 90J[2]^2 q_1 - 30L[4]\,q_1 - 39H[2]\,q_1^2 + 90J[2]\,q_1^2 + 300q_1^3,$$

$$H[10] \to 3H[2]\times H[4]\,J[2]^2 - 40H[4]\,J[2]^3 + 63H[2]\,J[2]^4 - 114J[2]^5 + 2H[2]\times H[4]\times L[4] - 6H[4]\times J[2]\times L[4] - 42H[2]\,J[2]^2 L[4] + 246J[2]^3 L[4] + 4H[2]\,L[4]^2 - 24J[2]\,L[4]^2 + 6H[2]\times H[4]\times J[2]\,q_1 + 36H[4]\,J[2]^2 q_1 - 485H[2]\,J[2]^3 q_1 + 2442J[2]^4 q_1 - 30H[4]\times L[4]\,q_1 + 72H[2]\times J[2]\times L[4]\,q_1 - 348J[2]^2 L[4]\,q_1 + 20L[4]^2 q_1 - 17H[2]\times H[4]\,q_1^2 - 36H[4]\times J[2]\,q_1^2 + 363H[2]\,J[2]^2 q_1^2 - 2002J[2]^3 q_1^2 - 70H[2]\times L[4]\,q_1^2 + 618J[2]\times L[4]\,q_1^2 + 280H[4]\,q_1^3 - 807H[2]\times J[2]\,q_1^3 + 3630J[2]^2 q_1^3 - 140L[4]\,q_1^3 + 210H[2]\,q_1^4 - 2100J[2]\,q_1^4,$$

$$J[4] \to \frac{1}{2}\left(H[2]\times J[2] - 2J[2]^2 + 2L[4] - H[2]\,q_1 + 10J[2]\,q_1\right),$$

$$K[4] \to H[2]\times J[2] - 6J[2]^2 + 2L[4] - 2H[2]\,q_1 + 20J[2]\,q_1,$$

$$J[6] \to \frac{1}{4}\left(H[2]\,J[2]^2 - 6J[2]^3 + 2H[2]\times L[4] + 2J[2]\times L[4] + 4H[2]\times J[2]\,q_1 + 20J[2]^2 q_1 - 30L[4]\,q_1 - 9H[2]\,q_1^2 - 210J[2]\,q_1^2 + 300q_1^3\right),$$

$$K[6] \to \frac{1}{2}\left(H[2]\,J[2]^2 - 6J[2]^3 + 2H[2]\times L[4] - 6J[2]\times L[4] + 10H[2]\times J[2]\,q_1 - 40J[2]^2 q_1 - 30L[4]\,q_1 - 19H[2]\,q_1^2 - 110J[2]\,q_1^2 + 300q_1^3\right),$$

$$L[6] \to H[4]\times J[2] - 2H[2]\,J[2]^2 + 2H[2]\times L[4] - 8J[2]\times L[4] - H[4]\,q_1 + 11H[2]\times J[2]\,q_1 - 42J[2]^2 q_1 - 12L[4]\,q_1 - 19H[2]\,q_1^2 + 40J[2]\,q_1^2 + 150q_1^3,$$

$$M[6] \to \frac{1}{4}\left(H[2]\,J[2]^2 - 6J[2]^3 + 6J[2]\times L[4] - 2H[2]\times J[2]\,q_1 + 80J[2]^2 q_1 - 30L[4]\,q_1 + H[2]\,q_1^2 - 310J[2]\,q_1^2 + 300q_1^3\right),$$

$$H[8] \to \frac{1}{2}\left(H[2]\times H[4]\times J[2] - 4H[4]\,J[2]^2 - 9H[2]\,J[2]^3 + 30J[2]^4 + 6H[4]\times L[4] - 4H[2]\times J[2]\times L[4] - 14J[2]^2 L[4] - 12L[4]^2 - H[2]\times H[4]\,q_1 + 32H[4]\times J[2]\,q_1 - 36H[2]\,J[2]^2 q_1 + 44J[2]^3 q_1 - 10H[2]\times L[4]\,q_1 - 226J[2]\times L[4]\,q_1 - 40H[4]\,q_1^2 - 69H[2]\times J[2]\,q_1^2 - 390J[2]^2 q_1^2 + 120L[4]\,q_1^2 + 70H[2]\,q_1^3 + 1900J[2]\,q_1^3\right)\}$$

Multiplying out the denominators to get relations with integer coefficients.

In[111]:=
```
FormatAsEquations = Table[Normalizations〚i〛 × SecondaryInvariants〚i〛 →
   Collect[(Normalizations〚i〛 × SecondaryInvariants〚i〛 /. SolveNeccSuff), q₁], {i, 1, 9}]
```

Out[∘]=

$$\{H[6] \to 2\,H[4] \times J[2] - 8\,H[2]\,J[2]^2 + 24\,J[2]^3 + 4\,H[2] \times L[4] - 24\,J[2] \times L[4] + \left(15\,H[2] \times J[2] - 90\,J[2]^2 - 30\,L[4]\right) q_1 + (-39\,H[2] + 90\,J[2])\,q_1^2 + 300\,q_1^3,$$

$$H[10] \to 3\,H[2] \times H[4]\,J[2]^2 - 40\,H[4]\,J[2]^3 + 63\,H[2]\,J[2]^4 - 114\,J[2]^5 + 2\,H[2] \times H[4] \times L[4] - 6\,H[4] \times J[2] \times L[4] - 42\,H[2]\,J[2]^2\,L[4] + 246\,J[2]^3\,L[4] + 4\,H[2]\,L[4]^2 - 24\,J[2]\,L[4]^2 + \left(6\,H[2] \times H[4] \times J[2] + 36\,H[4]\,J[2]^2 - 485\,H[2]\,J[2]^3 + 2442\,J[2]^4 - 30\,H[4] \times L[4] + 72\,H[2] \times J[2] \times L[4] - 348\,J[2]^2\,L[4] + 20\,L[4]^2\right) q_1 + \left(-17\,H[2] \times H[4] - 36\,H[4] \times J[2] + 363\,H[2]\,J[2]^2 - 2002\,J[2]^3 - 70\,H[2] \times L[4] + 618\,J[2] \times L[4]\right) q_1^2 + \left(280\,H[4] - 807\,H[2] \times J[2] + 3630\,J[2]^2 - 140\,L[4]\right) q_1^3 + (210\,H[2] - 2100\,J[2])\,q_1^4,$$

$$2\,J[4] \to H[2] \times J[2] - 2\,J[2]^2 + 2\,L[4] + (-H[2] + 10\,J[2])\,q_1,$$

$$K[4] \to H[2] \times J[2] - 6\,J[2]^2 + 2\,L[4] + (-2\,H[2] + 20\,J[2])\,q_1,$$

$$4\,J[6] \to H[2]\,J[2]^2 - 6\,J[2]^3 + 2\,H[2] \times L[4] + 2\,J[2] \times L[4] + \left(4\,H[2] \times J[2] + 20\,J[2]^2 - 30\,L[4]\right) q_1 + (-9\,H[2] - 210\,J[2])\,q_1^2 + 300\,q_1^3,$$

$$2\,K[6] \to H[2]\,J[2]^2 - 6\,J[2]^3 + 2\,H[2] \times L[4] - 6\,J[2] \times L[4] + \left(10\,H[2] \times J[2] - 40\,J[2]^2 - 30\,L[4]\right) q_1 + (-19\,H[2] - 110\,J[2])\,q_1^2 + 300\,q_1^3,$$

$$L[6] \to H[4] \times J[2] - 2\,H[2]\,J[2]^2 + 2\,H[2] \times L[4] - 8\,J[2] \times L[4] + \left(-H[4] + 11\,H[2] \times J[2] - 42\,J[2]^2 - 12\,L[4]\right) q_1 + (-19\,H[2] + 40\,J[2])\,q_1^2 + 150\,q_1^3,$$

$$4\,M[6] \to H[2]\,J[2]^2 - 6\,J[2]^3 + 6\,J[2] \times L[4] + \left(-2\,H[2] \times J[2] + 80\,J[2]^2 - 30\,L[4]\right) q_1 + (H[2] - 310\,J[2])\,q_1^2 + 300\,q_1^3,$$

$$2\,H[8] \to H[2] \times H[4] \times J[2] - 4\,H[4]\,J[2]^2 - 9\,H[2]\,J[2]^3 + 30\,J[2]^4 + 6\,H[4] \times L[4] - 4\,H[2] \times J[2] \times L[4] - 14\,J[2]^2\,L[4] - 12\,L[4]^2 + \left(-H[2] \times H[4] + 32\,H[4] \times J[2] - 36\,H[2]\,J[2]^2 + 44\,J[2]^3 - 10\,H[2] \times L[4] - 226\,J[2] \times L[4]\right) q_1 + \left(-40\,H[4] - 69\,H[2] \times J[2] - 390\,J[2]^2 + 120\,L[4]\right) q_1^2 + (70\,H[2] + 1900\,J[2])\,q_1^3\}$$

Calculations with respect to Lemma 7.3

The presentation in the paper summarizes the main points in the following systematic analysis.

In[112]:=
```
GroebnerBasis[defns, Join[{β₃}, Table[qᵢ, {i, 1, 4}]], {α, γ₁, γ₂}]
```

Out[∘]=

$$\{-q_1 + \beta_3^2\}$$

We first eliminate α, $\gamma_1$and γ_2. We find that $q_1 \geq 0$ is necessary to solve for a real β_3. Also, $q_1 > 0$ is `generic’ while $q_1 = 0$ is `degenerate’ and a case we should look at further. The degenerate case corresponds to $\beta_3 = 0$. We first look at the `generic’ case, then we will return to the case $\beta_3 = 0$.

In[113]:=
```
GenericBeta = GroebnerBasis[defns, Join[Table[qᵢ, {i, 1, 4}], {α, γ₁, γ₂}], {β₃}]
```

Out[∘]=

$$\{-12\,\alpha^2 + q_4 - 4\,\gamma_1^2 - 4\,\gamma_2^2,\ -20\,\alpha^4 + q_3 - 32\,\alpha^3\,\gamma_1 - 8\,\alpha^2\,\gamma_1^2 - 4\,\gamma_1^4 - 8\,\alpha^2\,\gamma_2^2 - 8\,\gamma_1^2\,\gamma_2^2 - 4\,\gamma_2^4,\ -16\,\alpha^2 + q_2\}$$

In the generic case $q_1 > 0$, we have three relations for the remaining parameters in terms of the quantities q_i. We note that there is one equation isolating α, so we solve for α first. Also note that, α^2 is determined by q_2 but not α itself. In this system, the terms involving γ_2 only have even powers of α, so they are completely determined by the q_i. There however are terms in γ_1 that involve odd powers of α. Here, we should not eliminate α when we are trying to solve for γ_1.

In[114]:= `GroebnerBasis[GenericBeta, Join[{α}, Table[qi, {i, 1, 4}]], {γ1, γ2}]`

Out[•]= $\{16\,\alpha^2 - q_2\}$

We find that $q_2 \geq 0$ is necessary for a real solution α. The case $q_2 > 0$ is `generic', and the case $q_2 = 0$ is `degenerate'.

In[115]:= `Genericαβ = GroebnerBasis[GenericBeta, Join[{γ1, γ2}, Table[qi, {i, 1, 4}]], {α}]`

Out[•]= $\{q_2^4 - 4\,q_2^2\,q_3 + 16\,q_3^2 - 2\,q_2^3\,q_4 + 8\,q_2\,q_3\,q_4 + 2\,q_2^2\,q_4^2 - 8\,q_3\,q_4^2 - 2\,q_2\,q_4^3 + q_4^4 + 4\,q_2^3\,\gamma_2^2,\ 3\,q_2 - 4\,q_4 + 16\,\gamma_1^2 + 16\,\gamma_2^2\}$

This is a triangular system with two equations for γ_1 and γ_2. We can solve them in turn to find γ_2 and γ_1

In[116]:= `γ2eqn = GroebnerBasis[Genericαβ, Join[{γ2}, Table[qi, {i, 1, 4}]], {γ1}]`

Out[•]= $\{q_2^4 - 4\,q_2^2\,q_3 + 16\,q_3^2 - 2\,q_2^3\,q_4 + 8\,q_2\,q_3\,q_4 + 2\,q_2^2\,q_4^2 - 8\,q_3\,q_4^2 - 2\,q_2\,q_4^3 + q_4^4 + 4\,q_2^3\,\gamma_2^2\}$

We isolate γ_2. This equation has a unique solution for ${\gamma_2}^2$ provided $q_2 \neq 0$, which is true in the generic case $q_2 > 0$.

In[117]:= `Solve[(γ2eqn /. {γ2^2 → γ2sqrd}) == 0, γ2sqrd][[1]]`

Out[•]= $$\left\{\gamma2sqrd \to \frac{-q_2^4 + 4\,q_2^2\,q_3 - 16\,q_3^2 + 2\,q_2^3\,q_4 - 8\,q_2\,q_3\,q_4 - 2\,q_2^2\,q_4^2 + 8\,q_3\,q_4^2 + 2\,q_2\,q_4^3 - q_4^4}{4\,q_2^3}\right\}$$

Still in the generic case $q_2 > 0$ and so $\alpha > 0$, we may solve for γ_1 below.

In[118]:= `γ1eqns = GroebnerBasis[GenericBeta, Join[{γ1, α}, Table[qi, {i, 1, 4}]], {γ2}]`

Out[•]= $\{16\,\alpha^2 - q_2,\ \alpha\,q_2^2 - 8\,\alpha\,q_3 - 2\,\alpha\,q_2\,q_4 + 2\,\alpha\,q_4^2 + q_2^2\,\gamma_1,\ q_2^2 - 8\,q_3 - 2\,q_2\,q_4 + 2\,q_4^2 + 16\,\alpha\,q_2\,\gamma_1\}$

In[119]:= `Solve[γ1eqns[[3]] == 0, γ1][[1]]`

Out[•]= $$\left\{\gamma_1 \to \frac{-q_2^2 + 8\,q_3 + 2\,q_2\,q_4 - 2\,q_4^2}{16\,\alpha\,q_2}\right\}$$

This concludes the discussion of the generic case, $q_2 \neq 0$, $q_1 \neq 0$.
We next specialize to the case $q_2 = 0$ which implies α=0.

In[120]:= `Specializedαeq0 = GenericBeta /. {α → 0, q2 → 0}`

Out[•]= $\{q_4 - 4\,\gamma_1^2 - 4\,\gamma_2^2,\ q_3 - 4\,\gamma_1^4 - 8\,\gamma_1^2\,\gamma_2^2 - 4\,\gamma_2^4,\ 0\}$

We obtain two equations for γ_1 and γ_2. We check for any additional necessary conditions for solvability by eliminating γ_1 and γ_2.

In[121]:= `GroebnerBasis[Specializedαeq0, Join[Table[qi, {i, 1, 4}]], {γ1, γ2}]`

Out[•]= $\{4\,q_3 - q_4^2\}$

We do obtain an additional solvability condition. In the case that $q_2 = 0$, we also need $q_3 = \frac{{q_4}^2}{4}$. We redefined the specialization to include this condition.

In[122]:= `Specializedαeq0 = GenericBeta /. {α → 0, q2 → 0, q3 → q4^2 / 4}`

Out[ ]= $\left\{q_4 - 4\gamma_1^2 - 4\gamma_2^2,\ \frac{q_4^2}{4} - 4\gamma_1^4 - 8\gamma_1^2\gamma_2^2 - 4\gamma_2^4,\ 0\right\}$

In[123]:= `GroebnerBasis[Specializedαeq0, {q4, γ1, γ2}]`

Out[ ]= $\left\{q_4 - 4\gamma_1^2 - 4\gamma_2^2\right\}$

We obtain a single equation for $\gamma_1{}^2 + \gamma_2{}^2$. This equation is solvable if $q_4 \geq 0$. We have a continuum of solutions in this case. This concludes the discussion of the case $q_1 > 0$, $q_2 = 0$, where we have the additional solvability conditions $4\,q_3 = q_4{}^2$ and $q_4 \geq 0$. The full solution is $\beta_3 = \sqrt{q_1}$, $\alpha = 0$, $\gamma_1{}^2 + \gamma_2{}^2 = \frac{q_4}{4}$.

We consider the last remaining case, the specialization $q_1 = 0$, $\beta_3 = 0$.

In[124]:= `β3eq0 = defns /. {β3 → 0, q1 → 0}`

Out[ ]= $\left\{0,\ -16\alpha^2 + q_2,\ -12\alpha^2 + q_4 - 4\gamma_1^2 - 4\gamma_2^2,\ -20\alpha^4 + q_3 - 32\alpha^3\gamma_1 - 8\alpha^2\gamma_1^2 - 4\gamma_1^4 - 8\alpha^2\gamma_2^2 - 8\gamma_1^2\gamma_2^2 - 4\gamma_2^4\right\}$

We now specialize to the degenerate case $\beta_3 = 0$.

In[125]:= `GroebnerBasis[β3eq0, {α, γ1, γ2}, {q1, q2, q3, q4}, MonomialOrder → EliminationOrder]`

Out[ ]= { }

We first check that setting $\beta_3 = 0$ does not impose conditions among the 'independent' quantities α, γ_1 and γ_2 which define R.

In[126]:= `GroebnerBasis[β3eq0, {q1, q2, q3, q4}, {α, γ1, γ2}, MonomialOrder → EliminationOrder]`

Out[ ]= { }

We determine all the consequences of having $\beta_3 = 0$. The only relation is $q_1 = 0$ and there are no additional solvability conditions from imposing $\beta_3 = 0$.

In[127]:= `GroebnerBasis[β3eq0, Join[{α}, Table[qi, {i, 1, 4}]], {γ1, γ2}]`

Out[ ]= $\left\{16\alpha^2 - q_2\right\}$

α is given by $16\,\alpha^2 = q_2$. We can solve this for a real α if $q_2 \geq 0$. The degenerate case is α=0, $q_2 = 0$. We first consider the generic case $q_2 > 0$. We begin by eliminating α, which now has a generic value.

In[128]:= `Genericαβeq0 = GroebnerBasis[β3eq0, Join[{γ1, γ2}, Table[qi, {i, 1, 4}]], {α}]`

Out[ ]= $\left\{q_2^4 - 4q_2^2q_3 + 16q_3^2 - 2q_2^3q_4 + 8q_2q_3q_4 + 2q_2^2q_4^2 - 8q_3q_4^2 - 2q_2q_4^3 + q_4^4 + 4q_2^3\gamma_2^2,\ 3q_2 - 4q_4 + 16\gamma_1^2 + 16\gamma_2^2\right\}$

This is the same triangular system as before with two equations for γ_1 and γ_2 and the solutions are still the same. We get $\gamma_2{}^2 = \frac{-q_2^4+4q_2^2q_3-16q_3^2+2q_2^3q_4-8q_2q_3q_4-2q_2^2q_4^2+8q_3q_4^2+2q_2q_4^3-q_4^4}{4q_2^3}$, $\gamma_1 = \frac{-q_2^2+8q_3+2q_2q_4-2q_4^2}{16\alpha q_2}$.

We now consider the 'doubly degenerate' case $q_1 = q_2 = 0$ which implies $\alpha = \beta_3 = 0$.

In[129]:=
```
αeq0βeq0 = defns /. {α → 0, β3 → 0, q1 → 0, q2 → 0}
```

Out[∘]= $\{0, 0, q_4 - 4\gamma_1^2 - 4\gamma_2^2, q_3 - 4\gamma_1^4 - 8\gamma_1^2\gamma_2^2 - 4\gamma_2^4\}$

In[130]:=
```
GroebnerBasis[αeq0βeq0, {q1, q2, q3, q4}, {γ1, γ2}, MonomialOrder → EliminationOrder]
```

Out[∘]= $\{-4 q_3 + q_4^2\}$

Just as before, we get the additional solvability condition $4 q_3 = q_4{}^2$. We now find the equations for γ_1 and γ_2 by imposing this additional condition.

In[131]:=
```
αeq0βeq0 /. {q3 → q4^2 / 4}
```

Out[∘]= $\{0, 0, q_4 - 4\gamma_1^2 - 4\gamma_2^2, \frac{q_4^2}{4} - 4\gamma_1^4 - 8\gamma_1^2\gamma_2^2 - 4\gamma_2^4\}$

In[132]:=
```
GroebnerBasis[{0, 0, q4 - 4 γ1^2 - 4 γ2^2, q4^2/4 - 4 γ1^4 - 8 γ1^2 γ2^2 - 4 γ2^4}, {q4, γ1, γ2}]
```

Out[∘]= $\{q_4 - 4\gamma_1^2 - 4\gamma_2^2\}$

We again get a continuum of solutions contingent on $q_4 \geq 0$.

In summary, in both cases $q_1 > 0$ and $q_1 = 0$, we get $\beta_3 = \sqrt{q_1}$ and we get the same set of equations for the parameters α, γ_1 and γ_2.

If $q_2 > 0$, we get the solvability condition $-q_2^4 + 4 q_2^2 q_3 - 16 q_3^2 + 2 q_2^3 q_4 - 8 q_2 q_3 q_4 - 2 q_2^2 q_4^2 + 8 q_3 q_4^2 + 2 q_2 q_4^3 - q_4^4 >= 0$. The parameters are given by $\alpha = \frac{\sqrt{q_2}}{4}$, $\gamma_2{}^2 = \frac{-q_2^4+4 q_2^2 q_3-16 q_3^2+2 q_2^3 q_4-8 q_2 q_3 q_4-2 q_2^2 q_4^2+8 q_3 q_4^2+2 q_2 q_4^3-q_4^4}{4 q_2^3}$, $\gamma_1 = \frac{-q_2^2+8 q_3+2 q_2 q_4-2 q_4^2}{16 \alpha q_2}$.

If $q_2 = 0$, we get the solvability condition $q_4{}^2 = 4 q_3$ and the parameters are given by $\alpha = 0$, $\gamma_1^2 + \gamma_2^2 = \frac{q_4}{4}$.

Example 7.4

In[133]:=
```
n = 3;
vars = Table[x[i], {i, 1, n}];
f = Sum[2 i x[i]^3, {i, 1, n}] + (3 x[1]^2 × x[2] - x[2]^3) - 12 x[1] × x[2] × x[3]
```

Out[∘]= $2 x[1]^3 + 3 x[1]^2 x[2] + 3 x[2]^3 - 12 x[1] \times x[2] \times x[3] + 6 x[3]^3$

This is an explicit numerical example.

In[134]:=
```
Γ = Simplify[Table[D[f, x[i], x[j], x[k]], {k, 1, n}, {j, 1, n}, {i, 1, n}] / 6];
u =  Simplify[Table[D[Laplacian[f, vars], x[i]], {i, 1, n}] / 6];
f3 = (n + 2) f - 3 (u.vars) (vars.vars);
𝒟 = Simplify[Table[D[f3, x[i], x[j], x[k]], {k, 1, n}, {j, 1, n}, {i, 1, n}] / 6];
Clear[H, J, K, L, M, Q];
v = Simplify[TensorContract[(𝒟⊗𝒟)⊗𝒟, {{1, 4}, {2, 5}, {3, 7}, {6, 8}}]];
w = Simplify[𝒟.u.u];
Q = Simplify[TensorContract[𝒟⊗𝒟, {{1, 4}, {2, 5}}]];
γuu = w;
FundamentalValues = {H[2] → Simplify[Tr[Q]],
   H[4] → Simplify[Tr[Q.Q]], J[2] → Simplify[u.u], L[4] → Simplify[γuu.u]};
SecondaryValues =
  { H[6] → Simplify[v.v], H[10] - Simplify[𝒟.v.v.v] , J[4] - Simplify[u.Q.u],
   K[4] - Simplify[Tr[Q.(𝒟.u)]], J[6] - Simplify[(u.Q).γuu], K[6] - Simplify[v.w],
   L[6] - Simplify[(u.Q).v], M[6] - Simplify[γuu.γuu], H[8] - Simplify[(u.Q).(Q.v)]};
```

In[145]:=
```
FundamentalValues
```

Out[•]= $\{H[2] \to 1060,\ H[4] \to 518\,384,\ J[2] \to 56,\ L[4] \to -4528\}$

In[146]:=
```
Specialization = FundamentalRelations /. FundamentalValues
```

Out[•]=
$$\Big\{1060 - 10\left(6\alpha^2 + \beta_3^2 + 10\gamma_1^2 + 10\gamma_2^2\right),\ 518\,384 - 32\alpha^2\beta_3^2 - \left(32\alpha^2 + 6\beta_3^2\right)^2 - 800\alpha^2\gamma_2^2 - 4\left(13\alpha^2 + \beta_3^2 - 10\alpha\gamma_1 + 25\gamma_1^2 + 25\gamma_2^2\right)^2 - 4\left(\beta_3^2 + (\alpha + 5\gamma_1)^2 + 25\gamma_2^2\right)^2,\ 56 - 16\alpha^2 - \beta_3^2,\ -4528 - 2\left(-48\alpha^2\beta_3^2 + \beta_3^4 + 32\alpha^3(3\alpha - 5\gamma_1)\right)\Big\}$$

In[147]:=
```
GroebnerBasis[Specialization, Coeffs]
```

Out[•]=
$$\left\{332 + 15\beta_3^2,\ 3\,173\,103\,609 + 125\,768\,785\gamma_2^2,\ -52\,993\,421\,209 + 1\,509\,225\,420\gamma_1^2,\ 230\,203\alpha - 85\,849\gamma_1\right\}$$